
Dragi prieteni,

Manualul este elaborat în conformitate cu Curriculumul disciplinar de informatică şi 
are drept scop însuşirea de către elevi a cunoştinţelor necesare pentru formarea culturii 
informaţionale şi dezvoltarea gîndirii algoritmice. Cu ajutorul acestui manual veţi studia 
funcţiile şi procedurile limbajului PASCAL, structurile dinamice de date şi metodele de 
elaborare a produselor program. De asemenea, veţi studia cele mai răspîndite tehnici de 
programare: trierea, tehnica Greedy, reluarea, metoda desparte şi stăpîneşte, programarea 
dinamică, metoda ramifi că şi mărgineşte, algoritmii euristici. În manual sînt expuse meto-
de de estimare a necesarului de memorie şi a timpului cerut de algoritmi, recomandări 
ce vizează utilizarea recursiei şi iterativităţii.

Modul de expunere a materialului este similar celui din manualele de informatică 
pentru clasele precedente. Mai întîi se prezintă sintaxa şi semantica unităţilor respective 
ale limbajului PASCAL, urmate de exemple de aplicare şi recomandări pentru elaborarea 
de programe ce pot fi  lansate pe calculator. În cazul tehnicilor de programare, sînt expuse 
noţiunile de bază şi suportul matematic al tehnicii respective, urmate de modul de orga-
nizare a datelor, descrierea algoritmilor, elaborarea şi depanarea programelor PASCAL. 
O atenţie deosebită se acordă metodelor de implementare a tehnicilor de programare, in-
terdependenţei dintre performanţele calculatorului şi complexitatea problemelor ce pot 
fi  rezolvate cu ajutorul mijloacelor respective.

Implementarea tehnicilor de programare este ilustrată cu ajutorul mai multor proble-
me frecvent întîlnite în viaţa cotidiană şi studiate în cadrul disciplinelor şcolare din ciclul 
liceal. Totodată, în manual au fost incluse şi probleme de o reală importanţă practică, 
rezolvarea cărora este posibilă doar cu aplicarea calculatorului.

Fiind strîns legate de cunoştinţele din alte domenii, temele din manual sînt axate pe 
metodele de rezolvare a problemelor ce necesită un volum foarte mare de calcul, evi-
denţiindu-se rolul esenţial al gîndirii matematice în apariţia şi dezvoltarea informati-
cii. Exemplele, exerciţiile şi sarcinile individuale din manual vor contribui la perceperea 
adecvată a rolului şi  locului calculatorului, a infl uenţei lui asupra dezvoltării matema-
ticii, fi zicii, chimiei, ştiinţelor socioumane. Pentru majoritatea temelor din manual au 
fost elaborate programe destinate instruirii asistate de calculator, fapt ce permite indivi-
dualizarea procesului de predare–învăţare, organizarea lecţiilor practice şi dezvoltarea 
capacităţilor creative ale fi ecărui elev.

În ansamblu, materialul inclus în Manualul de informatică pentru clasa a XI-a va contri-
bui la dezvoltarea următoarelor competenţe: analiza structurală a problemei;  divizarea 
problemelor complexe în probleme mai simple şi reducerea lor la cele deja rezolvate; 
estimarea complexităţii algoritmilor destinaţi soluţionării problemelor propuse; utiliza-
rea metodelor formale pentru elaborarea algoritmilor şi scrierea programelor respective.

Evident, aceste calităţi sînt strict necesare nu numai viitorilor informaticieni, dar şi 
fi ecărui om cult care, la sigur, va trăi şi va lucra într-un mediu bazat pe cele mai moderne 
tehnologii informaţionale.

Autorul

Capitolul 1

FUNCŢII  ŞI  PROCEDURI

1.1. Subprograme
E cunoscut faptul că o problemă complexă poate fi  rezolvată prin divizarea ei 

într-un set de părţi mai mici (subprobleme). Pentru fi ecare parte se scrie o anumită 
secvenţă de instrucţiuni, denumită subprogram. Problema în ansamblu se rezolvă 
cu ajutorul programului principal, în care pentru rezolvarea subproblemelor se folo-
sesc apelurile subprogramelor respective. Cînd în programul principal se întîlneşte 
un apel, execuţia continuă cu prima instrucţiune din programul apelat (fi g. 1.1). Cînd 
se termină executarea instrucţiunilor din subprogram, se revine la instrucţiunea ime-
diat următoare apelului din programul principal.

 Programul
principal

 Apeluri de 
subprograme

Subprogram 1

Subprogram 2

Programul
principal

Fig. 1.1. Interacţiunea între program și subprogram

În limbajul PASCAL există două tipuri de subprograme, şi anume, funcţii şi pro-
ceduri:

<Subprograme> ::= { <Funcţie>; | <Procedură>; }

Funcţiile sînt subprograme care calculează şi returnează o valoare. Limbajul 
PASCAL conţine un set de funcţii predefi nite, cunoscute oricărui program: sin, 
cos, eof etc. În completare, programatorul poate defi ni funcţii proprii, care se ape-
lează în acelaşi mod ca şi funcţiile-standard. Prin urmare, conceptul de funcţie extin-
de noţiunea de expresie PASCAL.

Procedurile sînt subprograme care efectuează prelucrarea datelor comunicate în 
momentul apelului. Limbajul conţine procedurile predefi nite read, readln, wri-
te, writeln ş.a., studiate în clasele precedente. În completare, programatorul poate 
defi ni proceduri proprii, care se apelează în acelaşi mod ca procedurile-standard. 
Prin urmare, conceptul de procedură extinde noţiunea de instrucţiune PASCAL.

Subprogramele se defi nesc, în întregime, în partea declarativă a unui program. 
Evident, apelurile de funcţii şi proceduri se includ în partea executabilă a programului.

Un subprogram poate fi  apelat chiar de el însuşi, caz în care apelul este recursiv.

Întrebări şi exerciţii
Explicaţi termenii program principal și subprogram.
Cum interacţionează programul și subprogramul?
Care este diferenţa dintre proceduri și funcţii?
Cum se apelează o funcţie? În care instrucţiuni ale limbajului pot apărea apeluri de funcţii?
Cum se apelează o procedură?
Numiţi tipul argumentului și tipul rezultatului furnizat de funcţiile predefi nite abs, chr, 

eof, eoln, exp, ord, sin, sqr, sqrt, pred, succ, trunc.
Numiţi tipul parametrilor actuali ai procedurilor read și write.
Ce prelucrări de date efectuează procedurile read și write?

1.2. Funcţii
Textul PASCAL al unei declaraţii de funcţie are forma:

function f(x1, x2, ..., xn) : tr;
D;
 begin
  ...
  f := e;
  ...
 end;

Prima linie este antetul funcţiei, format din:
f — numele funcţiei;
(x1, x2, ..., xn) — lista opţională de parametri formali reprezentînd argumentele 

funcţiei;
tr — tipul rezultatului; acesta trebuie să fi e numele unui tip simplu sau tip referinţă.

Antetul este urmat de corpul funcţiei, format din declaraţiile locale opţionale D 
şi instrucţiunea compusă begin ... end.

Declaraţiile locale sînt grupate în secţiunile (eventual vide)  label, const, type, 
var, function/procedure.

Numele f al funcţiei apare cel puţin o dată în partea stîngă a unei instrucţiuni de 
atribuire care se execută: f := e. Ultima valoare atribuită lui f va fi  întoarsă în progra-
mul principal.

În mod obişnuit, un parametru formal din lista (x1, x2, ..., xn) are forma:

v1, v2, ..., vk : tp

unde v1, v2, ..., vk sînt identifi catori, iar tp este un nume de tip.
Utilizarea funcţiei f se specifi că printr-un apel de forma:

f (a1, a2, ..., an)

unde (a1, a2, ..., an) este lista de parametri actuali. De obicei, parametrii actuali sînt expresii, 
valorile cărora sînt comunicate funcţiei. Corespondenţa între un parametru actual şi para-
metrul formal se face prin poziţia ocupată de aceştia în cele două liste. Parametrul actual 
trebuie să fi e compatibil din punctul de vedere al atribuirii cu tipul parametrului formal.

Exemplu:

Program P97;
 {Declararea şi utilizarea funcţiei Putere }
 type Natural=0..MaxInt;
 var a : real;
   b : Natural;
   c : real;
   s : integer;
   t : integer;
   v : real;

 function Putere(x : real; n : Natural) : real;
  {calcularea lui x la puterea n }
 var p : real;
   i : integer;
 begin
  p:=1;
  for i:=1 to n do p:=p*x;
  Putere:=p;
 end; { Putere }

 begin
  a:=3.0;
  b:=2;
  c:=Putere(a, b);
  writeln(a:10:5, b:4, c:10:5);

  s:=2;
  t:=4;
  v:=Putere(s, t);
  writeln(s:5, t:4, v:10:5);
  readln;
 end.

Funcţia Putere are doi parametri formali: x de tipul real şi n de tipul Natural. 
Funcţia returnează o valoare de tipul real. În corpul funcţiei sînt declarate variabi-
lele locale p şi i.

La execuţia apelului Putere(a,b) valorile 3.0 şi 2 ale parametrilor actuali a, b 
se transmit parametrilor formali, respectiv, x şi n. De menţionat că tipul lui a coinci-
de cu tipul lui x şi tipul lui b coincide cu tipul lui n.

În cazul apelului Putere(s,t) tipul parametrilor actuali s,t nu coincide cu 
tipul parametrilor formali, respectiv, x şi n. Totuşi apelul este corect, întrucît tipurile 
respectve sînt compatibile din punctul de vedere al atribuirii.

Întrebări şi exerciţii
Se consideră următoarea declaraţie:

function Factorial(n : integer) : integer;
var p, i : integer;
begin
  p:=1;
  for i:=1 to n do p:=p * i;
  Factorial:=p;
end;

 Numiţi tipul parametrului formal și tipul rezultatului returnat de funcţie. Precizaţi varia-
bilele declarate în corpul funcţiei. Elaboraţi un program care afi șează pe ecran valorile n! 
pentru n = 2, 3 și 7.

În care loc al programului principal se includ declaraţiile de funcţii?
Comentaţi programul ce urmează:

Program P98;
 { Eroare }
function Factorial(n : 0..7) : integer;
var p, i : integer;
begin
 p:=1;
 for i:=1 to n do p:=p*i;
 Factorial:=p;
end; { Factorial }
begin
 writeln(Factorial(4));
 readln;
end.

Se consideră antetul

function F(x : real; y : integer; z : char) : boolean;

 Care din apelurile ce urmează sînt corecte:

a) F(3.18, 4, ’a’) e) F(3.18, 4, 4)

b) F(4, 4, ’4’) f ) F(’3.18’, 4, ’4’)

c) F(4, 4, 4) g) F(15, 21, ’3’)

d) F(4, 3.18, ’a’) h) F(15,21,3)

Elaboraţi o funcţie care calculează:
 a) suma numerelor reale a, b, c, d;
 b) media numerelor întregi i, j, k, m;
 c) minimumul din numerele a, b, c, d;
 d) numărul de vocale într-un șir de caractere;
 e) numărul de consoane într-un șir de caractere;
 f ) rădăcina ecuaţiei ax + b = 0;
 g) cel mai mic divizor al numărului întreg n > 0, diferit de 1;
 h) cel mai mare divizor comun al numerelor naturale a, b;
 i) cel mai mic multiplu comun al numerelor naturale a, b;
 j) ultima cifră în notaţia zecimală a numărului întreg n > 0;
 k) cîte cifre sînt în notaţia zecimală a numărului întreg n > 0;
 l) cifra superioară în notaţia zecimală a numărului întreg n > 0;
 m) numărul de apariţii ale caracterului dat într-un șir de caractere.
Se consideră următoarele declaraţii:

const nmax=100;
type Vector=array [1..nmax] of real;

 Elaboraţi o funcţie care calculează:
 a) suma componentelor unui vector;
 b) media componentelor vectorului;
 c) componenta maximă;
 d) componenta minimă.
Se consideră următoarele tipuri de date:

type Punct=record
  x, y : real
      end;
     Segment=record
    A, B : Punct

       end;

     Triunghi=record
     A, B, C : Punct
    end;
     Dreptunghi=record
       A, B, C, D : Punct
      end;
     Cerc=record
  Centru : Punct;
  Raza : real
     end;

 Elaboraţi o funcţie care calculează:
 a) lungimea segmentului;
 b) lungimea cercului;
 c) aria cercului;
 d) aria triunghiului;
 e) aria dreptunghiului.
Variabila A este introdusă prin declaraţia

var A : set of char;

 Elaboraţi o funcţie care returnează numărul de caractere din mulţimea A.
Elaboraţi o funcţie care să calculeze diferenţa în secunde între două momente de timp 

date prin oră, minute și secunde.
Un triunghi este defi nit prin coordonatele vîrfurilor sale. Scrieţi funcţii care, pentru două 

triunghiuri date, să studieze dacă:
 a) au aceeași arie;
 b) sînt asemenea;
 c) primul este în interiorul celui de-al doilea.

1.3. Proceduri
Forma generală a textului unei declaraţii de procedură este:

procedure p(x1, x2, ..., xn);
 D;
begin
...
end;

În antetul procedurii apar:
p — numele procedurii;
x1, x2, ..., xn — lista opţională de parametri formali;
În corpul procedurii sînt incluse:
D — declaraţiile locale (opţionale) grupate după aceleaşi reguli ca şi în cazul func-

ţiilor;

begin ... end — instrucţiune compusă; ea nu conţine vreo atribuire asupra 
numelui procedurii.

Procedura poate să întoarcă mai multe rezultate, dar nu prin numele ei, ci prin 
variabile desemnate special (cu prefi xul var) în lista de parametri formali.

Parametrii din listă introduşi prin declaraţii de forma

v1, v2, ..., vk : tp

se numesc parametri-valoare. Aceştia servesc pentru transmiterea de valori din pro-
gramul principal în procedură.

Parametrii formali introduşi în listă prin declaraţii de forma

var v1, v2, ..., vk : tp

se numesc parametri-variabilă şi servesc pentru întoarcerea rezultatelor din proce-
dură în programul principal.

Activarea unei proceduri se face printr-un apel de forma

p(a1, a2, ..., an)

unde a1, a2, ..., an este lista de parametri actuali. Spre deosebire de funcţie, 
apelul de procedură este o instrucţiune; aceasta se inserează în programul principal 
în locul în care sînt dorite efectele produse de execuţia procedurii.

În cazul unui parametru-valoare drept parametru actual poate fi  utilizată orice 
expresie de tipul respectiv, în particular o constantă sau o variabilă. Modifi cările 
parametrilor-valoare nu se transmit în exteriorul subprogramului.

În cazul unui parametru-variabilă drept parametri actuali pot fi  utilizate numai 
variabile. Evident, modifi cările parametrilor în studiu vor fi  transmise programului 
apelant.

Exemplu:

Program P99;
 {Declararea şi utilizarea procedurii Lac }
var a, b, c, t, q : real;

procedure Lac(r : real; var l, s : real);
 {lungimea şi aria cercului        }
 {r - raza; l - lungimea; s - aria }
const Pi=3.14159;
begin
  l:=2*Pi*r;
  s:=Pi*sqr(r);
end; {Lac }

begin
  a:=1.0;
  Lac(a, b, c);
  writeln(a:10:5, b:10:5, c:10:5);

  Lac(3.0, t, q);
  writeln(3.0:10:5, t:10:5, q:10:5);

  readln;
end.

Procedura Lac are trei parametri formali: r, l şi s. Parametrul r este un parame-
tru-valoare, iar l şi s sînt parametri-variabilă.

Execuţia instrucţiunii Lac(a,b,c) determină transmiterea valorii 1.0 drept va-
loare a parametrului formal r şi a locaţiilor (adreselor) variabilelor b şi c drept locaţii 
(adrese) ale parametrilor formali l şi s. Prin urmare, secvenţa de instrucţiuni

a:=1.0;
Lac(a, b, c)

este echivalentă cu secvenţa

b:=2*Pi*1.0;
c:=Pi*sqr(1.0).

În mod similar, instrucţiunea

Lac(3.0,t,q)

este echivalentă cu secvenţa

t:=2*Pi*3.0;
q:=Pi*sqr(3.0).

Întrebări şi exerciţii
Care este diferenţa dintre un parametru-valoare și un parametru-variabilă?
Se consideră declaraţiile:

var k, m, n : integer;
  a, b, c : real;
procedure P(i : integer; var j : integer;
       x : real; var y : real);
begin
 {...}
end.

Care din apelurile ce urmează sînt corecte?

a) P(k,m,a,b) d) P(m,m,a,b)

b) P(3,m,a,b) e) P(m,k,6.1,b)

c) P(k,3,a,b) f ) P(n,m,6,b)

g) P(n,m,6,20) i) P(i,i,i,i)

h) P(a,m,b,c) j) P(a,a,a,a)

 Argumentaţi răspunsul.

Comentaţi programul ce urmează:

Program P100;
 {Eroare }
var a : real; 
  b : integer;
procedure P(x : real; var y : integer);
begin
 {... }
end; { P }
begin
 P(a, b);
 P(0.1, a);
 P(1, b);
 P(a, 1);
end.

Ce va afi șa pe ecran programul ce urmează?

Program P101;
 {Parametru-valoare şi parametru-variabilă }
var a, b : integer;

procedure P(x : integer; var y : integer);
begin
 x:=x+1;
 y:=y+1;
 writeln(’x=’, x, ’ y=’, y);
end; {P }

begin
 a:=0;
 b:=0;
 P(a, b);
 writeln(’a=’, a, ’ b=’, b);
 readln;
end.

 Argumentaţi răspunsul.

Elaboraţi o procedură care:
 a) calculează rădăcinile ecuaţiei ax2 + bx + c = 0;
 b) radiază dintr-un șir caracterul indicat în apel;
 c) încadrează un șir de caractere între simbolurile „#”;

 d) ordonează componentele unui tablou array [1..100] of real în ordine cres-
cătoare;

 e) ordonează componentele unui fi șier fi le of integer în ordine descrescătoare;
 f) calculează și depune într-un tablou numerele prime mai mici decît un număr natural dat n.
Se consideră următoarele tipuri de date

type Data = record
              Ziua : 1..31;
              Luna : 1..12;
              Anul : integer;
            end;
     Persoana = record
                  NumePrenume : string;
                  DataNasterii : Data;
                end;
     ListaPersoane = array [1..50] of Persoana;

 Elaboraţi o procedură care primește din programul principal o listă de persoane și resti-
tuie:

 a) persoanele născute în ziua z a lunii;
 b) persoanele născute în luna l a anului;
 c) persoanele născute în anul a;
 d) persoanele născute pe data z.l.a;
 e) persoana cea mai în vîrstă;
 f ) persoana cea mai tînără;
 g) vîrsta fi ecărei persoane în ani, luni, zile;
 h) lista persoanelor care au mai mult de v ani;
 i) lista persoanelor în ordine alfabetică;
 j) lista persoanelor ordonată conform datei nașterii;
 k) lista persoanelor de aceeași vîrstă (născuţi în același an).
Elaboraţi o procedură care:
 a) creează o copie de rezervă a unui fi șier text;
 b) exclude dintr-un fi șier text liniile vide;
 c) numerotează liniile unui fi șier text;
 d) concatenează două fi șiere text într-unul singur;
 e) concatenează n fi șiere text (n >2) într-unul singur.
Vom numi mari numerele naturale care conţin mai mult de 20 de cifre semnifi cative. Să 

se defi nească un tip de date pentru numerele naturale mari și să se scrie proceduri care 
să adune și să scadă astfel de numere.

1.4. Domenii de vizibilitate
Corpul unui program sau subprogram se numeşte bloc. Deoarece subprogramele 

sînt incluse în programul principal şi pot conţine la rîndul lor alte subprograme, re-

zultă că blocurile pot fi  imbricate (incluse unul în altul). Această imbricare de blocuri 
este denumită structura de bloc a programului PASCAL.

Într-o structură fi ecărui bloc i se ataşează cîte un nivel de imbricare. Programul 
principal este considerat de nivel 0, un bloc defi nit în programul principal este de 
nivel 1. În general, un bloc defi nit în nivelul n este de nivelul n + 1.

Pentru exemplifi care, în fi gura 1.2 este prezentată structura de bloc a programu-
lui P105.

Fig. 1.2. Structura de bloc a unui program PASCAL

Program P105;                                 {nivel 0}
 { Structura de bloc a programului }
var a : real;
{1}

    procedure P(b : real);               {nivel 1}
    var c : real;
    {2}

        procedure Q(d : integer);   {nivel 2}
        {3}
        var c : char;
        {4}
        begin
          c:=chr(d);
          writeln(’In procedura Q  c=’, c);
        end; {5}

    begin
      writeln(’b=’, b);
      c:=b+1;
      writeln(’In procedura P  c=’, c);
      Q(35);
    end; {6}

    function F(x : real) : real;         {nivel 1}
    begin
      f:=x/2;
    end;

begin
   a:=F(5);
   writeln(’a=’, a);
   P(a);
   readln;
end {7}.

De regulă, un bloc PASCAL include declaraţii de etichete, variabile, funcţii, para-
metri ş.a.m.d. O declaraţie introduce un nume, care poate fi  o etichetă sau un identi-
fi cator. O declaraţie dintr-un bloc poate redefi ni un nume declarat în exteriorul lui. 
În consecinţă, în diferite părţi ale programului unul şi acelaşi nume poate desemna 
obiecte diferite.

Prin domeniul de vizibilitate al unei declaraţii se înţelege textul de program, 
în care numele introdus desemnează obiectul specifi cat de declaraţia în studiu. 
Domeniul de vizibilitate începe imediat după terminarea declaraţiei şi se sfîrşeşte 
odată cu textul blocului respectiv. Deoarece blocurile pot fi  imbricate, domeniul de 
vizibilitate nu este neapărat o porţiune continuă din textul programului. Domeniul 
de vizibilitate al unei declaraţii dintr-un bloc inclus acoperă domeniul de vizibilitate 
al declaraţiei ce implică acelaşi nume din blocul exterior.

De exemplu, în programul P105 domeniul de vizibilitate al declaraţiei var a : 
real este textul cuprins între punctele marcate {1} şi {7}. Domeniul de vizibilitate 
al declaraţiei var c : real este format din două fragmente de text cuprinse între 
{2}, {3} şi {5}, {6}. Domeniul de vizibilitate al declaraţiei var c : char este 
textul cuprins între {4} şi {5}.

Cunoaşterea domeniilor de vizibilitate ale declaraţiilor este necesară pentru de-
terminarea obiectului curent desemnat de un nume.

De exemplu, identifi catorul c din instrucţiunea

c:=chr(d)

a programului P105 desemnează o variabilă de tip char. Acelaşi identifi cator din 
instrucţiunea

c:=b+1

desemnează o variabilă de tip real.
De reţinut că declaraţia unui nume de funcţie/procedură se consideră terminată 

la sfîrşitul antetului. Prin urmare, domeniul de vizibilitate al unei astfel de declaraţii 
include şi corpul funcţiei/procedurii respective. Acest fapt face posibil apelul re-
cursiv: în corpul funcţiei/procedurii aceasta poate fi  referită, fi ind vizibilă. Evident, 
declaraţia unui parametru formal este vizibilă numai în corpul subprogramului res-
pectiv.

De exemplu, domeniul de vizibilitate al declaraţiei procedure Q este textul 
cuprins între punctele marcate {3}şi {6}. Domeniul de vizibilitate al declaraţiei 
d:integer este textul cuprins între {3}şi {5}.

Întrebări şi exerciţii
Cum se determină domeniul de vizibilitate al unei declaraţii?
Determinaţi domeniile de vizibilitate ale declaraţiilor b : real și x : real din pro-

gramul P105 (fi g. 1.2).
Precizaţi structura de bloc a programului ce urmează. Indicaţi domeniul de vizibilitate al 

fi ecărei declaraţii și determinaţi obiectele desemnate de fi ecare apariţie a identifi catori-
lor c și x.

Program P106;
  {Redefi nirea constantelor }
 const c=1;

 function F1(x : integer) : integer;
 begin
  F1:=x+c;
 end; { F1 }

 function F2(c : real) : real;
 const x=2.0;
 begin
  F2:=x+c;
 end; { F2 }

 function F3(x : char) : char;
 const c=3;
 begin
  F3:=chr(ord(x)+c);
 end; { F3 }
 
 begin
  writeln(’F1=’, F1(1));
  writeln(’F2=’, F2(1));
  writeln(’F3=’, F3(’1’));
  readln;
 end.

 Ce va afi șa pe ecran programul în studiu?

Determinaţi domeniile de vizibilitate ale identifi catorilor P și F din programul P105 (fi g. 1.2).
Comentaţi programul ce urmează:

Program P107;
 { Eroare }
 var a : real;

 procedure P(x : real);
 var a : integer;
 begin
  a:=3.14;
  writeln(x+a);
 end; { P }

 begin
  a:=3.14;
  P(a);
 end.

Cum se determină obiectul desemnat de apariţia unui nume într-un program PASCAL?

1.5. Comunicarea prin variabile globale
Execuţia unui apel de subprogram presupune transmiterea datelor de prelucrat 

funcţiei sau procedurii respective. După executarea ultimei instrucţiuni din subpro-
gram, rezultatele produse trebuie întoarse în locul de apel. Cunoaştem deja că datele de 
prelucrat şi rezultatele produse pot fi  transmise prin parametri. Parametrii formali se 
specifi că în antetul funcţiei sau procedurii, iar parametrii actuali — în locul apelului.

În completare la modul de transmitere a datelor prin parametri, limbajul PASCAL 
permite comunicarea prin variabile globale.

Orice variabilă este locală în subprogramul în care a fost declarată. O variabilă 
este globală relativ la un subprogram atunci cînd ea este declarată în programul 
sau subprogramul ce îl cuprinde fără să fi e redeclarată în subprogramul în studiu. 
Întrucît variabilele globale sînt cunoscute atît în subprogram, cît şi în afara lui, ele 
pot fi  folosite pentru transmiterea datelor de prelucrat şi returnarea rezultatelor.

Exemplu:

Program P108;
  {Comunicarea prin variabile globale }
 var a,            {variabilă globală în P }
     b : real;     {variabilă globală în P, F }

 procedure P;
 var c : integer;  {variabilă locală în P }
 begin
  c:=2;
  b:=a*c;
 end; { P }

 function F : real;
 var a : 1..5;     {variabilă locală în F }
 begin
  a:=3;
  F:=a+b;
 end; { F }

 begin
  a:=1;
  P;
  writeln(b);     {se afi şează 2.0000000000E+00 }
  writeln(F);     {se afi şează 5.0000000000E+00 }
  readln;
 end.

Datele de prelucrat se transmit procedurii P prin variabila globală a. Rezultatul 
produs de procedură se returnează în blocul de apel prin variabila globală b. Valoarea 

argumentului funcţiei F se transmite prin variabila globală b. Menţionăm că varia-
bila a este locală în F şi nu poate fi  folosită pentru transmiterea datelor în această 
funcţie.

De obicei, comunicarea prin variabile globale se utilizează în cazurile în care mai 
multe subprograme prelucrează aceleaşi date. Pentru exemplifi care amintim func-
ţiile cu argumente de tip tablou, procedurile care prelucrează tablouri şi fi şiere de 
angajaţi, persoane, elevi etc.

Întrebări şi exerciţii
Explicaţi termenii variabilă globală relativ la un subprogram și variabilă locală într-un 

subprogram.
Numiţi variabilele globale și variabilele locale din programul P105 (fi g. 1.2).
 Poate fi  oare o variabilă locală în același timp și o variabilă globală relativ la un subpro-

gram?
Numiţi variabilele globale și variabilele locale din programul ce urmează. Ce va afi șa pe 

ecran acest program?

Program P109;
 {Comunicarea prin variabile globale }
 var a : integer;

 procedure P;
 var b, c, d : integer;

 procedure Q;
 begin
   c:=b+1;
 end; { Q }

 procedure R;
 begin
   d:=c+1;
 end; { R }
 begin
  b:=a;
  Q;
  R;
  a:=d;
 end; { P }

 begin
  a:=1;
  P;
  writeln(a);
  readln;
 end.

Se consideră declaraţiile

Type Ora=0..23;
     Grade=-40..+40;
     Temperatura=array [Ora] of Grade;

 Componentele unei variabile de tip Temperatura reprezintă temperaturile măsurate 
din oră în oră pe parcursul a 24 de ore. Elaboraţi o procedură care:

 a) indică maximumul și minimumul temperaturii;
 b) indică ora (orele) la care s-a înregistrat o temperatură maximă;
 c) înscrie ora (orele) la care s-a înregistrat o temperatură minimă într-un fi șier text.
 Comunicarea cu procedurile respective se va face prin variabile globale.
Se consideră fi șiere arbitrare de tip text. Elaboraţi o funcţie care:
 a) returnează numărul de linii dintr-un fi șier;
 b) calculează numărul de vocale dintr-un text;
 c) calculează numărul de cuvinte dintr-un text (cuvintele reprezintă șiruri de caractere 

separate prin spaţiu sau sfîrșit de linie);
 d) returnează lungimea medie a liniilor din text;
 e) calculează lungimea medie a cuvintelor din text;
 f ) returnează numărul semnelor de punctuaţie din text.
 Comunicarea cu funcţiile respective se va face prin variabile globale.

1.6. Efecte colaterale
Destinaţia unei funcţii este să întoarcă ca rezultat o singură valoare. În mod 

obişnuit, argumentele se transmit funcţiei prin parametri-valoare, iar rezultatul 
calculat se returnează în locul de apel prin numele funcţiei. În completare, limba-
jul PASCAL permite transmiterea argumentelor prin variabile globale şi parame-
tri-variabilă.

Prin efect colateral se înţelege o atribuire (în corpul funcţiei) a unei valori la o va-
riabilă globală sau la un parametru formal variabilă. Efectele colaterale pot infl uenţa 
în mod neaşteptat execuţia unui program şi complică procesele de depanare.

Prezentăm în continuare exemple defectuoase de programare, care folosesc func-
ţii cu efecte colaterale.

Program P110;
 {Efect colateral - atribuire la o variabilă globală}
 var a : integer; { variabilă globală }

 function F(x : integer) : integer;
 begin
  F:=a*x;
  a:=a+1;     {atribuire defectuoasă }
 end; { F }
begin
  a:=1;



21

  writeln(F(1));  { se afi şează 1 }
  writeln(F(1));  { se afi şează 2 }
  writeln(F(1));  { se afi şează 3 }
  readln;
 end.

În programul P110 funcţia F returnează valoarea expresiei a*x. Pe lîngă aceasta 
însă, atribuirea a:=a+1 alterează valoarea variabilei globale a. În consecinţă, pentru 
una şi aceeaşi valoare 1 a argumentului x funcţia returnează rezultate diferite, fapt 
ce nu se încadrează în conceptul uzual de funcţie.

Program P111;
 {Efect colateral - atribuire la un parametru formal}
 var a : integer;

 function F(var x : integer) : integer;
 begin
     F:=2*x;
     x:=x+1;    { atribuire defectuoasă }
 end; { F }

 begin
  a:=2;
  writeln(F(a));     { se afi şează 4 }
  writeln(F(a));     { se afi şează 6 }
  writeln(F(a));     { se afi şează 8 }
  readln;
 end.

În programul P111 funcţia F returnează valoarea expresiei 2*x. Întrucît x este 
un parametru formal variabilă, atribuirea x:=x+1 schimbă valoarea parametrului 
actual din apel, şi anume a variabilei a din programul principal. Faptul că apelurile 
textual identice F(a), F(a) şi F(a) returnează rezultate ce diferă poate crea con-
fuzii în procesul depanării.

În cazul procedurilor, atribuirile asupra variabilelor globale produc efecte co-
laterale similare celor discutate pentru astfel de atribuiri la funcţii. Întrucît mijlo-
cul-standard de întoarcere de rezultate din procedură este prin parametri formali 
variabilă, atribuirile asupra unor astfel de parametri nu sînt considerate ca efecte 
colaterale.

Efectele colaterale introduc abateri de la procesul-standard de comunicare, prin 
care variabilele participante sînt desemnate explicit ca parametri formali în decla-
raţie şi parametri actuali în apel. Consecinţele efectelor colaterale se pot propaga 
în domeniul de vizibilitate al declaraţiilor globale şi pot interfera cu cele similare, 
produse la execuţia altor proceduri şi funcţii. În astfel de condiţii, utilizarea varia-
bilelor globale devine riscantă. Prin urmare, la elaborarea programelor complexe se 
vor aplica următoarele recomandări:

1. Comunicarea funcţiilor cu mediul de chemare se va face prin transmiterea de 
date spre funcţie prin parametri formali valoare şi întoarcerea unui singur rezultat 
prin numele ei.

2. Comunicarea procedurilor cu mediul de chemare se va face prin transmiterea 
de date prin parametri formali valoare sau variabilă şi întoarcerea rezultatelor prin 
parametri formali variabilă.

3. Variabilele globale pot fi  folosite pentru transmiterea datelor în subprograme, 
însă valorile lor nu trebuie să fi e schimbate de acestea.

Întrebări şi exerciţii
Care este cauza efectelor colaterale? Ce consecinţe pot avea aceste efecte?
Precizaţi ce vor afi șa pe ecran programele ce urmează:

Program P112;
  {Efecte colaterale }
 var a, b : integer;

  function F(x : integer) : integer;
  begin
  F:=a*x;
  b:=b+1;
  end; { F }

  function G(x : integer) : integer;
  begin
   G:=b+x;
   a:=a+1;
  end; { G }
begin
 a:=1; b:=1;
 writeln(F(1));
 writeln(G(1));
 writeln(F(1));
 writeln(G(1));
 readln;
end.

Program P113;
 {Efecte colaterale }
 var a : integer;
     b : real;

  function F(var x : integer) : integer;
  begin
   F:=x;
   x:=x+1;
  end; { F }

  procedure P(x,y:integer; var z:real);
  begin
   z:=x/y;
  end; { P }

begin
 a:=1;
 P(F(a), a, b);
 writeln(a, ’   ’, b);
 readln;
end.

Program P114;
  {Efecte colaterale }
 var a, b : real;

  procedure P(var x, y : real);
  {Interschimbarea valorilor variabilelor x, y }
  begin
   a:=x;
   x:=y;
   y:=a;
  end; { P }

begin
 a:=1; b:=2;
 P(a, b);
 writeln(a, b);
 a:=3; b:=4;
 P(a, b);
 writeln(a, b);
 readln;
end.

 Cum pot fi  evitate efectele colaterale?

1.7. Recursia
Recursia se defi neşte ca o situaţie în care un subprogram se autoapelează fi e di-

rect, fi e prin intermediul altei funcţii sau proceduri. Subprogramul care se autoape-
lează se numeşte recursiv.

De exemplu, presupunem că este defi nit tipul

type Natural = 0..MaxInt;

Funcţia factorial

dacă
dacă

poate fi  exprimată în PASCAL, urmînd direct defi niţia, în forma:

function F(n : Natural) : Natural;
begin
 if n=0 then F:=1
 else F:=n*F(n-1)
end; {F}

Efectul unui apel F(7) este declanşarea unui lanţ de apeluri ale funcţiei F pentru 
parametrii actuali 6, 5, ..., 2, 1, 0:

F(7) -> F(6) -> F(5) -> ... -> F(1) -> F(0).

Apelul F(0) determină evaluarea directă a funcţiei şi oprirea procesului repe-
titiv; urmează revenirile din apeluri şi evaluarea lui F pentru 1, 2, ..., 6, 7, ultima 
valoare fi ind întoarcerea în locul primului apel.

Funcţia
dacă
dacă

dacă

are ca valori numerele lui Fibonacci. Urmînd defi niţia, obţinem:

function Fib(n:Natural):Natural;
begin
 if n=0 then Fib:=0
 else if n=1 then Fib:=1
            else Fib:=Fib(n-1)+Fib(n-2)
end; {Fib}

Fiecare apel al funcţiei Fib pentru n > 1 generează două apeluri Fib(n-1), 
Fib(n-2) ş.a.m.d., de exemplu:

Fib(4) ->
  Fib(3), Fib(2) -> 
    Fib(2), Fib(1), Fib(1), Fib(0) -> 
      Fib(1), Fib(0).

Din exemplele în studiu se observă că recursia este utilă pentru programarea unor cal-
cule repetitive. Repetiţia este asigurată prin execuţia unui subprogram care conţine un apel 
la el însuşi: cînd execuţia ajunge la acest apel, este declanşată o nouă execuţie ş.a.m.d.

Evident, orice subprogram recursiv trebuie să includă condiţii de oprire a procesu-
lui repetitiv. De exemplu, în cazul funcţiei factorial procesul repetitiv se opreşte cînd n 
ia valoarea 0; în cazul funcţiei Fib procesul se opreşte cînd n ia valoarea 0 sau 1.

La orice apel de subprogram, în memoria calculatorului vor fi  depuse următoa-
rele informaţii:

— valorile curente ale parametrilor transmişi prin valoare;
— locaţiile (adresele) parametrilor-variabilă;
— adresa de retur, adică adresa instrucţiunii ce urmează după apel.
Prin urmare, la apeluri recursive spaţiul ocupat din memorie va creşte rapid, 

riscînd depăşirea capacităţii de memorare a calculatorului. Astfel de cazuri pot fi  
evitate, înlocuind recursia prin iteraţie (instrucţiunile for, while, repeat). Pentru 
exemplifi care prezentăm o formă nerecursivă a funcţiei factorial:

function F(n: Natural): Natural;
var i, p : Natural;
begin
 p:=1;
 for i:=1 to n do p:=p*i;
 F:=p;
end; {F}

Recursia este deosebit de utilă în cazurile în care elaborarea unor algoritmi ne-
recursivi este foarte complicată: translatarea programelor PASCAL în limbajul cod-
maşină, grafi ca pe calculator, recunoaşterea formelor ş.a.

Întrebări şi exerciţii
Cum se execută un subprogram recursiv? Ce informaţii se depun în memoria calculato-

rului la execuţia unui apel recursiv?
Care este diferenţa dintre recursie și iteraţie?
Elaboraţi o formă nerecursivă a funcţiei lui Fibonacci.
Scrieţi un subprogram recursiv care:
 a) calculează suma S(n) = 1 + 3 + 5 + ... + (2n – 1);
 b) calculează produsul P(n) = 1  4  7  ...  (3n – 2);
 c) inversează un șir de caractere;
 d) calculează produsul P(n) = 2  4  6  ...  2n.
Elaboraţi un program care citește de la tastatură numerele naturale m, n și afi șează pe 

ecran valoarea funcţiei lui Ackermann:

dacă
dacă

dacă

 Calculaţi a(0, 0), a(1, 2), a(2, 1) și a(2, 2). Încercaţi să calculaţi a(4, 4) și a(10, 10). Explicaţi 
mesajele afi șate pe ecran.

Se consideră declaraţia

type Vector=array [1..20] of integer;

 Elaboraţi un subprogram recursiv care:
 a) afi șează componentele vectorului pe ecran;
 b) calculează suma componentelor;

 c) inversează componentele vectorului;
 d) calculează suma componentelor pozitive;
 e) verifi că dacă cel puţin o componentă a vectorului este negativă;
 f ) calculează produsul componentelor negative;
 g) verifi că dacă cel puţin o componentă a vectorului este egală cu un număr dat.
Elaboraţi o formă nerecursivă a funcţiei ce urmează:

function S(n:Natural):Natural;
 begin
  if n=0 then S:=0
  else S:=n+S(n-1)
 end; {S}

Scrieţi o funcţie recursivă care returnează valoarea true dacă șirul de caractere s este 
conform defi niţiei

 <Număr> ::= <Cifră> | <Cifră> <Număr>

 Indicaţie. Forma unei astfel de funcţii derivă din formula metalingvistică. Varianta nerecursivă

function N(s : string) : boolean;
var i : integer; 
 p : boolean
begin
 p:=(s<>’’);
 for i=1 to length(s) do
 p:=p and (s[i] in [’0’..’9’]);
 N:=p;
end;

 derivă din defi niţia
 <Număr> ::= <Cifră> {<Cifră>}

Se consideră următoarele formule metalingvistice:
 <Număr> ::= <Cifră> {<Cifră>}
 <Semn> ::= + | –
 <Expresie> ::= <Număr> | <Expresie> <Semn> <Expresie>
 Scrieţi o funcţie recursivă care returnează valoarea true dacă șirul de caractere s este 

conform defi niţiei unităţii lexicale <Expresie>.

1.8. Sintaxa declaraţiilor şi apelurilor de subprograme
În general, defi nirea unei funcţii se face cu ajutorul următoarelor formule meta-

lingvistice:

<Funcţie> ::=    <Antet funcţie>; <Corp> 
                         | <Antet funcţie>; <Directivă>
                         | function <Identifi cator>; <Corp>

<Antet funcţie> ::= 
                      function <Identifi cator> [<Listă parametri formali>] : <Identifi cator>

Diagramele sintactice sînt prezentate în fi gura 1.3.

function

function

Fig. 1.3. Sintaxa declaraţiilor de funcţii

Procedurile se defi nesc cu ajutorul următoarelor formule:

<Procedură>::=<Antet procedură>;<Corp> | <Antet procedură>;<Directivă> 
                                                                         | procedure <Identifi cator>;<Corp>
<Antet procedură> := procedure <Identifi cator> [<Listă parametri formali>]

Diagramele sintactice sînt prezentate în fi gura 1.4.

procedure

procedure

Antet procedură

<Antet procedură>

Fig. 1.4. Sintaxa declaraţiilor de proceduri

Listele de parametri formali au următoarea sintaxă:

<Listă parametri formali> ::= 
                        (<Parametru formal> {; <Parametru formal>})
<Parametru formal> ::= 
                         [var] <Identifi cator> {, <Identifi cator>} : <Identifi cator> 
                         | <Antet funcţie> 
                         | <Antet procedură>

Diagrama sintactică este prezentată în fi gura 1.5.

<Listă parametri formali>

Identificator

Antet funcţie

Antet procedură

Identificator

,

;

( ):var

Fig. 1.5. Diagrama sintactică <Listă parametri formali>

Amintim că în lipsa cuvîntului-cheie var identifi catorii din listă specifi că para-
metrii-valoare. Cuvîntul var prefi xează parametrii-variabilă. Antetul unei funcţii 
(proceduri) din listă specifi că un parametru-funcţie (procedură). În Turbo PASCAL 
astfel de parametri se declară explicit ca aparţinînd unui tip procedural şi au forma 
parametrilor-valoare. Limbajul PASCAL extinde sensul uzual al noţiunii de funcţie, 
permiţînd returnarea valorilor nu numai prin numele funcţiei, ci şi prin parametri-
variabilă.

Un apel de funcţie are forma:

<Apel funcţie> ::= <Nume funcţie > [<Listă parametri actuali>]

iar o instrucţiune apel de procedură:

<Apel procedură> ::= <Nume procedură > [<Listă parametri actuali>]

Parametrii actuali se specifi că cu ajutorul formulelor:

<Listă parametri actuali> ::= (<Parametru actual > {,<Parametru actual>})
<Parametru actual > ::=<Expresie> | <Variabilă> | <Nume funcţie> 
                                                                                    | <Nume procedură >
Diagrama sintactică este prezentată în fi gura 1.6.

<Apel funcţie>

<Apel procedură>

<Nume funcţie>

Nume procedură

Listă parametri actuali

Listă parametri actuali

<Listă parametri actuali>

Expresie

Variabilă

Nume funcţie

Nume procedură

)(

,

Fig. 1.6. Sintaxa apelurilor de funcţii și proceduri

Corespondenţa între un parametru actual şi parametrul formal se face prin pozi-
ţia ocupată de aceştia în cele două liste.

În cazul unui parametru-valoare drept parametru actual poate fi  utilizată ori-
ce expresie, în particular, o constantă sau o variabilă. Expresia respectivă trebuie 
să fi e compatibilă din punctul de vedere al atribuirii cu tipul parametrului formal. 
Modifi cările parametrilor-valoare nu se transmit în exteriorul subprogramului.

În cazul unui parametru-variabilă drept parametri actuali pot fi  utilizate numai va-
riabile. Modifi cările parametrilor-variabilă se transmit în exteriorul subprogramului.

În cazul unui parametru-funcţie (procedură) drept parametru actual poate fi  uti-
lizat orice nume de funcţie (procedură), antetul căreia are forma specifi cată în lista 
parametrilor formali.

Întrebări şi exerciţii
Cînd se utilizează declaraţiile de forma

function <Identifi cator>; <Corp> ?
Indicaţi pe diagramele sintactice din fi gurile 1.3 și 1.5 drumurile care corespund declara-

ţiilor de funcţii din programul P106, paragraful 1.4.
Care este diferenţa dintre un parametru-valoare și un parametru-variabilă?
Indicaţi pe diagramele sintactice din fi gurile 1.4 și 1.5 drumurile care corespund declara-

ţiilor de proceduri din programul P101, paragraful 1.3.
Indicaţi pe diagramele sintactice din fi gurile 1.3–1.6 drumurile care corespund declaraţi-

ilor și apelurilor de subprograme din programul P105, paragraful 1.4.

Capitolul 2

STRUCTURI  DINAMICE
DE  DATE

2.1. Variabile dinamice. Tipul referinţă
Variabilele declarate în secţiunea var a unui program sau subprogram se nu-

mesc variabile statice. Numărul variabilelor statice se stabileşte în momentul scrierii 
programului şi nu poate fi  schimbat în timpul execuţiei. Există însă situaţii în care 
numărul necesar de variabile nu este cunoscut din timp.

De exemplu, presupunem că este necesară prelucrarea datelor referitoare la per-
soanele care formează un şir de aşteptare (o coadă) la o casă de bilete. Lungimea 
cozii este nedefi nită. De fi ecare dată cum apare o persoană nouă, trebuie să se creeze 
o variabilă de tipul respectiv. După ce persoana pleacă, variabila corespunzătoare 
devine inutilă.

Variabilele care sînt create şi eventual distruse în timpul execuţiei programului se 
numesc variabile dinamice.

Accesul la variabilele dinamice se face prin intermediul variabilelor de tip referin-
ţă. De obicei, un tip referinţă se defi neşte printr-o declaraţie de forma:

type Tr = ^Tb;

unde Tr este numele tipului referinţă, iar Tb este tipul de bază. Semnul „^” se citeşte 
„adresă”. Evident, pot fi  utilizate şi tipuri referinţă anonime. Diagrama sintactică 
<Tip referinţă> este prezentată în fi gura 2.1.

<Tip referinţă>

Tip^

Fig. 2.1. Diagrama sintactică <Tip referinţă>

Mulţimea de valori ale unui tip de date referinţă constă din adrese. Fiecare adresă 
identifi că o variabilă dinamică ce aparţine tipului de bază. La această mulţime de 
adrese se mai adaugă o valoare specială, notată nil (zero), care nu identifi că nicio 
variabilă.

Exemplu:

type AdresaInteger=^integer;
     AdresaChar=^char;

var  i : AdresaInteger;
     r : ^real;
     c : AdresaChar;

Valoarea curentă a variabilei i va indica o variabilă dinamică de tipul inte-
ger. Într-un mod similar, variabilele de tip referinţă r şi c identifi că variabile de ti-
pul real şi, respectiv, char. Subliniem faptul că tipurile de date AdresaInteger, 
AdresaChar şi tipul anonim ^real sînt tipuri referinţă distincte.

Operaţiile care se pot face cu valorile unui tip de date referinţă sînt = şi <>. Valorile 
de tip referinţă nu pot fi  citite de la tastatură şi afi şate pe ecran.

Crearea unei variabile dinamice se realizează cu procedura predefi nită new (nou). 
Apelul acestei proceduri are forma

new(p)

unde p este o variabilă de tip referinţă.
Procedura alocă spaţiu de memorie pentru variabila nou -creată şi returnează 

adresa zonei respective prin variabila p. În continuare variabila dinamică poate fi  
accesată prin aşa-zisa dereperare: numele variabilei de tip referinţă p este urmat de 
semnul de „^”. Dereperarea unei variabile de tip referinţă cu conţinutul nil va de-
clanşa o eroare de execuţie.

Exemplu:
new(i); i^:=1 — crearea unei variabile dinamice de tipul integer; variabilei 

create i se atribuie valoarea 1;
new(r); r^:=2.0 — crearea unei variabile dinamice de tipul real; variabilei 

create i se atribuie valoarea 2.0;
new(c); c^:=’*’ — crearea unei variabile dinamice de tipul char; variabilei 

create i se atribuie valoarea ’*’.
Subliniem faptul că variabila dinamică p^ obţinută printr-un apel new(p) este 

distinctă de toate variabilele create anterior. Prin urmare, executarea instrucţiu-
nilor

new(p); new(p); ...; new(p)

conduce la crearea unui şir v1 , v2 , ..., vn de variabile dinamice. Numai ultima variabilă 
creată, vn , este referită prin p^. Întrucît valorile variabilelor de tip referinţă reprezin-
tă adresele anumitor zone din memoria internă a calculatorului, variabilele în studiu 
se numesc indicatori de adresă.

Distrugerea unei variabile dinamice şi eliberarea zonei respective de memorie se 
realizează cu procedura predefi nită dispose (a dispune). Apelul acestei proceduri 
are forma:

dispose(p)

unde p este o variabilă de tip referinţă.

Exemple:

dispose(i); dispose(r); dispose(c)

După executarea instrucţiunii dispose(p) valoarea variabilei de tip referinţă p 
este nedefi nită.

Asupra variabilelor dinamice se pot efectua toate operaţiile admise de tipul de 
bază.

Exemplu:

Program P117;
 {Operaţii cu variabile dinamice }
 type AdresaInteger=^integer;
 var i, j, k : AdresaInteger;
     r, s, t : ^real;
 begin
  {crearea variabilelor dinamice de tipul integer }
  new(i); new(j); new(k);
  {operaţii cu variabilele create }
  i^:=1; j^:=2;
  k^:=i^+j^;
  writeln(k^);
  {crearea variabilelor dinamice de tipul real }
  new(r); new(s); new(t);
  {operaţii cu variabilele create }
  r^:=1.0; s^:=2.0;
  t^:=r^/s^;
  writeln(t^);
  {distrugerea variabilelor dinamice }
  dispose(i); dispose(j); dispose(k);
  dispose(r); dispose(s); dispose(t);
  readln;
 end.

Spre deosebire de variabilele statice, care ocupă zone de memorie stabilite de 
compilator, variabilele dinamice ocupă zone de memorie oferite de procedura new. 
Zonele respective sînt eliberate de procedura dispose şi pot fi  reutilizate pentru 
crearea unor variabile dinamice noi. Prin urmare, procedurile new şi dispose asi-
gură alocarea (rezervarea) dinamică a memoriei: spaţiul de memorie este atribuit 
unei variabile dinamice numai pe durata existenţei ei.

Numărul de variabile dinamice ce pot exista concomitent în timpul execuţiei unui 
program PASCAL depinde de tipul variabilelor şi spaţiul de memorie disponibil. 
În cazul în care tot spaţiul de memorie este deja ocupat, apelul procedurii new va 
declanşa o eroare de execuţie.

Exemplu:

Program P118;
 {Eroare: depăşirea capacităţii memoriei }
 label 1;
 var i : ^integer;

 begin
  1 : new(i);
      goto 1;
 end.

Alocarea dinamică a memoriei necesită o atenţie sporită din partea programato-
rului care este obligat să asigure crearea, distrugerea şi referirea corectă a variabile-
lor dinamice.

Întrebări şi exerciţii
Care este diferenţa dintre variabilele statice și variabilele dinamice?
Cum se identifi că variabilele dinamice?
Indicaţi pe diagrama sintactică din fi gura 2.1 drumurile care corespund declaraţiilor de 

tipuri referinţă din programul P117.
  Se consideră declaraţiile:

type AdresaReal=^real;
var  r : AdresaReal;

 Precizaţi mulţimea de valori ale tipului de date AdresaReal și mulţimea de valori pe 
care le poate lua variabila dinamică r^.

Ce operaţii se pot efectua cu valorile unui tip de date referinţă? Cu variabilele dinamice?
Se consideră declaraţiile:

type AdresaTablou = ^array [1..10] of integer;
var  t : AdresaTablou;

 Precizaţi mulţimea de valori ale tipului de date AdresaTablou și mulţimea de valori 
pe care le poate lua variabila dinamică t^.

Comentaţi programul:

Program P119;
 {Eroare }
 var r, s : ^real;
 begin
  r^:=1; s^:=2;
  writeln(’r^=’, r^, ’  s^=’, s^); 
  readln;
 end.

Elaboraţi un program în care se creează două variabile dinamice de tipul șir de caractere. 
Atribuiţi valori variabilelor create și afi șaţi la ecran rezultatul concatenării șirurilor respective.

 Ce va afi șa pe ecran programul ce urmează?

Program P120;
 var i : ^integer;

 begin
  new(i); i^:=1;
  new(i); i^:=2;
  new(i); i^:=3;
  writeln(i^);
  readln;
 end.

Comentaţi programul:

Program P121;
 {Eroare }
 var i, j : ^integer;
 begin
  new(i);
  i^:=1;
  dispose(i);
  new(j);
  j^:=2;
  dispose(j);
  writeln(’i^=’, i^, ’  j^=’, j^);
  readln;
 end.

 Explicaţi expresia alocarea dinamică a memoriei.

2.2. Structuri de date
O structură de date este formată din datele propriu-zise şi relaţiile dintre ele. În 

funcţie de modul de organizare, o structură de date poate fi  implicită sau explicită.
Tablourile, şirurile de caractere, articolele, fi şierele şi mulţimile studiate în capito-

lele precedente sînt structuri implicite de date. Relaţiile dintre componentele acestor 
sructuri sînt predefi nite şi nemodifi cabile. De exemplu, toate componentele unui şir 
de caractere au un nume comun, iar caracterul s[i+1] este succesorul caracterului 
s[i] în virtutea poziţiei ocupate.

Întrucît structura tablourilor, şirurilor de caractere, articolelor, mulţimilor şi fi şi-
erelor nu se modifi că în timpul execuţiei oricărui program sau subprogram, variabi-
lele respective reprezintă structuri statice de date.

Folosind date cu structură implicită, putem rezolva reprezentativ o clasă limitată 
de probleme. În multe cazuri relaţiile dintre componente nu numai că se modifi că 
dinamic, dar în acelaşi timp pot deveni deosebit de complexe.

De exemplu, în cazul unui fi r de aşteptare la o casă de bilete relaţiile dintre per-
soane se modifi că: persoanele nou-sosite se aşază la rînd; persoanele în criză de timp 
pleacă fară să-şi mai procure bilete; persoanele care au plecat pentru un timp îşi păs-
trează rîndul ş.a.m.d. În cazul proiectării asistate de calculator a reţelelor de circula-

ţie, staţiile, rutele, capacitatea de trafi c ş.a. pot fi  stabilite interactiv de către utilizator. 
În astfel de situaţii utilizarea datelor cu structură implicită devine nenaturală, difi cilă 
şi inefi cientă.

Prin urmare, este necesară folosirea unor structuri de date în care relaţiile dintre 
componente să fi e reprezentate şi prelucrate în mod explicit. Acest efect se poate 
obţine ataşînd fi ecărei componente o informaţie ce caracterizează relaţiile acesteia 
cu alte date ale structurii. În cele mai multe cazuri, informaţia suplimentară, numită 
informaţie de structură, se reprezintă prin variabilele de tipul referinţă.

Structurile de date componentele cărora sînt create şi eventual distruse în timpul exe-
cuţiei programului se numesc structuri dinamice de date. Structurile dinamice frecvent 
utilizate sînt: listele unidirecţionale, listele bidirecţionale, stivele, cozile, arborii ş.a.

O structură de date este recursivă dacă ea poate fi  descompusă în date cu aceeaşi 
structură. Pentru exemplifi care menţionăm listele unidirecţionale şi arborii care vor 
fi  studiaţi în paragrafele următoare.

Întrebări şi exerciţii
Explicaţi termenul structură de date. Daţi exemple.
Care este diferenţa dintre structurile implicite și structurile explicite de date?
O structură de date este omogenă dacă toate componentele sînt de același tip. În caz 

contrar, structura de date este eterogenă. Daţi exemple de structuri omogene și struc-
turi eterogene de date.

Care este diferenţa dintre structurile statice și structurile dinamice de date?
Explicaţi termenul structură recursivă de date.

2.3. Liste unidirecţionale
Listele unidirecţionale sînt structuri explicite şi dinamice de date formate din 

celule. Fiecare celulă este o variabilă dinamică de tipul record ce conţine, în prin-
cipal, două cîmpuri: cîmpul datelor şi cîmpul legăturilor. Cîmpul datelor memo-
rează informaţia prelucrabilă asociată celulei. Cîmpul legăturilor furnizează indi-
catorul de adresă corespunzător celulei la care se poate ajunge din celula curentă. 
Se consideră că orice celulă poate fi  atinsă pornind de la o celulă privilegiată, nu-
mită baza listei.

Pentru exemplifi care, în fi gura 2.2 este prezentată o listă unidirecţională formată 
din 4 celule. Celulele conţin elementele A, B, C şi D.

Datele necesare pentru crearea şi prelucrarea unei liste unidirecţionale pot fi  de-
fi nite prin declaraţii de forma:

type AdresaCelula=^Celula;
 Celula=record
     Info : string;



36

     Urm  : AdresaCelula;
   end;
var P : AdresaCelula;

Informaţia utilă asociată unei celule se memorează în cîmpul Info, iar adresa 
celulei următoare în cîmpul Urm. Pentru simplifi care se consideră că cîmpul Info 
este de tipul şir de caractere. Ultima celulă din listă va avea în cîmpul Urm valoarea 
nil. Adresa primei celule (adresa de bază) este memorată în variabila de tip refe-
rinţă P (fi g. 2.2).

celulă cîmp legături

cîmp
date

indicator
de adresă

nil

DCBA

P

Fig. 2.2. Lista unidirecţională

Subliniem faptul că în defi niţia tipului referinţă AdresaCelula tipul de bază 
Celula încă nu este cunoscut. Conform regulilor limbajului PASCAL, acest lucru 
este posibil numai în cazul variabilelor dinamice, cu condiţia ca tipul de bază să fi e 
defi nit mai tîrziu în aceeaşi declaraţie.

O listă unidirecţională poate fi  creată adăugînd la vîrful listei cîte un element. 
Iniţial lista în curs de construcţie este vidă, adică nu conţine nicio celulă.

Exemplu:

Program P122;
 {Crearea listei unidirecţionale A->B->C->D }
 type AdresaCelula=^Celula;
      Celula=record
                   Info : string;
                    Urm : AdresaCelula;
             end;
 var P,         {adresa de bază }
     R, V : AdresaCelula;
 begin
  {1 - iniţial lista este vidă }
  P:=nil;
  {2 - adăugarea celulei A }
  new(R);       {crearea unei celule }
  P:=R;         {iniţializarea adresei de bază }

  R^.Info:=’A’; {încărcarea informaţiei utile }
  R^.Urm:=nil;  {înscrierea indicatorului ”sfîrşit de listă” }
  V:=R;         {memorarea adresei vîrfului }
  {3 - adăugarea celulei B }
  new(R);       {crearea unei celule }
  R^.Info:=’B’; {încărcarea informaţiei utile }
  R^.Urm:=nil;  {înscrierea indicatorului ”sfîrşit de listă” }
  V^.Urm:=R;    {crearea legăturii A -> B }
  V:=R;         {actualizarea adresei vîrfului }
  {4 - adăugarea celulei C }
  new(R);       {crearea unei celule }
  R^.Info:=’C’; {încărcarea informaţiei utile }
  R^.Urm:=nil;  {înscrierea indicatorului ”sfîrşit de listă” }
  V^.Urm:=R;    {crearea legăturii B -> C }
  V:=R;         {actualizarea adresei vîrfului }
  {5 - adăugarea celulei D }
  new(R);       {crearea unei celule }
  R^.Info:=’D’; {încărcarea informaţiei utile }
  R^.Urm:=nil;  {înscrierea indicatorului ”sfîrşit de listă” }
  V^.Urm:=R;    {crearea legăturii C -> D }
  V:=R;         {actualizarea adresei vîrfului }
  {afi şarea listei create }
  R:=P;
  while R<>nil do begin
                    writeln(R^.Info);
                    R:=R^.Urm
                  end;
  readln;
 end.

Procesul de construire a listei în studiu este prezentat în fi gura 2.3. Variabila V 
(adresa vîrfului) din programul P122 reţine adresa ultimei celule deja create pentru 
a-i poziţiona indicatorul de adresă V^.Urm. Se procedează astfel pentru că în mo-
mentul în care completăm cîmpurile R^.Info şi R^.Urm ale celulei curente încă nu 
se cunoaşte adresa celulei ce urmează.

Listele cu un număr arbitrar de celule pot fi  create şi afi şate cu ajutorul procedu-
rilor respective din programul P123:

Program P123;
{ Crearea listelor unidirectionale}
type AdresaCelula=^Celula;
     Celula=record
             Info : string;
              Urm : AdresaCelula;
            end;

A

1 – iniţial lista este vidă

2 – adăugarea celulei A

3 – adăugarea celulei B

4 – adăugarea celulei C

5 – adăugarea celulei D

B

B C

A

A

A B C D

P

P V

P V

P V

VV

Fig. 2.3. Crearea listei unidirecţionale

var p,q,r : AdresaCelula;
    s : string;
    i: integer;
procedure Creare;
begin
 p:=nil;
write(’s=’); readln(s);
 new(r); r^.Info:=s; r^.Urm:=nil;
 p:=r; q:=r;
 write(’s=’);
  while not eof do
   begin
     readln(s);write(’s=’);
     new(r); r^.Info:=s; r^.Urm:=nil;
     q^.Urm:=r; q:=r
   end;
end; { Creare }
procedure Afi sare;
begin
 r:=p;
 while r<>nil do
   begin
     writeln(r^.Info);
     r:=r^.Urm;
   end;
readln;
end; { Afi sare }

begin
 Creare;
 Afi sare;
end.

Orice listă unidirecţională poate fi  defi nită recursiv după cum urmează:
a) o celulă este o listă unidirecţională;
b) o celulă ce conţine o legătură către o altă listă unidirecţională este o listă unidi-

recţională.
Pentru a sublinia faptul că listele unidirecţionale sînt structuri recursive de date, 

declaraţiile respective pot fi  transcrise în forma:

type  Lista=^Celula;
 Celula=record
          Info : string;
          Urm  : Lista
       end;
var P : Lista;

Proprietăţile listelor unidirecţionale pot fi  reproduse parţial prin memorarea ele-
mentelor respective într -un tablou unidimensional. De exemplu, datele din fi gura 2.2 
pot fi  reprezentate în forma:

var L : array [1..4] of string;
   ...
 L[1]:= ’A’;
 L[2]:= ’B’;
 L[3]:= ’C’;
 L[4]:= ’D’;
 ...

Însă o astfel de reprezentare nu permite crearea şi prelucrarea structurilor de date 
cu un număr arbitrar de elemente.

Întrebări şi exerciţii
Cum se defi nesc datele necesare pentru crearea unei liste?
Care este destinaţia cîmpului datelor din componenţa unei celule? Care este destinaţia 

cîmpului legăturilor? Ce informaţie se înscrie în acest cîmp?
Scrieţi un program care creează lista unidirecţională din fi gura 2.2, adăugînd cîte un ele-

ment la baza listei.
Elaboraţi o procedură care schimbă cu locul două elemente din listă.
De la tastatură se citesc numere întregi diferite de zero. Se cere să se creeze două liste, 

una a numerelor negative, iar alta – a numerelor pozitive.
Prin ce se explică faptul că listele unidirecţionale sînt structuri recursive de date?

2.4. Prelucrarea listelor unidirecţionale
Operaţiile frecvent utilizate în cazul listelor unidirecţionale sînt:
– parcurgerea listei şi prelucrarea informaţiei utile asociate fi ecărei celule;
– căutarea unui anumit element, identifi cat prin valoarea sa;
– includerea (inserarea) unui element într-un anumit loc din listă;
– excluderea (ştergerea) unui element dintr-o listă ş.a.
Presupunem că există o listă nevidă (fi g. 2.2) defi nită prin declaraţiile:

type  AdresaCelula=^Celula;
   Celula=record;
        Info : string;
        Urm  : AdresaCelula
       end;

Variabila P indică adresa de bază a listei în studiu.
Parcurgerea listei se realizează conform următoarei secvenţe de instrucţiuni:

R:=P; {poziţionare pe celula de bază }
while R<>nil do
begin
 {prelucrarea informaţiei din cîmpul R^.Info }
 R:=R^.Urm; { poziţionare pe celula următoare }
end;

Căutarea celulei ce conţine elementul specifi cat de variabila Cheie este realizată 
de secvenţa:

R:=P;
while R^.Info<>Cheie do R:=R^.Urm;

Adresa celulei în studiu va fi  reţinută în variabila R.
Subliniem faptul că această secvenţă se execută corect numai în cazul în care lista 

include cel puţin o celulă ce conţine informaţia căutată. În caz contrar, se ajunge la 
vîrful listei, variabila R ia valoarea nil, iar dereperarea R^ provoacă o eroare de 
execuţie. Pentru a evita astfel de erori, se utilizează secvenţa:

 R:=P;
 while R<>nil do
 begin
  if R^.Info=Cheie then goto 1;
  R:=R^.Urm
 end;
1: ...

Întrucît listele unidirecţionale sînt structuri recursive de date, operaţia de căutare 
poate fi  realizată şi de un subprogram recursiv:

type Lista=^AdresaCelula;
  Celula=record;
           Info : string;
            Urm : Lista;
         end;
var P : Lista;
...
function Caut(P : Lista; Cheie : string):Lista;
begin
  if P=nil then Caut:=nil
           else
             if P^.Info=Cheie then Caut:=P 
                              else Caut:=Caut(P^.Urm, Cheie)
end;

Funcţia Caut returnează adresa de bază a sublistei ce conţine în prima celulă, 
dacă există, elementul specifi cat de parametrul Cheie.

Includerea celulei referite de indicatorul Q după celula referită de indicatorul R 
(fi g. 2.4) este realizată de secvenţa de instrucţiuni:

Q

R

...
...

Q

R

...
...

a)

b)

Fig. 2.4. Includerea unui element în listă:
a – lista pînă la includere; b – lista după includere

Q^.Urm:=R^.Urm;
R^.Urm:=Q;

Excluderea celulei R din listă necesită afl area adresei Q a celulei precedente şi 
modifi carea indicatorului de adresă Q^.Urm (fi g. 2.5):

Q:=P;
while Q^.Urm<>R do Q:=Q^.Urm;
Q^.Urm:=R^.Urm;

Menţionăm că includerea sau excluderea elementului din baza listei necesită ac-
tualizarea adresei de bază P.

Exemplu:

Program P124;
 {Crearea şi prelucrarea unei liste unidirecţionale }
 type AdresaCelula=^Celula;
      Celula=record
               Info : string;
              Urm : AdresaCelula;
             end;

RQ

...
...

Q

...
...

a)

b)

Fig. 2.5. Excluderea unui element din listă:
a – lista pînă la excludere; b – lista după excludere

 var P : AdresaCelula;  {adresa de bază }
     c : char;
procedure Creare;
var R, V : AdresaCelula;
begin 
  if P<>nil then writeln(’Lista există deja’)
    else begin
           writeln(’Daţi lista:’);
           while not eof do
             begin
               new(R);
               readln(R^.Info);
               R^.Urm:=nil;
               if P=nil then begin P:=R; V:=R end
                     else begin V^.Urm:=R; V:=R end;
        end;
        end;
end; {Creare }

procedure Afi s;
var R : AdresaCelula;
begin
  if P=nil then writeln(’Lista este vidă’)
           else begin
                  writeln(’Lista curentă:’);
                  R:=P;
                  while R<>nil do 

      begin
        writeln(R^.Info);
        R:=R^.Urm;
      end;
         end;
  readln;
 end; {Afi s }

procedure Includ;
label 1;
var Q, R : AdresaCelula;
   Cheie : string;
begin
  new(Q);
  write(’Daţi elementul ce urmează’);
  writeln(’ să fi e inclus:’);
  readln(Q^.Info);
  write(’Indicaţi elementul după care’);
  writeln(’ se va face includerea:’);
  readln(Cheie); 
  R:=P;
  while R<>nil do
     begin
       if R^.Info=Cheie then goto 1;
       R:=R^.Urm;
     end;
 1:if R=nil then begin
                   writeln(’Element inexistent’);
                   dispose(Q);
                 end;
            else begin
                   Q^.Urm:=R^.Urm;
                   R^.Urm:=Q;
                 end;
end; { Includ }

procedure Exclud;
label 1;
var Q, R : AdresaCelula;
   Cheie : string;
begin
  write(’Daţi elementul ce urmează’);
  writeln(’ să fi e exclus:’);
  readln(Cheie);
  R:=P;
  Q:=R;

  while R<>nil do
     begin
       if R^.Info=Cheie then goto 1;
       Q:=R;
       R:=R^.Urm;
     end;
 1:if R=nil then writeln(’Element inexistent’)
            else begin
                   if R=P then P:=R^.Urm else Q^.Urm:=R^.Urm;
                   dispose(R);
                 end;
end; { Exclud }
begin
  P:=nil; { iniţial lista este vidă }
  repeat
    writeln(’Meniu:’);
    writeln(’C - Crearea listei’);
    writeln(’I - Includerea elementului’);
    writeln(’E - Excluderea elementului’);
    writeln(’A - Afi şarea listei la ecran’);
    writeln(’O - Oprirea programului’);
    write(’Opţiunea=’); readln(c);
    case c of
        ’C’ : Creare;
        ’I’ : Includ;
        ’E’ : Exclud;
        ’A’ : Afi s;
        ’O’ :
        else writeln(’Opţiune necunoscută’)
    end;
  until c=’O’;
 end.

Procedura Creare formează o listă unidirecţională cu un număr arbitrar de ce-
lule. Informaţia utilă asociată fi ecărei celule se citeşte de la tastatură. Procedura Afi s 
afi şează elementele listei la ecran. Procedura Includ citeşte de la tastatură elemen-
tul ce urmează să fi e inclus şi elementul după care se va face includerea. În continua-
re se caută celula ce conţine elementul specifi cat, după care, dacă există, este inclusă 
celula nou -creată. Procedura Exclud caută şi elimină, dacă există, celula ce conţine 
elementul citit de la tastatură.

Întrebări şi exerciţii
Scrieţi o funcţie nerecursivă care returnează adresa vîrfului listei unidirecţionale. 

Transcrieţi această funcţie într-o formă recursivă.
Transcrieţi procedurile Includ și Exclud din programul P124, fără a utiliza instrucţi-

unea goto.

Se consideră următoarele tipuri de date:

type  AdresaCandidat=^Candidat;
   Candidat=record
        NumePrenume : string;
        NotaMedie : real;
        Urm : AdresaCandidat
          end;

 Elaboraţi un program care:
 a) creează o listă unidirecţională cu componente de tipul Candidat;
 b) afi șează lista pe ecran;
 c) exclude din listă candidatul care își retrage actele;
 d) include în listă candidatul care depune actele;
 e) afi șează pe ecran candidaţii cu media mai mare de 7,5;
 f ) creează o listă suplimentară formată din candidaţii cu media mai mare de 9,0;
 g) exclude din listă toţi candidaţii cu media mai mică de 6,0.
Elaboraţi o procedură care:
 a) reordonează elementele listei unidirecţionale conform unui anumit criteriu;
 b) concatenează două liste unidirecţionale;
 c) descompune o listă în două liste;
 d) selectează din listă elementele care corespund unui anumit criteriu.
Elementele listei unidirecţionale sînt memorate într-un tablou unidirecţional. Elaboraţi 

procedurile necesare pentru:
 a) parcurgerea listei;
 b) căutarea unui anumit element;
 c) includerea unui element;
 d) excluderea unui element.
 Care sînt avantajele și neajunsurile acestei reprezentări? Se consideră că lista va conţine 

cel mult 100 de elemente.
Scrieţi o funcţie recursivă ce returnează numărul de celule dintr-o listă unidirecţională.
Scrieţi un subprogram recursiv care exclude o anumită celulă din listă.
Prin cuvînt se înţelege orice secvenţă nevidă formată din literele alfabetului latin. 

Elaboraţi un program care formează lista cuvintelor întîlnite într-un fi șier text și calculea-
ză numărul de apariţii al fi ecărui cuvînt. Examinaţi cazurile:

 a) cuvintele se includ în listă în ordinea primei lor apariţii în text;
 b) cuvintele se includ în listă în ordine alfabetică.
 Se consideră că literele mari și mici sînt identice.

2.5. Stiva
Prin stivă (în limba engleză stack) înţelegem o listă unidirecţională cu proprieta-

tea că operaţiile de introducere şi extragere a elementelor se fac la un singur capăt 

al ei. Poziţia ocupată în stivă de ultimul element introdus poartă numele de vîrf. O 
stivă fără niciun element se numeşte stivă vidă.

Pentru exemplifi care, în fi gura 2.6 este prezentată o stivă care conţine elementele 
A, B, C.

S

C

B

A

S

a)

b)

C

B

A

Fig. 2.6. Stiva:
a – reprezentarea detaliată; b – reprezentarea generalizată

Datele necesare pentru crearea şi prelucrarea unei stive pot fi  defi nite prin decla-
raţii de forma:

type AdresaCelula=^Celula;
    Celula=record
          Info : string;
          Prec : AdresaCelula
          end;
var  S : AdresaCelula;

Adresa vîrfului stivei este memorată în variabila de tip referinţă S. Adresa celulei 
precedente din stivă este memorată în cîmpul Prec.

Operaţia de introducere a unui element în stivă (fi g. 2.7) este efectuată de secven-
ţa de instrucţiuni:

new(R); { crearea unei celule }

{încărcarea informaţiei utile în cîmpul R^.Info }

R^.Prec:=S; { crearea legăturii către celula precedentă 
              din stivă }

S:=R; { actualizarea adresei vîrfului }

unde R este o variabilă de tipul AdresaCelula.

S

S

a)

C

B

A

c)

B

A

b)

C

D

B

A

S

Fig. 2.7. Introducerea și extragerea elementelor din stivă:
a – stiva iniţială; b – introducerea elementului D; c – extragerea elementelor D, C

Extragerea unui element din stivă (fi g. 2.7) este efectuată de secvenţa:

R:=S; { memorarea adresei celulei extrase }

 { prelucrarea informaţiei din cîmpul R^.Info }

S:=S^.Prec; { eliminarea celulei din stivă }
dispose(R); { distrugerea celulei extrase }

Exemplu:

Program P127;
 {Crearea şi prelucrarea unei stive }
 type AdresaCelula=^Celula;
      Celula=record
                   Info : string;
                   Prec : AdresaCelula;
            end;
 var S : AdresaCelula; {adresa vîrfului }
     c : char;
 procedure Introduc;
 var R : AdresaCelula;
 begin
  new(R);
  write(’Daţi elementul ce urmează’);
  writeln(’ să fi e introdus:’);
  readln(R^.Info);
  R^.Prec:=S;
  S:=R;
 end; { Includ }

 procedure Extrag;
 var R : AdresaCelula;
 begin
  if S=nil then writeln(’Stiva este vidă’)
           else begin
                 R:=S;
                 write(’Este extras’);
                 writeln(’ elementul:’);
                 writeln(R^.Info);
                 S:=S^.Prec;
                 dispose(R);
                end;
 end; { Extrag }

 procedure Afi s;
 var R : AdresaCelula;
 begin
  if S=nil then writeln(’Stiva este vidă’)
    else begin
           writeln(’Stiva include elementele:’);
           R:=S;
           while R<>nil do begin
                             writeln(R^.Info);
                             R:=R^.Prec;
                           end;
         end;
  readln;
 end; { Afi s }
 begin
  S:=nil;  { iniţial stiva este vidă }
  repeat
    writeln(’Meniu:’);
    writeln(’I - Introducerea elementului;’);
    writeln(’E - Extragerea elementului’);
    writeln(’A - Afi şarea stivei pe ecran’);
    writeln(’O - Oprirea programului’);
    write(’Opţiunea=’); readln(c);
    case c of
        ’I’ : Introduc;
        ’E’ : Extrag;
        ’A’ : Afi s;
        ’O’ :
        else writeln(’Opţiune necunoscută’)
    end;
  until c=’O’;
 end.

Stivele mai poartă şi numele de liste LIFO (last in, fi rst out — ultimul element care 
a intrat în stivă va fi  primul care va ieşi din ea) şi sînt frecvent utilizate pentru aloca-
rea dinamică a memoriei în cazul procedurilor şi funcţiilor recursive. Evident, stivele 
pot fi  simulate utilizînd tablourile unidimensionale array [1..n] of ..., însă o 
astfel de reprezentare este limitată de cele n componente ale tablourilor.

Întrebări şi exerciţii
Care este ordinea de introducere și extragere a datelor din stivă?
De la tastatură se citesc mai multe numere naturale. Afi șaţi numerele în studiu pe ecran 

în ordinea inversă citirii.
În fi gura 2.8 este reprezentată schema de manevrare a vagoanelor de tren într-un de-

pou. Elaboraţi un program care citește de la tastatură și afi șează pe ecran datele despre 
fi ecare vagon intrat sau ieșit din depou. Datele în studiu includ:

 – numărul de înmatriculare (integer);
 – staţia de înmatriculare (string);
 – anul fabricării (1960..2000);
 – tipul vagonului (string);
 – capacitatea de încărcare (real);
 – proprietarul vagonului (string).

Intrare Ieşire

Linie moartă

Fig. 2.8. Schema de manevrare a vagoanelor de tren

Colectivele temporare de muncă sînt formate și desfi inţate în ordinea „ultimul angajat va 
fi  primul care va fi  concediat”. Elaboraţi un program care citește de la tastatură și afi șează 
pe ecran datele despre fi ecare persoană angajată sau concediată. Datele în studiu includ:

 – numele (string);
 – prenumele (string);
 – anul nașterii (1930..1985);
 – data angajării (ziua, luna, anul).
Se consideră șiruri fi nite de caractere formate din parantezele (, ), [, ], {, }. Un șir este co-

rect numai atunci cînd el poate fi  construit cu ajutorul următoarelor reguli:
 a) șirul vid este corect;



51

 b) dacă A este un șir corect, atunci (A), [A] și {A} sînt șiruri corecte;
 c) dacă A și B sînt șiruri corecte, atunci AB este un șir corect.
 De exemplu, șirurile ( ), [ ], { }, [( )], ((({[ ]}))([ ])) sînt corecte, iar șirurile ([, ( )[ ]{{, ([)] nu sînt 

corecte. Elaboraţi un program care verifi că dacă șirul citit de la tastatură este corect.
 Indicaţie. Problema poate fi  rezolvată printr-o singură parcurgere a șirului supus verifi cării. 

Dacă caracterul curent este (, [ sau {, el este depus în stivă. Dacă vîrful stivei și caracterul 
curent formează una din perechile ( ), [ ] sau { }, paranteza respectivă este scoasă din stivă. 
În cazul unui șir corect, după examinarea ultimului caracter din șir stiva rămîne vidă.

Elementele stivei sînt memorate într-un tablou unidimensional. Elaboraţi procedurile ne-
cesare pentru introducerea și extragerea elementelor din stivă. Care sînt avantajele și ne-
ajunsurile acestei reprezentări? Se consideră că stiva conţine cel mult 100 de elemente.

2.6. Cozi
Prin coadă (în engleză queue) înţelegem o listă unidirecţională în care toate in-

troducerile se efectuează la unul din capete, iar extragerile se efectuează la celălalt 
capăt. O coadă fără niciun element se numeşte coadă vidă.

Pentru exemplifi care, în fi gura 2.9 este prezentată o coadă care conţine elementele A, B, C.

a)

P

PU

U

C B A

b)

ABC

Fig. 2.9. Coada:
a – reprezentarea detaliată; b – reprezentarea generalizată

Datele necesare pentru crearea şi prelucrarea unei cozi pot fi  defi nite prin decla-
raţii de forma:

type  AdresaCelula=^Celula;
   Celula=record
        Info : string;

        Urm : AdresaCelula
    end;
var P, U : AdresaCelula;

Adresa primului element din coadă este memorată în variabila de tip referinţă P, 
iar adresa ultimului element în variabila U. Adresa celulei următoare din coadă este 
memorată în cîmpul Urm.

Operaţia de introducere a unui element (fi g. 2.10) este efectuată de secvenţa de 
instrucţiuni:

new(R); {crearea unei celule }

{încărcarea informaţiei utile în cîmpul R^.Info }

R^.Urm:=nil; { înscrierea indicatorului ”ultimul element” }
U^.Urm:=R;   { adăugarea celulei la coadă }
U:=R;        { actualizarea adresei ultimei celule }

a)

PU

ABC

b)

PU

B ACD

c)

PU

CD

Fig. 2.10. Introducerea și extragerea elementelor din coadă:
a – coada iniţială; b – introducerea elementului D;

c – extragerea elementelor A, B

Extragerea unui element din coadă (fi g. 2.10) este efectuată de secvenţa:

R:=P; { memorarea adresei primei celule }

{prelucrarea informaţiei din cîmpul R^.Info }

P:=P^.Urm; {eliminarea primei celule }
dispose(R); {distrugerea celulei extrase }

Exemplu:

Program P128;
 {Crearea şi prelucrarea unei cozi }
 type AdresaCelula=^Celula;
      Celula=record
               Info : string;
               Urm : AdresaCelula;
             end;
 var P,                {adresa primului element }
     U : AdresaCelula; {adresa ultimului element }
     c : char;
 
 procedure Introduc;
 var R : AdresaCelula;
 begin
  new(R);
  write(’Daţi elementul ce urmează’);
  writeln(’ să fi e introdus:’);
  readln(R^.Info);
  R^.Urm:=nil;
  if P=nil then begin P:=R; U:=R end
           else begin U^.Urm:=R; U:=R end;
 end; { Introduc }
 
 procedure Extrag;
 var R : AdresaCelula;
 begin
  if P=nil then writeln(’Coada este vidă’)
           else begin
                  R:=P;
                  write(’Este extras’);
                  writeln(’ elementul:’);
                  writeln(R^.Info);
                  P:=P^.Urm;
                  dispose(R);
                end;
 end; { Extrag }

 procedure Afi s;
 var R : AdresaCelula;
 begin
  if P=nil then writeln(’Coada este vidă’)
           else begin
                 write(’Coada include’);
                 writeln(’ elementele:’);
                 R:=P;
                 while R<>nil do
                  begin
                    writeln(R^.Info);
                    R:=R^.Urm;
                  end;
                end;
  readln;
 end; { Afi s }
 
 begin
  P:=nil; U:=nil; { iniţial coada este vidă }
  repeat

    writeln(’Meniu:’);
    writeln(’I - Introducerea elementului;’);
    writeln(’E - Extragerea elementului;’);
    writeln(’A - Afi şarea cozii la ecran;’);
    writeln(’O - Oprirea programului’);
    write(’Opţiunea=’); readln(c);
    case c of
        ’I’ : Introduc;
        ’E’ : Extrag;
        ’A’ : Afi s;
        ’O’ :
        else writeln(’Opţiune necunoscută’)
    end;
  until c=’O’;
 end.

Cozile mai poartă numele de liste FIFO (fi rst in, fi rst out — primul element intrat 
în coadă va fi  primul ieşit din coadă). Menţionăm că simularea cozilor cu ajutorul ta-
blourilor unidimensionale este inefi cientă din cauza migrării elementelor cozii spre 
ultima componentă a tabloului.

Întrebări şi exerciţii
 Elaboraţi o funcţie care returnează numărul elementelor unei cozi.
Avioanele care solicită aterizarea pe o anumită pistă a unui aeroport formează un fi r de 

așteptare. Elaboraţi un program care citește de la tastatură și afi șează pe ecran datele 

despre fi ecare avion care solicită aterizarea și avionul care aterizează. Datele în studiu 
includ:

 – numărul de înmatriculare (integer);
 – tipul avionului (string);
 – numărul rutei (integer).
Prin coadă cu priorităţi vom înţelege o coadă în care elementul de introdus se inserează 

nu după ultimul element al cozii, ci înaintea tuturor elementelor cu o prioritate mai mică. 
Priorităţile elementelor se indică prin numere întregi. Elaboraţi un program care:

 a) creează o coadă cu priorităţi;
 b) introduce în coadă elementele specifi cate de utilizator;
 c) extrage elementele din coadă;
 d) afi șează coada cu priorităţi pe ecran.

2.7. Arbori binari
Prin nod se înţelege o variabilă dinamică de tipul record care conţine un cîmp 

destinat memorării informaţiilor utile şi doi indicatori de adresă.
Arborele binar se defi neşte recursiv după cum urmează:
a) un nod este un arbore binar;
b) un nod ce conţine legături către alţi doi arbori binari este un arbore binar.
Prin convenţie, arborele vid nu conţine niciun nod. Pentru exemplifi care, în fi gu-

ra 2.11 este prezentat un arbore binar nodurile căruia conţin informaţia utilă A, B, C, 
D, E, F, G, H, I, J. Datele necesare pentru crearea şi prelucrarea unui arbore binar pot 
fi  defi nite prin declaraţii de forma:

type AdresaNod=^Nod;
   Nod=record
      Info : string;
      Stg, Dr : AdresaNod
     end;
var  T : AdresaNod;

Pentru a sublinia faptul că arborii binari sînt structuri recursive de date, declara-
ţiile în studiu pot fi  transcrise în forma:

type Arbore=^Nod;
   Nod=record
      Info : string;
      Stg, Dr : Arbore;
    end;
var  T : Arbore;

Nodul spre care nu este îndreptată nicio legătură se numeşte rădăcină. Adresa 
rădăcinii se memorează în variabila de tip referinţă T. În cazul unui arbore vid 
T=nil.

Cei doi arbori conectaţi la rădăcină se numesc subarborele stîng şi subarborele 
drept. Adresa subarborelui stîng se memorează în cîmpul Stg, iar adresa subarbo-
relui drept — în cîmpul Dr.

Nivelul unui nod este, prin convenţie, 0 pentru nodul-rădăcină şi i + 1, pentru no-
dul conectat la un nod de nivelul i. În mod obişnuit, în reprezentarea grafi că a unui 
arbore binar nodurile se desenează pe niveluri: rădăcina se afl ă pe nivelul 0, vîrfurile 
conectate la rădăcină — pe nivelul 1 ş.a.m.d. (fi g. 2.11).

D E F

H I J

T

nivel 0

nivel 1

nivel 2

nivel 3

D

B

E F

H

G

JI

A

a)

Fig. 2.11. Arborele binar:
a – reprezentarea detaliată; b – reprezentarea generalizată

Nodurile de pe nivelul i + 1, conectate la un nod de pe nivelul i, se numesc des-
cendenţii acestuia. În fi gura 2.11 nodul B este descendentul stîng, iar nodul C este 
descendentul drept al nodului A; nodul D este descendentul stîng, iar nodul E — 
descendentul drept al nodului B ş.a.m.d.

Dacă un nod x este descendentul altui nod y, îl numim pe acesta din urmă pă-
rintele nodului x. În fi gura 2.11 nodul A este părintele nodurilor B şi C; nodul B este 
părintele nodurilor D şi E ş.a.m.d.

Un nod la care nu este conectat niciun subarbore este un nod terminal. În caz con-
trar, nodul este neterminal. Prin înălţimea arborelui binar înţelegem numărul de nivel 
maxim asociat nodurilor terminale. Arborele din fi gura 2.11 are înălţimea 3; nodurile 
D, H, F, I şi J sînt noduri terminale; nodurile A, B, C, E şi G sînt noduri neterminale.

Arborii binari pot fi  construiţi în memoria calculatorului cu ajutorul algoritmilor 
iterativi sau algoritmilor recursivi.

Algoritmul iterativ creează nodurile în ordinea apariţiei lor pe niveluri:
– se creează nodul-rădăcină;
– nodul-rădăcină se introduce într-o coadă;
– pentru fi ecare nod extras din coadă se creează, dacă există, descendentul stîng 

şi descendentul drept;
– nodurile nou-create se introduc în coadă;
– procesul de construire a arborelui se încheie cînd coada devine vidă.
Nodurile arborelui din fi gura 2.11 vor fi  create de algoritmul iterativ în ordinea: 

A, B, C, D, E, F, G, H, I, J.
Un algoritm similar poate fi  utilizat pentru parcurgerea arborelui binar şi afi şarea 

nodurilor respective pe ecran:
– se creează o coadă care conţine un singur element – nodul-rădăcină;
– fi ecare nod extras din coadă este afi şat pe ecran;
– descendenţii nodului extras se introduc în coadă;
– procesul de afi şare se încheie cînd coada devine vidă.

Exemplu:

Program P129;
 {Crearea unui arbore binar - iteraţie }
 type AdresaNod=^Nod;
      Nod=record
                Info : string;
             Stg, Dr : AdresaNod
          end;

      AdresaCelula=^Celula;
      Celula=record
                   Info : AdresaNod;
                   Urm : AdresaCelula
             end;
 
 var  T : AdresaNod;    {rădăcina }
    Prim,               {primul element din coadă }
  Ultim : AdresaCelula; { ultimul element din coadă }

 procedure IntroduInCoada(Q : AdresaNod);
 var R : AdresaCelula;
 begin
 new(R);
  R^.Info:=Q;
  R^.Urm:=nil;
  if Prim=nil then begin Prim:=R; Ultim:=R end
           else begin Ultim^.Urm:=R; Ultim:=R end;
 end; {IntroduInCoada}
 
 procedure ExtrageDinCoada(var Q : AdresaNod);
 var R : AdresaCelula;
 begin
  if Prim=nil then writeln(’Coada este vidă’)
           else begin
                  R:=Prim;
                  Q:=R^.Info;
                  Prim:=Prim^.Urm;
                  dispose(R);
                end;
 end; { ExtrageDinCoada }
 procedure CreareArboreBinar;
 var R, Q : AdresaNod;
        s : string;
 begin
  T:=nil;                {iniţial arborele este vid }
  Prim:=nil; Ultim:=nil; {iniţial coada este vidă }
  writeln(’Daţi rădăcina:’); readln(s);
  if s<>’’ then
      begin
      new(R); {crearea rădăcinii }
      R^.Info:=s;
      T:=R;   {iniţializarea adresei rădăcinii }
      IntroduInCoada(T);
     end;
  while Prim<>nil do  {cît coada nu e vidă }
     begin
      ExtrageDinCoada(R);
      writeln(’Daţi descendenţii nodului’, R^.Info);
      write(’  stîng:  ’); readln(s);
      if s=’’ then R^.Stg:=nil
         else
             begin
              new(Q); R^.Stg:=Q;
              Q^.Info:=s;
              IntroduInCoada(Q);
             end; { else }

     write(’  drept:  ’); readln(s);
      if s=’’ then R^.Dr:=nil
         else
             begin
              new(Q); R^.Dr:=Q;
              Q^.Info:=s;
              IntroduInCoada(Q);
             end; { else }
     end; { while }
 end; { CreareArboreBinar }
 procedure Afi sareArboreBinar;
 var R : AdresaNod;
 begin
  if T=nil then writeln(’Arbore vid’)
     else
      begin
       writeln(’Arborele este format din:’);
       Prim:=nil; Ultim:=nil;
       IntroduInCoada(T);
       while Prim<>nil do
             begin
              ExtrageDinCoada(R);
              writeln(’Nodul  ’, R^.Info);
              write(’  descendenţi: ’);
              if R^.Stg=nil then write(’nil, ’)
                 else begin
                       write(R^.Stg^.Info, ’, ’);
                       IntroduInCoada(R^.Stg);
                      end;
 
              if R^.Dr=nil then writeln(’nil’)
                 else begin
                       writeln(R^.Dr^.Info);
                       IntroduInCoada(R^.Dr);
                      end;
             end; { while }
      end; { else }
  readln;
 end; { Afi sareArboreBinar }
 
 begin
  CreareArboreBinar;
  Afi sareArboreBinar;
 end.

Informaţia utilă asociată fi ecărui nod se citeşte de la tastatură. Absenţa descen-
dentului se semnalează prin apăsarea tastei <ENTER> (programul citeşte de la tas-

tatură un şir vid de caractere). Menţionăm că coada creată de programul P129 nu 
conţine nodurile propriu-zise, ci adresele acestor noduri.

Algoritmul recursiv construieşte arborii binari urmînd direct defi niţia respec-
tivă:

– se creează nodul-rădăcină;
– se construieşte subarborele stîng;
– se construieşte subarborele drept.
Nodurile arborelui binar din fi gura 2.11 vor fi  create de algoritmul recursiv în 

ordinea: A, B, D, E, H, C, F, G, I, J.

Exemplu:

Program P130;
 {Crearea unui arbore binar - recursie }
 type Arbore=^Nod;
      Nod=record
                Info : string;
              Stg, Dr : Arbore
          end;
 
 var  T : Arbore;    {rădăcina }
 
 function Arb : Arbore;
  {crearea arborelui binar }
 var R : Arbore;
     s : string;
 begin
  readln(s);
  if s=’’ then Arb:=nil
     else begin
          new(R);
          R^.Info:=s;
          write(’Daţi descendentul stîng’);
          writeln(’ al nodului ’, s, ’:’);
          R^.Stg:=Arb;
          write(’Daţi descendentul drept’);
          writeln(’al nodului ’, s, ’:’);
          R^.Dr:=Arb;
          Arb:=R;
         end;
 end; {Arb }

procedure Afi sArb(T : Arbore; nivel : integer);
  {afi sarea arborelui binar }
 var i : integer;
begin
  if T<>nil then

     begin
      Afi sArb(T^.Stg, nivel+1);
      for i:=1 to nivel do write(’    ’);
      writeln(T^.Info);
      Afi sArb(T^.Dr, nivel+1);
     end;
 end; {Afi sareArb }
 
 begin
  writeln(’Daţi rădăcina:’);
  T:=Arb;
  Afi sArb(T, 0);
  readln;
 end.

Funcţia Arb citeşte de la tastatură informaţia utilă asociată nodului în curs de cre-
are. Dacă se citeşte un şir vid, nu se creează niciun nod şi funcţia returnează valoarea 
nil. În caz contrar, funcţia creează un nod, înscrie şirul de caractere în cîmpul Info 
şi returnează adresa nodului. În momentul cînd trebuie completate cîmpurile Stg 
(adresa subarborelui stîng) şi Dr (adresa subarborelui drept), funcţia se autoapelea-
ză, trecînd astfel la construcţia subarborelui respectiv.

Procedura Afi sArb afi şează arborele binar pe ecran. Se afi şează subarborele stîng, 
rădăcina şi apoi subarborele drept. Nivelul fi ecărui nod este redat prin inserarea 
numărului respectiv de spaţii.

Comparînd programele P129 şi P130, se observă că prelucrarea structurilor re-
cursive de date, şi anume a arborilor binari, este mai naturală şi mai efi cientă în cazul 
utilizării unor algoritmi recursivi.

Arborii binari au numeroase aplicaţii, una dintre cele specifi ce fi ind reprezentarea 
expresiilor în scopul prelucrării acestora în translatoarele limbajelor de programare.

Întrebări şi exerciţii
Cum se defi nește un arbore binar? Explicaţi termenii: rădăcină, subarborele stîng, subar-

borele drept, descendent, nivel, nod terminal, nod neterminal, înălţimea arborelui binar.
Formulaţi algoritmii iterativi destinaţi creării și afi șării arborilor binari.
Cum se construiește un arbore binar cu ajutorul algoritmului recursiv?
Elaboraţi un program care construiește arborele genealogic propriu pe parcursul a trei 

sau patru generaţii. Nodul-rădăcină conţine numele, prenumele și anul nașterii, iar no-
durile descendente conţin datele respective despre părinţi.

Cum trebuie modifi cată procedura Afi sArb din programul P130 pentru ca arborele 
binar să fi e afi șat în ordinea: subarborele drept, nodul-rădăcină, subarborele stîng?

Scrieţi o funcţie recursivă care returnează numărul nodurilor unui arbore binar. Transcrieţi 
această funcţie într-o formă nerecursivă.

Organizarea unui turneu „prin eliminare” este redată cu ajutorul unui arbore binar. 
Nodurile arborelui în studiu conţin următoarea informaţie:

 – numele (string);
 – prenumele (string);
 – data nașterii (ziua, luna, anul);
 – cetăţenia (string).
 Fiecărui jucător îi corespunde un nod terminal, iar fi ecărui meci – un nod neterminal. În 

fi ecare nod neterminal se înscriu datele despre cîştigătorul meciului la care au participat 
cei doi jucători din nodurile descendente. Evident, rădăcina arborelui va conţine datele 
despre cîştigătorul turneului.

 Scrieţi un program care creează în memoria calculatorului şi afi şează pe ecran arborele 
unui turneu prin eliminare.

 Indicaţie: Se porneşte de la o listă a jucătorilor. Cîştigătorii meciurilor din prima etapă se 
includ într-o altă listă. În continuare se formează lista cîştigătorilor meciurilor din etapa 
a doua ş.a.m.d.

Cum trebuie modifi cată funcţia Arb din programul P130 pentru ca arborele binar să se 
construiască în ordinea: A, C, G, J, I, F, B, E, H, D?

Funcţia Arb din programul P130 construiește arborii binari în ordinea: nodul-rădăcină, 
subarborele stîng, subarborele drept. Scrieţi o procedură nerecursivă care construiește 
arborii binari în aceeași ordine.

 Indicaţie: Se utilizează o stivă elementele căreia sînt noduri. Iniţial stiva va conţine numai 
nodul-rădăcină. Pentru fi ecare nod din vîrful stivei se va construi subarborele stîng, iar 
apoi – subarborele drept. Nodurile nou-create se introduc în stivă. După construirea 
subarborelui drept, nodul respectiv este scos din stivă.

2.8. Parcurgerea arborilor binari
Operaţiile care se pot efectua asupra arborilor binari se împart în două mari ca-

tegorii:
– operaţii care modifi că structura arborelui (inserarea sau eliminarea unui nod);
– operaţii care păstrează intactă structura arborelui (căutarea unei informaţii, ti-

părirea informaţiilor asociate unui nod etc.).
Una din problemele care apar în mod frecvent la efectuarea acestor operaţii este 

necesitatea de a parcurge sau a traversa arborele binar.
Prin parcurgerea unui arbore se înţelege examinarea în mod sistematic a noduri-

lor sale astfel încît informaţia din fi ecare nod să fi e prelucrată o singură dată. Există 
trei modalităţi de parcurgere a arborilor binari: nodurile pot fi  vizitate în preordine, 
inordine şi postordine. Aceste trei metode sînt defi nite recursiv: dacă arborele este 
vid, atunci el este parcurs fără a se face nimic; astfel parcurgerea se face în trei etape.

Parcurgerea în preordine sau traversarea RSD:
1) se vizitează rădăcina;
2) se traversează subarborele stîng;
3) se traversează subarborele drept.
Parcurgerea în inordine sau traversarea SRD:
1) se traversează subarborele stîng;
2) se vizitează rădăcina;
3) se traversează subarborele drept.

Parcurgerea în postordine sau traversarea SDR:
1) se traversează subarborele stîng;
2) se traversează subarborele drept;
3) se vizitează rădăcina.
Notaţiile RSD, SRD şi SDR reprezintă ordinea în care vor fi  vizitate rădăcina (R), 

subarborele stîng (S) şi subarborele drept (D). Metodele de parcurgere a arborilor 
binari sînt ilustrate în fi gura 2.12.

preordine (RSD) inordine (SRD) postordine (SDR)

R

S D S D S D

R R

Fig. 2.12. Metodele de parcurgere a arborilor binari

Pentru arborele din fi gura 2.11 parcurgerea în preordine furnizează nodurile în 
ordinea:

A, B, D, E, H, C, F, G, I, J;

parcurgerea în inordine furnizează nodurile în ordinea:

D, B, E, H, A, F, C, I, G, J;

iar parcurgerea în postordine conduce la:

D, H, E, B, F, I, J, G, C, A.

Prezentăm mai jos un program PASCAL de parcurgere a unui arbore binar după 
toate cele trei metode.

Program P131;
 {Parcurgerea arborelui binar }
 type Arbore=^Nod;
      Nod=record
               Info : string;
             Stg, Dr : Arbore
          end;
var  T : Arbore;    {rădăcina }

 function Arb : Arbore;
  {crearea arborelui binar }
 var R : Arbore;
     s : string;
 begin
  readln(s);
  if s=’’ then Arb:=nil
     else begin
           new(R);
           R^.Info:=s;
           write(’Daţi descendentul stîng’);
           writeln(’ al nodului ’, s, ’:’);
           R^.Stg:=Arb;
           write(’Daţi descendentul drept’);
           writeln(’ al nodului ’, s, ’:’);
           R^.Dr:=Arb;
           Arb:=R;
          end;
 end; {Arb }
 
 procedure Afi sArb(T : Arbore; nivel : integer);
  {afi şarea arborelui binar }
 var i : integer;
 begin
  if T<>nil then
     begin
      Afi sArb(T^.Stg, nivel+1);
      for i:=1 to nivel do write(’    ’);
      writeln(T^.Info);
     Afi sArb(T^.Dr, nivel+1);
     end;
 end; {Afi sareArb }
 
 procedure Preordine(T : Arbore);
  {traversare RSD }
 begin
  if T<>nil then begin
                   writeln(T^.Info);
                   Preordine(T^.Stg);
                   Preordine(T^.Dr)
                 end;
 end; {Preordine }

procedure Inordine(T : Arbore);
  {traversare SRD }
 begin
  if T<>nil then begin

                   Inordine(T^.Stg);
                   writeln(T^.Info);
                   Inordine(T^.Dr)
                 end;
 end; {Preordine }
 
 procedure Postordine(T : Arbore);
  {traversare SDR }
 begin
  if T<>nil then begin
                   Postordine(T^.Stg);
                   Postordine(T^.Dr);
                   writeln(T^.Info)
                 end;
 end; { Postordine }
 
 begin
  writeln(’Daţi rădăcina:’);
  T:=Arb;
  Afi sArb(T, 0);
  readln;
  writeln(’Parcurgere în preordine:’);
  Preordine(T);
  readln;
 writeln(’Parcurgere în inordine:’);
  Inordine(T);
  readln;
  writeln(’Parcurgere în postordine:’);
  Postordine(T);
  readln;
 end.

Menţionăm că funcţia Arb creează nodurile, parcurgînd arborele binar în curs de 
construcţie în preordine. Procedura Afi sArb afi şează nodurile, parcurgînd arborele 
binar în inordine.

Întrebări şi exerciţii
Ce operaţii pot fi  efectuate asupra arborilor binari?
Explicaţi metodele de parcurgere a arborilor binari. Daţi exemple.
Scrieţi listele de noduri obţinute în urma celor trei metode de parcurgere a arborelui 

binar din fi gura 2.13.
Transcrieţi procedurile Preordine, Inordine și Postordine din programul P131 

în formă nerecursivă.
Scrieţi un subprogram care returnează înălţimea arborelui binar.

Fig. 2.13. Arborele binar

Elaboraţi un program care afi șează pe ecran toate nodurile afl ate pe un nivel dat într-un 
arbore binar.

Elaboraţi o procedură recursivă care parcurge un arbore binar în ordinea:
 a) RDS (rădăcina – subarborele drept – subarborele stîng);
 b) DRS (subarborele drept – rădăcina – subarborele stîng);
 c) DSR (subarborele drept – subarborele stîng – rădăcina).
 Transcrieţi procedura elaborată într-o formă nerecursivă.
Scrieţi un subprogram care afi șează la ecran nivelul fi ecărui nod dintr-un arbore binar.
Se dă un arbore binar în care nodurile terminale reprezintă numere întregi, iar cele ne-

terminale – operaţiile binare +, -, *, mod, div. Arborele în studiu poate fi  considerat ca o 
reprezentare a unei expresii aritmetice. Valoarea acestei expresii se calculează efectuînd 
operaţia din nodul-rădăcină asupra valorilor subexpresiilor reprezentate de subarborele 
stîng și subarborele drept.

 Scrieţi o funcţie care returnează valoarea expresiilor aritmetice reprezentate prin arbori 
binari.

Se consideră expresiile aritmetice formate din operanzi și operatorii binari +, -, *, /. Operanzii 
sînt variabile numele cărora este format dintr-o singură literă și constante alcătuite dintr-o 
cifră. Fiecărei expresii aritmetice i se poate asocia un arbore binar după cum urmează:

 a) expresiei aritmetice formate dintr-un singur operand i se asociază un arbore binar 
format doar din nodul ce conţine operandul respectiv;

 b) expresiei aritmetice de forma E1°E2, unde E1 și E2 sînt expresii aritmetice, i se asociază 
un arbore binar care are în nodul-rădăcină operatorul °, ca subarbore stîng arborele aso-
ciat expresiei E1, iar ca subarbore drept arborele asociat expresiei E2.

 Valoarea expresiei se calculează efectuînd operaţia din nodul-rădăcină asupra valorilor 
subexpresiilor reprezentate de subarborele stîng și subarborele drept.

 Scrieţi un program care:
 a) construiește arbori binari asociaţi expresiilor aritmetice citite de la tastatură;
 b) evaluează expresiile aritmetice reprezentate prin arborii binari.
 Indicaţie. Algoritmul va urma defi niţia recursivă a arborelui în studiu. Ca operator curent 

„°” se poate desemna orice operator +, - din expresia supusă prelucrării. Operatorii *, / 
pot fi  desemnaţi ca operatori curenţi numai cînd expresia supusă prelucrării nu conţine 
operatorii +, -.

2.9. Arbori de ordinul m
Se consideră variabile dinamice de tipul record care au în cîmpul legăturilor indi-

catori de adresă. Ca şi în cazul arborilor binari vom numi astfel de variabile noduri.
Arborele de ordinul m se defi neşte recursiv după cum urmează:
a) un nod este un arbore de ordinul m;
b) un nod ce conţine cel mult m legături către alţi arbori de ordinul m este un ar-

bore de ordinul m.
Se consideră că în arbore există cel puţin un nod care subordonează exact m sub-

arbori. Prin convenţie, arborele vid nu conţine niciun nod.
Arborii de ordinul 2 se numesc arbori binari şi au fost studiaţi în paragrafele preceden-

te. Arborii de ordinul 3, 4, 5 ş.a.m.d. se numesc arbori multicăi (în engleză multiway tree).
Pentru exemplifi care, în fi gura 2.14 este prezentat un arbore de ordinul 4. Evident, 

pentru arborii multicăi termenii rădăcină, subarbore, nivel, părinte, descendent, nod 
terminal, nod neterminal, înălţime au aceeaşi semnifi caţie ca şi pentru arborii binari. 
Terminologia utilizată în structurile de date în studiu include chiar cuvinte ca fi u, 
tată, fraţi, unchi, veri, străbunic etc. cu înţeles similar celui din vorbirea curentă pentru 
noduri afl ate pe diverse niveluri. Într-un limbaj simplist, structurile de date în studiu 
exprimă relaţii de „ramifi care” între noduri, asemănătoare confi guraţiei arborilor 
din natură, cu deosebirea că în informatică arborii „cresc” în jos.

DB C

E F G H

J K

I

Anivel 0

nivel 1

nivel 2

nivel 3

Fig. 2.14. Arborele de ordinul 4

Datele necesare pentru crearea şi prelucrarea unui arbore binar de ordinul m pot 
fi  defi nite prin declaraţii de forma:

type Arbore = ^Nod;
     Nod = record;
   Info : string;
   Dsc : array [1..m] of Arbore
      end;
var T : Arbore;

Adresele descendenţilor unui nod se memorează în componentele Dsc[1], 
Dsc[2], ..., Dsc[m] ale tabloului Dsc. Adresa rădăcinii se reţine în variabila de 
tip referinţă T.

Cele mai uzuale metode de parcurgere a arborilor de ordinul m sînt parcurgerea 
în lăţime şi parcurgerea în adîncime.

Parcurgerea în lăţime presupune vizitarea nodurilor în ordinea apariţiei lor pe 
niveluri. De exemplu, pentru arborele din fi gura 2.14 nodurile vor fi  vizitate în ordi-
nea: A, B, C, D, E, F, G, H, I, J, K.

În mod obişnuit, parcurgerea în lăţime se realizează cu ajutorul unui algoritm 
iterativ care utilizează o structură auxiliară de date, şi anume o coadă formată din 
adresele nodurilor care vor fi  vizitate.

Parcurgerea în adîncime se defi neşte recursiv: dacă arborele este vid, el este par-
curs fără a se face nimic; altfel se vizitează întîi rădăcina, apoi subarborii de la stînga 
la dreapta. Pentru arborele din fi gura 2.14 parcurgerea în adîncime furnizează nodu-
rile în ordinea: A, B, C, E, F, J, K, G, H, D, I. Parcurgerea în adîncime se realizează 
foarte simplu cu ajutorul unui algoritm recursiv.

Exemplu:

Program P133;
 {Arbori de ordinul m }
 const m=4;
 type Arbore=^Nod;
      Nod=record
                Info : string;
                Dsc : array [1..m] of Arbore
          end;
 
      AdresaCelula=^Celula;
      Celula=record
                   Info : Arbore;
                    Urm : AdresaCelula
             end;
 
 var  T : Arbore;    {rădăcina }
    Prim,            {primul element din coadă }
  Ultim : AdresaCelula; {ultimul element din coadă }

 procedure IntroduInCoada(Q : Arbore);
 var R : AdresaCelula;
 begin
  new(R);
  R^.Info:=Q;
  R^.Urm:=nil;
  if Prim=nil then begin Prim:=R; Ultim:=R end
           else begin Ultim^.Urm:=R; Ultim:=R end;
 end; {IntroduInCoadă }

 procedure ExtrageDinCoada(var Q : Arbore);
 var R : AdresaCelula;
 begin
  if Prim=nil then writeln(’Coada este vidă’)
           else begin                  R:=Prim;
                  Q:=R^.Info;
                  Prim:=Prim^.Urm;
                  dispose(R);
                end;
 end; {ExtrageDinCoadă }

procedure CreareArbore(var T : Arbore);
 var R, Q : Arbore;
        s : string;
        i : integer;
 begin
 T:=nil;                {iniţial arborele este vid }
  Prim:=nil; Ultim:=nil; {iniţial coada este vidă }
  writeln(’Daţi rădăcina: ’); readln(s);
  if s<>’’ then
     begin
       new(R); {crearea rădăcinii }
       R^.Info:=s;
       T:=R;   {iniţializarea adresei rădăcinii }
       IntroduInCoada(T);
     end;
  while Prim<>nil do  {cît coada nu e vidă }
     begin
       ExtrageDinCoada(R);
       for i:=1 to m do R^.Dsc [i]:=nil;
       writeln(’Daţi descendenţii nodului’,R^.Info);
       i:=1; readln(s);
       while (i<=m) and (s<>’’) do
         begin
           new(Q); R^.Dsc [i]:=Q; Q^.Info:=s;
           IntroduInCoada(Q);
           i:=i+1; readln(s);
         end;
  end;
 end; {CreareArbore }

 procedure Afi sareArbore(T : Arbore);
 var R : Arbore;
     i : integer;
begin
  if T=nil then writeln(’Arbore vid’)

       begin
         writeln(’Arborele este format din:’);
         Prim:=nil; Ultim:=nil;
         IntroduInCoada(T);
         while Prim<>nil do
            begin
              ExtrageDinCoada(R);
              writeln(’Nodul ’, R^.Info);
              write(’  Descendenţi: ’);
              for i:=1 to m do
                if R^.Dsc [i]<>nil then
                   begin
                      write(R^.Dsc [i]^.Info, ’ ’);
                      IntroduInCoada(R^.Dsc [i]);
                   end; {then }
              writeln;
            end; {while }
       end; {else }
  readln;
 end; {Afi sareArbore }
 
 procedure InLatime(T : Arbore);
 var R : Arbore;
     i : integer;
 begin
  if T<>nil then
     begin
       Prim:=nil; Ultim:=nil;
       IntroduInCoada(T);
       while Prim<>nil do
         begin
           ExtrageDinCoada(R);
           writeln(R^.Info);
           for i:=1 to m do
            if R^.Dsc [i]<>nil then IntroduInCoada(R^.Dsc [i]);
         end; {while }
       end; {then }
 end; {InLatime }

 procedure InAdincime(T : Arbore);
 var i : integer;
 begin
  if T<>nil then
    begin
       writeln(T^.Info);
       for i:=1 to m do InAdincime(T^.Dsc [i]);
     end;
 end; {InAdincime }

 begin
  CreareArbore(T);
  Afi sareArbore(T);
  writeln(’Parcurgerea arborelui în lăţime:’);
  InLatime(T);
  readln;
  writeln(’Parcurgerea arborelui în adîncime:’);
  InAdincime(T);
  readln;
 end.

Informaţia utilă asociată fi ecărui nod se citeşte de la tastatură. Absenţa des-
cendenţilor se semnalează prin apăsarea tastei <ENTER>. Menţionăm că proce-
dura CreareArbore creează nodurile parcurgînd în lăţime arborele în curs de 
construcţie. Evident, procedura AfisareArbore vizitează nodurile în ordinea 
creării.

Operaţiile frecvent efectuate asupra arborilor multicăi sînt: inserarea sau elimi-
narea unui nod, căutarea unei informaţii, prelucrarea informaţiilor utile asociate no-
durilor ş.a. De obicei, arborii multicăi sînt utilizaţi în cazul aplicaţiilor care necesită 
prelucrarea unor mari cantităţi de date organizate ierarhic pe suporturile externe 
de informaţie. De exemplu, amintim modul de organizare a discurilor magnetice şi 
optice în sistemele de operare MS-DOS, UNIX etc. Arborii în studiu sînt de asemenea 
utilizaţi în aplicaţiile grafi ce pentru reprezentarea relaţiilor dinamice dintre compo-
nentele imaginilor procesate.

Întrebări şi exerciţii
Daţi exemple de arbori de ordinul 3, 5, 6.
Cum se defi nește un arbore de ordinul m? Ce operaţii pot fi  efectuate asupra arborilor în 

studiu?
Explicaţi metodele de parcurgere a arborilor multicăi. Daţi exemple.
Scrieţi un program recursiv care construiește în memoria calculatorului un arbore mul-

ticăi. Informaţia utilă asociată nodurilor se citește de la tastatură.
Scrieţi o funcţie care returnează:
 a) numărul nodurilor unui arbore multicăi;
 b) nivelul unui anumit nod din arbore;
 c) înălţimea arborelui.
Transcrieţi procedura InAdincime din programul P133 într-o formă nerecursivă.
Cum trebuie modifi cată procedura InLatime din programul P133 ca nodurile arbore-

lui din fi gura 6.18 să fi e vizitate în ordinea: A, D, C, B, I, H, G, F, E, K, J?
Cum trebuie modifi cată procedura InAdincime din programul P133 pentru ca nodu-

rile arborelui din fi gura 2.14 să fi e vizitate în ordinea: A, D, I, C, H, G, F, K, J, E, B?
 Se dă un arbore multicăi, informaţiile din noduri fi ind șiruri de caractere. Să se afi șeze pe 

ecran toate șirurile de caractere de lungime pară.

Organizarea datelor de pe discurile magnetice este redată cu ajutorul unui arbore mul-
ticăi. Nodurile terminale reprezintă fi șierele, iar nodurile neterminale – directoarele. 
Informaţia utilă asociată fi ecărui nod include:

 – numele fi șierului sau directorului (string[8]);
 – extensia (string[3]);
 – data și ora ultimei actualizări (respectiv ziua, luna, anul și ore, minute, secunde);
 – lungimea (integer);
 – atributele (’A’, ’H’, ’R’, ’S’).
 Elaboraţi un program care simulează operaţiile de căutare, creare și ștergere a fi șierelor 

și directoarelor.
 În unele cazuri ordinul m al arborelui multicăi nu este cunoscut în momentul scrierii 

programului, fapt ce nu permite utilizarea structurilor de date de tipul array[1..m] 
of Arbore. Pentru a depăși acest inconvenient, tabloul respectiv poate fi  înlocuit cu o 
listă uni- sau bidirecţională.

 Elaboraţi subprogramele necesare pentru crearea și prelucrarea arborilor multicăi de 
ordin arbitrar.

2.10. Tipul de date  pointer
Acest paragraf se referă în întregime la implementarea Turbo PASCAL.
Mulţimea de valori ale tipului predefi nit de date pointer (indicator) constă din 

adrese şi valoarea specială nil. Însă, spre deosebire de tipurile de date referinţă 
adresele cărora identifi că numai variabilele dinamice ce aparţin tipului de bază, va-
lorile de tip pointer pot identifi ca variabile dinamice de orice tip. Evident, valoarea 
nil nu identifi că nicio variabilă dinamică. Prin convenţie, tipul de date pointer 
este compatibil cu orice tip de date referinţă.

Operaţiile care se pot face cu valori de tipul de date pointer sînt = şi <>. Valorile 
de acest tip nu pot fi  citite de la tastatură şi afi şate pe ecran.

O variabilă de tip pointer se introduce printr-o declaraţie de forma:

var p : pointer;

Întrucît astfel de declaraţii nu conţin informaţii despre tipul de bază, tipul varia-
bilei dinamice p^ este necunoscut. Prin urmare, variabilele de tip pointer nu pot fi  
dereperate, iar scrierea caracterului ^ după astfel de variabile constituie o eroare.

Programul ce urmează ilustrează utilizarea variabilelor de tip pointer pentru 
memorarea temporară a valorilor variabilelor de tip referinţă.

Program P134;
  { Tipul de date pointer }
 var p : pointer;
  i, j : ^integer;
  x, y : ^real;
  r, s : ^string;
begin
  {p va identifi ca o variabilă dinamică de tipul integer }
  new(i); i^:=1;

  p:=i;
  new(i); i^:=2;
  j:=p;
  writeln(’j^=’, j^); { se afi şează 1 }
  {p va identifi ca o variabilă dinamică de tipul real }
  new(x); x^:=1;
  p:=x;
  new(x); x^:=2;
  y:=p;  writeln(’y^=’, y^); {se afi şează 1.0000000000E+00 }
  {p va identifi ca o variabilă dinamică de tipul string }
  new(r); r^:=’AAA’;
  p:=r;
  new(r); r^:=’BBB’;
  s:=p;
  writeln(’s^=’, s^); { se afi şează AAA }
  readln;
 end.

Domeniul principal de utilizare a variabilelor de tip pointer este gestionarea 
memoriei interne a calculatorului. În Turbo PASCAL alocarea variabilelor dinamice 
se execută într-o zonă specială a memoriei interne numită heap (grămadă). Adresa 
de început a heap-ului, numită adresa de bază, este depusă în variabila predefi nită de 
tip pointer HeapOrg. Variabila de tip pointer HeapPtr conţine adresa primei 
locaţii libere, numită vîrful heap-ului (fi g. 2.15).

HeapPtr

Zonă
liberă

Zonă ocupată de
variabilele 
dinamiceHeapOrg

Fig. 2.15. Structura heap-ului

Variabilele dinamice sînt create şi depuse în heap de procedura new. Ori de cîte ori 
în vîrful heap-ului se creează o variabilă dinamică conţinutul variabilei HeapPtr este 
actualizat: valoarea curentă este incrementată cu dimensiunea spaţiului de memorie 
necesar variabilei dinamice.

Memoria ocupată în heap de o variabilă dinamică se eliberează printr-un apel al 
procedurii dispose. Dimensiunea spaţiului ce se eliberează depinde de tipul vari-
abilei dinamice.

Ordinea de apelare a procedurii dispose nu coincide în general cu ordinea cre-
ării variabilelor dinamice de către procedura new. În consecinţă, în heap pot apărea 
goluri. Golurile apărute pot fi  refolosite de procedura new, dacă variabila dinamică 
în curs de creare „încape” în spaţiul respectiv.

Eliberarea memoriei ocupate de o structură dinamică de date poate fi  efectuată 
apelînd procedura dispose pentru fi ecare componentă. Întrucît în program sînt cu-
noscute numai adresele componentelor privilegiate, de regulă baza şi vîrful listei, ră-
dăcina arborelui etc., căutarea celorlalte componente cade în sarcina programatoru-
lui. Mai mult decît atît, ordinea de apelare a procedurii dispose trebuie să asigure 
păstrarea legăturilor către componentele care încă nu au fost distruse. În caz contrar, 
componentele respective nu mai sînt referite de niciun indicator de adresă şi devin 
inaccesibile. Prin urmare, utilizarea procedurii dispose pentru eliberarea memoriei 
ocupate de structuri complexe de date este greoaie şi inefi cientă. Acest inconvenient 
poate fi  depăşit cu ajutorul procedurilor predefi nite mark şi release.

Apelul procedurii mark are forma:

mark(p)

unde p este o variabilă de tip pointer. Procedura memorează adresa vîrfului din 
HeapPtr în variabila p.

Apelul procedurii release are forma:

release(p)

Această procedură reface adresa vîrfului în starea înregistrată anterior cu proce-
dura mark: valoarea conţinută în variabila de tip pointer p este depusă în indica-
torul HeapPtr.

Zona de memorie destinată alocării variabilelor dinamice poate fi  gestionată cu 
ajutorul algoritmului ce urmează:

1) se memorează adresa vîrfului cu procedura mark;
2) se creează variabilele dinamice cu procedura new;
3) se utilizează variabilele dinamice create;
4) cînd variabilele dinamice nu mai sînt necesare, spaţiul ocupat din heap este 

eliberat cu procedura release.

Exemplu:
Se consideră următoarele declaraţii:

var i, j, k, m, n : ^integer;
    p : pointer;

Să presupunem că sînt executate instrucţiunile:

new(i); i^:=1;
new(j); j^:=2;
mark(p);
new(k); k^:=3;
new(m); m^:=4;
new(n); n^:=5;

Starea heap-ului este prezentată în fi gura 2.16a. Instrucţiunea mark(p) a memorat 
în variabila de tip pointer p valoarea actuală din HeapPtr înainte de crearea va-
riabilei dinamice k^.

HeapPtr

Zonă
liberă

a) b)

5

4

3

2

1

2

1

Zonă
liberă

HeapOrg

P
HeapPtr

HeapOrgi

j

k

m

n

i

j

k

m

n

Fig. 2.16. Starea heap-ului pînă (a) și după executarea instrucţiunii release(p) (b)

Dacă acum se execută instrucţiunea

release(p)

memoria ocupată de variabilele dinamice create după apelul procedurii mark, şi 
anume, k^, m^ şi n^, va fi  eliberată (fi g. 2.16b).

Deoarece variabila predefi nită HeapOrg reţine adresa de bază a heap-ului, tot spa-
ţiul destinat alocării variabileler dinamice poate fi  eliberat cu ajutorul instucţiunii

release(HeapOrg)

Programul ce urmează ilustrează utilizarea procedurilor mark şi release.

Program P135;
  {Gestionarea memoriei interne }
 type Lista=^Celula;
      Celula=record
                   Info : string;
                    Urm : Lista
             end;
      Stiva=Lista;
          end;
 var L : Lista;
     S : Stiva;
     T : Arbore;
     p : pointer;

function Lst : Lista;
  {crearea listei unidirecţionale }

 var R : Lista;
     s : string;
 begin
  write(’Info=’); readln(s);
  if s=’’ then Lst:=nil
     else
         begin
           new(R);
           R^.Info:=s;
           R^.Urm:=Lst;
           Lst:=R;
         end;
 end; {Lst }
 
 procedure Afi sLst(L : Lista);
  {afi şarea listei }
 begin
  if L<>nil then
     begin
       writeln(L^.Info);
       Afi sLst(L^.Urm);
     end;
 end; {Afi sLst }

 procedure Stv(var S : Stiva);
  {crearea unei stive }
 var R : Stiva;
    st : string;
 begin
  S:=nil;
  write(’Info=’); readln(st);
  while st<>’’ do
       begin
          new(R);
          R^.Info:=st;
          R^.Urm:=S;
          S:=R;
          write(’Info=’); readln(st);
        end;
 end; { Stv }

function Arb : Arbore;
  {crearea arborelui binar }
 var R : Arbore;
     s : string;

 begin
  readln(s);
  if s=’’ then Arb:=nil
     else begin
           new(R);
           R^.Info:=s;
           write(’Daţi descendentul stîng’);
           writeln(’ al nodului ’, s, ’:’);
           R^.Stg:=Arb;
           write(’Daţi descendentul drept’);
           writeln(’ al nodului ’, s, ’:’);
           R^.Dr:=Arb;
           Arb:=R;
          end;
 end; { Arb }
 
 procedure Afi sArb(T : Arbore; nivel : integer);
  { afi şarea arborelui binar }
 var i : integer;
 begin
  if T<>nil then
     begin
      Afi sArb(T^.Stg, nivel+1);
      for i:=1 to nivel do write(’    ’);
      writeln(T^.Info);
      Afi sArb(T^.Dr, nivel+1);
     end;
 end; {Afi sArb }
 begin
  writeln(’Daţi lista:’);
  L:=Lst;
  writeln(’Lista creată:’);
  Afi sLst(L);
  mark(p);    { p reţine adresa din HeapPtr }
  writeln(’Daţi rădăcina:’);
  T:=Arb;
  writeln(’Arborele creat:’);
  Afi sArb(T, 0);
  release(p);{eliberarea memoriei ocupate de arbore }
  writeln(’Daţi stiva:’);
  Stv(S);
  writeln(’Stiva creată’);
  Afi sLst(S);
  release(HeapOrg); {eliberarea memoriei ocupate de listă şi 
stivă }
  readln;
 end.

Subliniem faptul că în implementările actuale procedurile dispose şi relea-
se nu atribuie valoarea nil indicatorilor de adresă variabilele dinamice ale cărora 
au fost distruse (fi g. 2.16b). Întrucît memoria eliberată este refolosită, atribuirile 
efectuate asupra variabilelor distruse pot altera valorile variabilelor dinamice nou-
create.

Întrebări şi exerciţii
Care este mulţimea de valori ale tipului de date pointer? Ce operaţii pot fi  efectuate 

cu aceste valori?
Comentaţi următorul program:

Program P136;
  {Eroare }
 var i : ^integer; 
 j, k : integer;
 p : pointer;
 begin
  new(i); i^:=1;
  p:=i;
  new(i); i^:=2;
  j:=i^; k:=p^;
  writeln(’j+k=’, j+k);
 end.

Care este domeniul de utilizare a variabilelor de tip pointer?
Este oare heap-ul o structură de date de tip stivă? Argumentaţi răspunsul.
Lansaţi în execuţie programele ce urmează. Explicaţi rezultatele afi șate pe ecran.

Program P137;
 var i, j, k, m, n : ^integer;
                 p : pointer;
 begin
  {crearea variabilelor i^, j^, k^ }
  new(i); new(j); new(k);
  i^:=1; j^:=2; k^:=3;
  p:=j; {p reţine adresa din j }
  {distrugerea variabilei j^ şi crearea variabilei m^}
  dispose(j); new(m); m^:=4;
  j:=p; {refacerea adresei din j }
  writeln(’i^=’, i^, ’  j^=’, j^, ’  k^=’, k^);
  {distrugerea variabilei m^ şi crearea variabilei n^ }
  dispose(m); new(n); n^:=5;
  writeln(’i^=’, i^, ’  j^=’, j^, ’  k^=’, k^);
  readln;
 end.

Program P138;
 var i, j, k, m : ^integer;
 begin

  {crearea variabilelor i^, j^ }
  new(i); new(j);
  i^:=1; j^:=2;

  {eliberarea memoriei heap-ului }
  release(HeapOrg);

  {crearea variabilelor k^ şi m^ }
  new(k); new(m);

  k^:=1; m^:=2;
  writeln(’k^=’, k^, ’ m^=’, m^);
  i^:=3; j^:=4;
  writeln(’k^=’, k^, ’ m^=’, m^);
  readln;
 end.

Scrieţi o procedură care eliberează memoria ocupată de:
 a) o listă unidirecţională;
 b) un arbore binar;
 c) un arbore multicăi.
 Memoria trebuie eliberată apelînd procedura dispose pentru fi ecare componentă a 

structurii dinamice de date.
Scrieţi un program în care se creează mai întîi o coadă, iar apoi un arbore multicăi. Spaţiul 

de memorie eliberat după distrugerea cozii trebuie refolosit pentru alocarea arborelui.

Capitolul 3

METODE  DE  ELABORARE  
A  PRODUSELOR  PROGRAM

3.1. Programarea modulară
Programarea modulară urmăreşte reducerea complexităţii programelor mari prin 

descompunerea acestora în module.
Modulul este un produs program format din descrieri de date şi subprograme 

destinate prelucrării acestora. Modulele pot fi  scrise independent şi compilate se-
parat înainte de a fi  încorporate în programul în curs de elaborare. Menţionăm că 
pentru program se mai utilizează şi denumirea de modul principal.

Limbajul-standard nu prevede mijloace pentru programarea modulară. Se consi-
deră că programele PASCAL sînt entităţi monolit care trebuie compilate împreună 
cu subprogramele pe care, eventual, le conţin. Acest lucru devine incomod în cazul 
programelor mari care pot include zeci şi chiar sute de subprograme.

În versiunea Turbo PASCAL modulele sînt implementate prin unităţi de pro-
gram. Forma generală a unei unităţi de program este:

unit <Nume>;
interface
[uses <Nume> {,<Nume>};]
[<Constante>]
[<Tipuri>]
[<Variabile>]
[{<Antet funcţie>; | <Antet procedură>;}]

implementation
[uses <Nume> {,<Nume};]
[<Etichete>]
[<Constante>]
[<Tipuri>]
[<Variabile>]
[<Subprograme>]
[{function <Identifi cator>;
<Corp>; |
 procedure <Identifi cator>;
 <Corp>;}]
 [begin
 [<Instrucţiune> {; <Instrucţiune }]]
 end.

În esenţă, o unitate de program constă din trei secţiuni: de interfaţă, de imple-
mentare şi de iniţializare.

Secţiunea de interfaţă începe cu cuvîntul-cheie interface. Aici se declară con-
stantele, tipurile, variabilele şi subprogramele exportate de unitate. Aceste elemente 
pot fi  referite de orice modul care utilizează direct sau prin tranzitivitate unitatea 
respectivă. Menţionăm că în secţiunea de interfaţă apar doar antetele funcţiilor şi 
procedurilor exportate. Dacă unitatea actuală utilizează alte unităţi, numele acestora 
sînt specifi cate în clauza uses.

Secţiunea de implementare începe cu cuvîntul-cheie implementation. Această 
secţiune conţine declaraţii locale de etichete, constante, tipuri, variabile şi subpro-
grame. Elementele defi nite aici sînt „ascunse” şi nu pot fi  referite de modulele care 
utilizează unitatea actuală. După declaraţiile locale urmează corpul procedurilor şi 
funcţiilor, ale căror antete au fost defi nite în secţiunea de interfaţă. Fiecare subpro-
gram specifi cat în interfaţă trebuie să aibă un corp. După cuvîntul-cheie function 
sau procedure se scrie doar numele subprogramului. Menţionăm că nu este nece-
sară descrierea listei de parametri şi a valorii returnate.

Secţiunea de iniţializare începe, dacă există, cu cuvîntul-cheie begin. Secţiunea 
constă dintr-o secvenţă de instrucţiuni şi serveşte pentru atribuirea valorilor iniţiale 
variabilelor defi nite în secţiunea de interfaţă. Dacă un program utilizează mai multe 
unităţi, execuţia programului este precedată de execuţia secţiunilor de iniţializare în 
ordinea în care aceste unităţi apar în clauza uses din program.

Exemplu:

Unit U1;
  {Prelucrarea vectorilor }
 interface
 
  const nmax=100;
  type Vector=array [1..nmax] of real;
  var n : 1..nmax;
  function sum(V : Vector) : real;
  function min(V : Vector) : real;
  function max(V : Vector) : real;
  procedure Citire(var V : Vector);
  procedure Afi sare(V : Vector);
 
 implementation
 
  var i : 1..nmax;
      s : real;
 
  function sum;
  begin
       s:=0;
       for i:=1 to n do s:=s+V [i];
       sum:=s;
  end; {sum }

  function min;
  begin
       s:=V [1];
       for i:=2 to n do
       if s>V [i] then s:=V [i];
       min:=s;
  end; {min }
 
  function max;
  begin
       s:=V [1];
       for i:=2 to n do
       if s<V [i] then s:=V [i];
       max:=s;
  end; {max }
 
  procedure Citire;
  begin
       for i:=1 to n do readln(V [i]);
  end; {Citire }

  procedure Afi sare;
  begin
       for i:=1 to n do writeln(V [i]);
  end; {Afi sare }
 
 begin
  write(’n=’); readln(n);
 end.

Unitatea U1 exportă constanta nmax, tipul Vector, variabila n, funcţiile sum, 
min, max, procedurile Citire şi Afi sare. Valoarea iniţială a variabilei n este citită 
de la tastatură. Elementele în studiu pot fi  referite în orice program ce conţine clauza 
uses U1.

Exemplu:

Program P139;
  {Utilizarea unităţii U1 }
 uses U1;
var A : Vector;
 begin
  writeln(’Daţi un vector:’);
  Citire(A);
  writeln(’Aţi introdus:’);
  Afi sare(A);
  writeln(’sum=’, sum(A));

  writeln(’min=’, min(A));
  writeln(’max=’, max(A));
  readln;
 end.

Domeniile de vizibilitate ale declaraţiilor din unităţile de program se stabilesc 
conform regulilor ce urmează.

1. Declaraţiile din secţiunea de implementare sînt vizibile numai în unitatea ac-
tuală.

2. Declaraţiile din secţiunea de interfaţă sînt vizibile în:
– unitatea actuală;
– modulele care utilizează direct unitatea actuală;
– modulele care utilizează unitatea actuală prin tranzitivitate.
Referirea oricărui identifi cator id declarat într-o unitate v utilizată prin tranzitivi-

tate se face prin v.id.
3. Dacă unul şi acelaşi identifi cator este declarat în mai multe module, este luată 

în considerare declaraţia cea mai recentă.

Exemplu:

Program P140;
 uses U2;
 var x : integer;
 begin
  x:=4;
  writeln(’Programul P140:’);
  writeln(’n=’, U3.n);
  writeln(’m=’, m);
  writeln(’x=’, x);
  readln;
 end.

Unit U2;
interface
 uses U3;
 var m : integer;
     x : real;
 implementation
 begin
  writeln(’Unitatea U2:’);
  m:=2;
  writeln(’   m=’, m);
  x:=3.0;
  writeln(’   x=’, x);
 end.

Unit U3;
interface
 var n : integer;
implementation
begin
 writeln(’Unitatea U3:’);
 n:=1;
 writeln(’n=’, n);
end.

În programul P140 unitatea U2 este utilizată direct, iar unitatea U3 prin tranzi-
tivitate. Variabila n din modulul U3 este referită prin U3.n. Identifi catorul x apare 
în declaraţiile var x: real din unitatea U2 şi var x: integer din programul 
P140. Compilatorul ia în considerare ultima declaraţie.

Unităţile de program se clasifi că în unităţile-standard, livrate odată cu compi-
latorul Turbo PASCAL, şi unităţile scrise de utilizator. În continuare prezentăm o 
caracteristică succintă a unităţilor-standard frecvent utilizate.

System — conţine toate subprogramele predefi nite din Turbo PASCAL. Unitatea în 
studiu se încorporează automat în toate programele, fără a fi  necesară clauza uses.

Crt — permite utilizarea funcţiilor şi procedurilor referitoare la lucrul cu ecranul 
în mod text, precum şi cu tastatura şi difuzorul. Accesibilitatea subprogramelor se 
realizează prin clauza uses crt.

Graph — implementează subprogramele destinate unor prelucrări grafi ce: defi -
niri de ferestre şi pagini, defi niri de culori şi palete, desenarea arcurilor, cercurilor, 
poligoanelor şi a altor fi guri, salvarea imaginilor etc. Serviciile unităţii de program 
pot fi  accesate prin clauza uses graph.

Printer — asigură redirectarea operaţiilor de scriere în fi şierul text cu numele 
lst la imprimantă. Utilizînd unitatea în studiu, programatorul nu mai trebuie să de-
clare, să deschidă şi să închidă acest fi şier. Serviciile unităţii Printer devin accesibile 
unui program sau unei unităţi de program prin specifi carea clausei uses printer.

Destinaţia şi modul de utilizare a constantelor, tipurilor de date, variabilelor, 
funcţiilor şi procedurilor din unităţile-standard este inclusă în ghidurile de utilizare 
şi sistemele de asistenţă Turbo PASCAL’s Online Help.

Elaborînd propriile unităţi de program, orice utilizator îşi poate crea biblioteci de sub-
programe ce descriu algoritmi din diverse domenii: rezolvarea ecuaţiilor, calcule statisti-
ce, procesarea textelor, crearea şi prelucrarea structurilor dinamice de date etc. Divizarea 
unui program mare în module uşurează activitatea de elaborare a produselor program 
în echipă. În astfel de cazuri fi ecare programator scrie, testează şi documentează cîteva 
module relativ simple, ceea ce contribuie la îmbunătăţirea produsului program rezultat.

Întrebări şi exerciţii
Care sînt avantajele programării modulare? Prevede oare limbajul-standard mijloace 

pentru programarea modulară?
Care este forma-standard a unei unităţi de program? Precizaţi structura și destinaţia sec-

ţiunilor de interfaţă, de implementare și de iniţializare.

Cum se determină domeniile de vizibilitate ale declaraţiilor din unităţile de program?
Precizaţi ce va afi șa pe ecran programul ce urmează.

Program P141;
uses U4;
var s : string;
begin
  s:=’BBB’;
  writeln(’U5.k=’, U5.k);
  writeln(’U5.m=’, U5.m);
  writeln(’U5.s=’, U5.s);
  writeln(’U4.m=’, U4.m);
  writeln(’U4.s=’, U4.s);
  writeln(’m=’, m);
  writeln(’s=’, s);
  readln;
end.

Unit U4;
interface
uses U5;
var m : real;
 s : char;
implementation
begin
  m:=4.0;
  s:=’A’;
end.
Unit U5;
interface
var k, m : integer;
 s : real;
implementation
begin
  k:=1;
  m:=2;
  s:=3.0;
end.

Comentaţi programul:

Program P142;
  {Eroare }
uses U6;
begin
  writeln(’k=’, k);
  writeln(’m=’, m);
  readln;
end.

Unit U6;
interface
 var k : integer;
implementation
 var m : integer;
begin
  k:=1;
  m:=2;
end.

Completaţi unitatea de program U1 din paragraful în studiu cu un subprogram care:
 a) returnează media aritmetică a componentelor unui vector;
 b) aranjează componentele în ordine crescătoare;
 c) returnează produsul componentelor unui vector;
 d) returnează numărul componentelor pozitive;
 e) aranjează componentele în ordine descrescătoare.
Scrieţi o unitate de program care oferă descrieri de date și subprograme pentru prelu-

crarea matricelor.
Numerele întregi n, n  10254, pot fi  reprezentate în calculator prin șiruri formate din ca-

racterele ’+’, ’–’, ’0’, ’1’, ’2’, ..., ’9’. Elaboraţi o unitate de program care conţine 
funcţiile și procedurile necesare pentru efectuarea următoarelor operaţii:

 a) citirea numerelor de la tastatură;
 b) afi șarea numerelor pe ecran;
 c) +, -, *, mod, div;
 d) calcularea factorialului;
 e) citirea și scrierea numerelor în fi șiere secvenţiale.
Șirurile de caractere de lungime n, n  500, pot fi  reprezentate în programele Turbo 

PASCAL prin variabile de tipul:

type lungime = 0..500;
     SirDeCaractere = record
    n: lungime;
    s: array [1..500] of char
       end;

 Elaboraţi o unitate de program care conţine funcţiile și procedurile necesare pentru 
efectuarea următoarelor operaţii:

 a) citirea șirurilor de la tastatură;
 b) afi șarea șirurilor pe ecran;
 c) concatenarea șirurilor;
 d) compararea lexicografi că;
 e) calcularea lungimii unui șir.
Elaboraţi o unitate de program pentru prelucrarea:
 a) listelor unidirecţionale;
 b) cozilor;
 c) stivelor;

 d) arborilor binari;
 e) arborilor multicăi.
 Utilizaţi în acest scop declaraţiile de tipuri, funcţii și proceduri din capitolul 2.

 Găsiţi în sistemul de asistenţă Turbo PASCAL’ s Online Help descrierea unităţilor-standard 
instalate pe calculatorul dvs. Afi șaţi pe ecran textul fi ecărei unităţi, determinaţi destina-
ţia și modul de utilizare a funcţiilor și procedurilor respective.

3.2. Testarea şi depanarea programelor
Un program este corect dacă:
a) după lansarea în execuţie procesul de calcul se termină;
b) rezultatele obţinute reprezintă o soluţie a problemei pentru rezolvarea căreia a 

fost scris programul.
În caz contrar programul conţine erori.
Asigurarea corectitudinii unui program presupune execuţia sa pentru fi ecare 

combinaţie posibilă de valori ale datelor de intrare. În majoritatea cazurilor acest 
lucru este imposibil, deoarece domeniul de valori al datelor de intrare este, practic, 
infi nit, iar soluţiile respective sînt necunoscute.

Testarea este o etapă în elaborarea programelor ce are drept scop eliminarea ero-
rilor. Ea se realizează executînd programul cu anumite seturi de date de intrare nu-
mite date de testare sau, mai simplu, teste. În funcţie de modul de selecţie a datelor 
de testare, deosebim:

– testarea funcţională sau metoda cutiei negre;
– testarea structurală sau metoda cutiei transparente.
Amintim că termenul cutie neagră este folosit pentru un sistem, structura internă 

a căruia este necunoscută.
În cazul testării funcţionale datele de testare sînt astfel concepute, încît să se asi-

gure că fi ecare funcţie a programului este pe deplin realizată. Programul este văzut 
ca o cutie neagră, a cărei funcţionare este determinată prin introducerea unor date şi 
analiza rezultatelor obţinute. Selectarea datelor de intrare depinde, în mare măsură, 
de îndemînarea şi experienţa celui care efectuează testarea. În mod obişnuit, se selec-
tează valori tipice şi valori netipice din domeniul datelor de intrare.

Exemplu. Se consideră programul P143. Textul programului nu este deocamdată 
prezentat pentru a sublinia faptul că în metoda testării funcţionale el nu este necesar. 
Programul realizează următoarele funcţii:

– citeşte de la tastatură un şir de numere reale;
– afi şează pe ecran media aritmetică a numerelor pozitive din şir.
Evident, domeniul datelor de intrare este format din şiruri de numere reale.
Datele de testare vor include:
a) valorile netipice:
– şir vid;
– şir ce nu conţine numere pozitive;
– şir ce conţine un singur număr pozitiv;

b) valorile tipice:
– şir cu două numere pozitive;
– şir cu trei sau mai multe numere pozitive.
În testarea structurală testele sînt elaborate examinînd structura programului: 

declaraţiile de date, proceduri şi funcţii, instrucţiunile simple, instrucţiunile structu-
rate etc. Datele de testare vor asigura:

a) execuţia fi ecărei instrucţiuni simple (atribuiri, apeluri de proceduri, salturi goto);
b) execuţia fi ecărui ciclu for de zero, unu şi de mai multe ori;
c) execuţia fi ecărei instrucţiuni if, repeat, while pentru valorile true şi fal-

se ale expresiilor booleene de control;
d) execuţia fi ecărui caz din componenţa instrucţiunilor case.
Exemplu. Prezentăm textul programului care calculează media aritmetică a nume-

relor pozitive dintr-un şir:

Program P143;
  { Media numerelor pozitive dintr-un şir }
var n, k : integer;
    x, s : real;
begin
  n:=0;
  k:=0;
  s:=0;
  writeln(’Daţi un şir de numere reale:’);
 while not eof do
    begin
      readln(x);
      n:=n+1;
      if x>0 then
     begin
      k:=k+1;
      s:=s+x;
     end;
    end; { while }
 if n=0 then writeln(’şir vid’)
        else if k=0 then writeln(’Şirul nu conţine numere
                                  pozitive’)
               else writeln(’Media=’, s/k);
 readln;
end.

Testul trebuie să asigure execuţia instrucţiunilor while şi if pentru valorile 
true şi false ale expresiilor booleene not eof, x>0, n=0 şi k=0. Prin urmare, da-
tele de testare vor include:

– un şir vid (not eof=false, n=0);
– un şir nevid (not eof=true, n0);
– un şir ce conţine cel puţin un număr pozitiv (x>0, k0);
– un şir nevid ce nu conţine numere pozitive (x0, k=0).

Metoda cutiei transparente poate fi  utilizată independent sau împreună cu me-
toda cutiei negre pentru a îmbunătăţi un test deja obţinut. De exemplu, în cazul 
programului P143 şirul ce conţine cel puţin un număr pozitiv poate fi  înlocuit cu 
trei şiruri ce conţin, respectiv unul, două, trei sau mai multe numere pozitive. Aceste 
date vor asigura execuţia instrucţiunilor k:=k+1, s:=s+x şi calcularea expresiei 
s/k pentru valorile tipice şi valorile netipice ale variabilelor k şi s.

Difi cultăţile în aplicarea testării structurale sînt legate de prezenţa instrucţiunilor 
de decizie (if, case), a celor iterative (for, while, repeat) sau a celei de transfer 
(goto). Acestea determină apariţia unui număr foarte mare de combinări, în care 
instrucţiunile de atribuire şi apelurile de proceduri pot fi  executate.

Depanarea programului constă în localizarea zonelor din program care au 
condus la apariţia unei erori, identifi carea cauzelor erorii şi corectarea acesteia. 
Depanarea poate fi  făcută static (după executarea programului) şi dinamic (în tim-
pul executării).

În metoda depanării statice cauzele erorii se stabilesc analizînd rezultatele de-
rulării programului şi mesajele sistemului de operare. Pentru a facilita procesul de 
depanare, în program temporar se includ instrucţiuni care afi şează pe ecran valorile 
intermediare ale variabilelor-cheie.

În metoda depanării dinamice localizarea erorilor se face urmărind executarea 
programului la nivel de instrucţiuni. Implementările actuale ale limbajului permit 
efectuarea următoarelor operaţii de depanare dinamică:

– execuţia pas cu pas a programului;
– observarea valorilor unor expresii specifi cate;
– crearea şi eliminarea unor puncte de suspendare a executării;
– modifi carea valorilor unor variabile;
– trasarea apelurilor de funcţii şi proceduri;
– tratarea erorilor de intrare–ieşire, a erorilor de depăşire etc.
Descrierea detaliată a operaţiilor în studiu este inclusă în sistemul de asistenţă 

Turbo PASCAL’s Online Help.
Efi cienţa depanării depinde de modul în care este scris şi testat programul, calita-

tea mesajelor de eroare generate de calculator şi tipul erorii. De regulă, un test care 
semnalează prezenţa unei erori este urmat de alte texte organizate în aşa fel, încît să 
izoleze eroarea şi să furnizeze informaţii pentru corectarea ei.

S-a constatat că testarea şi depanarea ocupă mai mult de jumătate din perioada de 
timp necesară realizării unui produs program. Complexitatea acestor procese poate 
fi  redusă prin divizarea programelor mari în subprograme sau module şi aplicarea 
metodelor programării structurate.

Subliniem faptul că testarea programelor este un mijloc efi cient de a depista ero-
rile, însă nu şi un mijloc de a demonstra absenţa lor. Cu toate că testarea nu demon-
strează corectitudinea programului, ea este deocamdată singura metodă practică de 
certifi care a produselor program. În prezent se elaborează metode de verifi care baza-
te pe demonstrarea formală a corectitudinii programului, însă rezultatele cunoscute 
în această direcţie nu sînt aplicabile programelor complexe.

Întrebări şi exerciţii
Cînd un program PASCAL este corect? Cum poate fi  asigurată corectitudinea unui pro-

gram?
Cum se selectează datele de intrare în metoda testării funcţionale?
Elaboraţi un test funcţional pentru programul P124 din paragraful 2.4. Programul reali-

zează următoarele funcţii:
 – creează o listă unidirecţională;
 – afi șează lista pe ecran;
 – include un anumit element în listă;
 – exclude din listă elementul specifi cat de utilizator.
Precizaţi funcţiile realizate de programele ce urmează și elaboraţi testele funcţionale:
 a) P117 și P120 din paragraful 2.1;
 b) P122 și P123 din paragraful 2.2;
 c) P127 din paragraful 2.5 și P128 din paragraful 2.6.
Cum se selectează datele de intrare în metoda testării structurale?
Elaboraţi teste structurale pentru programele ce urmează:
 a) P117 și P120 din paragraful 2.1;
 b) P122 și P123 din paragraful 2.2;
 c) P127 din paragraful 2.5.
Care este diferenţa dintre depanarea statică și depanarea dinamică?
Găsiţi în sistemul de asistenţă Turbo PASCAL’ s Online Help descrierea operaţiilor de depa-

nare dinamică. Efectuaţi aceste operaţii pentru programele elaborate de dvs.

3.3. Elemente de programare structurată
Încă din primii ani de activitate în domeniul prelucrărilor de date s-a constatat că 

testarea, depanarea şi modifi carea programelor necesită un mare volum de muncă. 
Mai mult decît atît, programele complexe ce conţin sute şi mii de instrucţiuni devin 
greu accesibile chiar şi pentru autorii lor.

Programarea structurată reprezintă un stil, o manieră de concepere a programe-
lor potrivit unor reguli bine stabilite, bazate pe teorema de structură. Conform teo-
remei de structură, orice algoritm poate fi  reprezentat ca o combinaţie a trei structuri 
de control:

– secvenţa (succesiune de două sau mai multe atributuri şi/sau apeluri de pro-
ceduri);

– decizia (if... then... sau if... then... else...);
– ciclul cu test iniţial (while... do...).
Programarea structurată admite şi utilizarea altor structuri de control, cum sînt:
– selecţia (case... of...);
– ciclul cu test fi nal (repeat... until...);
– ciclul cu contor (for... do...).

Regulile de bază ale programării structurate sînt:
1. Structura oricărui program sau subprogram va fi  concepută ca o combinaţie a 

structurilor de control admise: secvenţa, decizia, selecţia, ciclul.
2. Structura datelor utilizate în program trebuie să corespundă specifi cului pro-

blemelor rezolvate.
3. Lungimea maximă a unei funcţii sau proceduri este de 50–100 de linii. Folosirea 

variabilelor globale nu este încurajată.
4. Identifi catorii folosiţi pentru constante, tipuri, variabile, funcţii, proceduri şi 

unităţi de program trebuie să fi e cît mai sugestivi.
5. Claritatea textului trebuie asigurată prin inserarea comentariilor şi alinierea 

textului în conformitate cu structura logică şi sintactică a instrucţiunilor.
6. Operaţiile de intrare–ieşire vor fi  localizate în subprograme separate. Co-

rectitudinea datelor de intrare se verifi că imediat după citirea lor.
7. Încuibarea insturcţiunilor if mai mult de trei ori trebuie evitată prin folosirea 

istrucţiunilor case.
Programele obţinute conform regulilor în studiu sînt testabile, clare, ordonate, 

fără salturi şi reveniri. Menţionăm că, conform teoremei de structură, orice program 
poate fi  scris fără a utiliza instrucţiunea goto. Totuşi unii autori admit utilizarea 
acestei instrucţiuni cu condiţia ca ea să fi e folosită la minimum, iar salturile să fi e 
efectuate numai în jos.

Întrebări şi exerciţii
Care este justifi carea teoretică a programării structurate?
Precizaţi structurile de control necesare și sufi ciente pentru reprezentarea oricărui algoritm.
Formulaţi regulile de bază ale programării structurate.
Care sînt avantajele programării structurate?
Corespund oare programele P124, P130 și P135 din capitolul 2 regulilor de bază ale 

programării structurate?
Programul ce urmează afi șează pe ecran toate reprezentările posibile ale numărului na-

tural n ca sumă de numere naturale consecutive.

Program P144;
var a,i,l,s,n : integer;
b : boolean;
begin
write(’n=’); readln(n);
b:=true;
for i:=1 to ((n+1) div 2) do
begin
a:=i;
s:=0;
while (s<n) do
begin
s:=s+a;
a:=a+1;
end;

if s=n then
begin
b:=false;
write(n, ’=’, i);
for l:=i+1 to (a-1) do write(’+’,l);
writeln;
end;
end;
if b then writeln(’Reprezentări nu există’);
readln;
end.

 De exemplu, pentru n = 15 obţinem:
 15=1+2+3+4+5;

 15=4+5+6;

 15=7+8.

 Reprezentările în studiu se calculează prin metoda trierii, examinîndu-se șirurile:
 1, 2, 3, ..., k;
 2, 3, ..., k;
 3, ..., k
 ș.a.m.d., unde k = (n + 1) div 2. Termenii 1, 2, 3, ..., k se calculează cu ajutorul relaţiei recu-

rente a = a + 1. Aliniaţi textul programului în conformitate cu structura logică și sintaxa 
fi ecărei instrucţiuni.

Capitolul 4

ANALIZA  ALGORITMILOR

4.1. Complexitatea algoritmilor
Valoarea practică a programelor PASCAL depinde în mod decisiv de complexi-

tatea algoritmilor ce stau la baza lor. Amintim că algoritmul reprezintă o succesiune 
fi nită de operaţii (instrucţiuni, comenzi) cunoscute, care, fi ind executate într-o ordi-
ne bine stabilită, furnizează soluţia unei probleme.

E cunoscut faptul că unul şi acelaşi algoritm poate fi  descris prin diverse metode: 
scheme logice, formule matematice, texte scrise într-un limbaj de comunicare între oa-
meni, cu ajutorul limbajelor de programare. Evident, şi în acest manual algoritmii pe 
care îi vom studia vor fi  descrişi cu ajutorul mijloacelor oferite de limbajul PASCAL: 
instrucţiuni, funcţii, proceduri şi programe ce pot fi  derulate pe calculator.

Complexitatea algoritmului se caracterizează prin necesarul de memorie şi durata 
de execuţie. Metodele de estimare a acestor indicatori se studiază într-un comparti-
ment special al informaticii, denumit analiza algoritmilor. În cadrul acestui compar-
timent se utilizează următoarele notaţii:

n − un număr natural ce caracterizează mărimea datelor de intrare ale unui al-
goritm. În majoritatea cazurilor n reprezintă numărul de elemente ale unei mulţimi, 
gradul unei ecuaţii, numărul de componente ale unui tablou etc.;

V(n) − volumul de memorie internă necesară pentru păstrarea datelor cu care 
operează algoritmul;

T(n) − timpul necesar executării algoritmului. De obicei, în acest timp nu se inclu-
de durata operaţiilor de introducere a datelor iniţiale şi de extragere a rezultatelor.

Evident, volumul de memorie V(n) şi timpul de execuţie T(n) depind, în primul 
rînd, de caracteristica n a datelor de intrare, fapt accentuat şi prin folosirea notaţiilor 
ce reprezintă funcţii de argumentul n.

Aplicarea practică a unui algoritm este posibilă numai atunci cînd necesarul de 
memorie şi timpul cerut nu încalcă restricţiile impuse de mediul de programare şi 
capacitatea de prelucrare a calculatorului utilizat.

De exemplu, presupunem că pentru rezolvarea unei probleme există doi algo-
ritmi diferiţi, notaţi prin A1 şi A2. Necesarul de memorie şi timpul cerut de algoritmul 
A1 este:

V1(n) = 100n2 + 4;
T1(n) = n3  10-3  ,

iar de algoritmul A2:
V2(n) = 10n + 12;
T2(n) = 2n  10-6  .

În aceste formule volumul de memorie se calculează în octeţi, iar timpul – în se-
cunde.

Necesarul de memorie şi timpul cerut de algoritmii A1, A2 pentru diferite valori 
ale lui n este prezentat în tabelul 4.1.

Tabelul 4.1.
Complexitatea algoritmilor A1 şi A2

n 10 20 30 40 50

V1(n) 9,77 Kocteţi 39,06 Kocteţi 87,89 Kocteţi 156,25 Kocteţi 244,14 Kocteţi

V2(n) 112 octeţi 212 octeţi 312 octeţi 412 octeţi 512 octeţi

T1(n) 1 secundă 8 secunde 9 secunde 16 secunde 25 secunde

T2(n) 0,001 secunde 1,05 secunde 18 secunde 13 zile 36 ani

Din tabelul 4.1 se observă că algoritmul A2 devine practic inutilizabil pentru da-
tele de intrare caracteristica cărora n > 30. Pentru astfel de date timpul de execuţie 
a algoritmului A1 este mult mai mic, însă necesarul de memorie V1(n) poate depăşi 
limita impusă de mediul de programare (64 Kocteţi pentru variabilele statice din pro-
gramele Turbo PASCAL 7.0).

Determinarea necesarului de memorie V(n) şi a timpului de execuţie T(n) pre-
zintă un interes deosebit la etapa de elaborare a algoritmilor şi a programelor res-
pective. Evident, anume pe parcursul acestei etape pot fi  eliminaţi din start acei 
algoritmi care necesită memorii prea mari sau care au un timp de execuţie inac-
ceptabil.

Menţionăm că existenţa unor calculatoare cu memorii din ce în ce mai performan-
te fac ca atenţia informaticienilor să fi e îndreptată în special asupra necesarului de 
timp sau, cu alte cuvinte, asupra complexităţii temporale a algoritmilor.

Întrebări şi exerciţii
Explicaţi termenul complexitatea algoritmului. Numiţi indicatorii ce caracterizează com-

plexitatea algoritmilor.
Cum credeţi, care factori infl uenţează complexitatea unui algoritm?
De ce depinde necesarul de memorie și timpul cerut de un algoritm? Cînd este posibilă 

aplicarea practică a unui algoritm?
Algoritmii A1 și A2 (vezi tabelul 4.1) vor derula în mediul de programare Turbo PASCAL 

7.0. Cum credeţi, care algoritm trebuie utilizat în cazul datelor de intrare cu caracteristi-
ca: a) n = 10; b) n = 20; c) n = 30? Pentru care valori ale lui n algoritmul A1 poate fi  utilizat 
în mediul de programare Turbo PASCAL 7.0?

 Complexitatea unui algoritm, notat prin A3, se caracterizează prin
V3(n) = 600n3 + 18;
T3(n) = 3 n  10 -2 .

 Cum credeţi, pentru care valori ale lui n algoritmul A3 poate derula în mediul de progra-
mare Turbo PASCAL 7.0?

Determinaţi mărimea datelor de intrare a celor mai reprezentativi algoritmi elaboraţi de 
dvs. în procesul studierii limbajului de programare PASCAL.

4.2. Estimarea necesarului de memorie
Evaluarea necesarului de memorie V(n) poate fi  făcută însumînd numărul de oc-

teţi alocaţi pentru fi ecare variabilă din program. Numărul de octeţi alocat unei varia-
bile nestructurate  integer, real, boolean, char, enumerare, subdomeniu, referin-
ţă  depinde de implementarea limbajului. Numărul de octeţi alocat unei variabile 
depinde de implementarea limbajului. În mediul de programare Turbo PASCAL 7.0 
memoria se alocă conform tabelului 4.2.

Tabelul 4.2
Alocarea memoriei interne în Turbo PASCAL 7.0

Tipul variabilei Numărul de octeţi

integer 2

real 6

boolean 1

char 1

enumerare 1

subdomeniu conform tipului de bază

referinţă 4

pointer 4

În cazul tipurilor structurate de date volumul de memorie necesar unei variabile 
se calculează însumînd numărul de octeţi alocaţi pentru fi ecare componentă.

De exemplu, necesarul de memorie pentru variabilele A, B, p şi s din declaraţiile:

var A : array[1..n, 1..n] of real;
    B : array[1..n] of integer;
    p : boolean;
    s : string[10];

este
V(n) = 6n2 + 2n + 11 (octeţi).

În general, necesarul de memorie al unui program PASCAL depinde nu numai de 
tipul variabilelor utilizate, dar şi de modul de gestionare a memoriei interne a calcu-

latorului. În procesul derulării unui program PASCAL spaţiul de memorie internă 
este divizat în trei secţiuni (fi g. 4.1):

 segmentul date, destinat alocării variabilelor globale. Aceste variabile se decla-
ră în secţiunea var a programului PASCAL;

 stiva, destinată alocării parametrilor actuali, variabilelor locale, valorilor retur-
nate de  funcţii şi adreselor de revenire pe durata execuţiei subprogramelor PASCAL. 
Apelul unui subprogram implică depunerea datelor respective în stivă, iar ieşirea din 
subprogram  eliminarea lor. Accentuăm că în cazul parametrului-variabilă în stivă 
se depune numai adresa variabilei din programul apelant, iar în cazul parametrului-
valoare în stivă va fi  depusă o copie a datelor din lista parametrilor actuali.

 heap-ul, utilizat pentru alocarea variabilelor dinamice. Aceste variabile sînt cre-
ate şi, eventual, distruse cu ajutorul procedurilor new şi dispose.

Segmentul
date

Vd(n)

Vs(n)

Vh(n)

Stiva Heap-ul

– variabile 
globale

– parametri 
actuali;

– variabile 
locale;
– valori 
de funcţii;
– adrese 
de revenire

– variabile 
dinamice

Fig. 4.1. Gestionarea memoriei interne

Prin urmare, estimarea necesarului de memorie presupune evaluarea următoare-
lor caracteristici ale unui program (fi g. 4.1):

Vd(n)  volumul de memorie ocupat de variabilele globale în segmentul date;
Vs(n)  volumul de memorie ocupat de parametrii actuali şi de variabilele locale 

în stivă;
Vh(n)  volumul de memorie ocupat de variabilele dinamice în heap.
De obicei, în mediul de programare Turbo PASCAL 7.0 se cere ca Vd(n)  64 Kocteţi, 

Vs(n)  16 Kocteţi şi Vh(n)  256 Kocteţi. Dimensiunile stivei şi ale heap-ului pot fi  modifi -
cate cu ajutorul directivelor de compilare sau a comenzilor mediului de programare.

Exemplu:

Program P145;
 { Gestionarea memoriei interne }
const n = 100;
type Matrice = array[1..n, 1..n] of real;
     Vector = array[1..n] of real;

var A : Matrice;
    i : integer;
 p, q : ^Matrice;

procedure Prelucrare(var B:Matrice);
var C : Vector;
begin
  {...prelucrarea elementelor matricei B...}
end; { Prelucrare }

begin
  {...introducerea matricei A...}
 Prelucrare(A);
 new(p);
 new(q);
  {...prelucrarea variabilelor dinamice p^ si q^...}
 dispose(p);
 dispose(q);
  {...afi şarea rezultatelor...}
 writeln(’Sfîrşit’);
 readln;
end.

Variabilele globale A, i, p şi q din programul P145 vor fi  depuse în segmentul date 
(fi g. 4.2). Necesarul de memorie pentru aceste variabile:

Vd(n) = 6n2 + 2 + 2  4 = 6n2 + 10.

Execuţia instrucţiunii apel de procedură Pr(A) implică depunerea în stivă a 
adresei matricei A, a adresei de revenire în programul principal şi a variabilei locale 
C. Necesarul de memorie pentru aceste date:

Vs(n) = 6n + 8.

După ieşirea din procedură, datele respective vor fi  eliminate din stivă.
Instrucţiunile new(p) şi new(q) creează în heap variabilele dinamice p^ şi q^ de 

tipul Matrice. Necesarul de memorie pentru aceste variabile:
Vh(n) = 6n2 + 6n2 = 12n2 .

După execuţia instrucţiunilor dispose(p) şi dispose(q), variabilele dinamice 
din heap sînt distruse, iar spaţiul respectiv de memorie devine liber.

Întrebări şi exerciţii
Determinaţi cu ajutorul sistemului de asistenţă al mediului de programare cu care lu-

craţi dvs. necesarul de memorie pentru variabilele nestructurate.
Cum se evaluează volumul de memorie internă necesar pentru înmagazinarea datelor 

unui algoritm?

Segmentul
date Stiva Heap-ul

Matricea A

i
p

q

Adresa de revenire

Adresa matricei A

Vectorul C

Matricea p^

Matricea q^

q

p

Fig. 4.2. Gestionarea memoriei în programul P145

Explicaţi cum se gestionează memoria internă în cazul unui program PASCAL.
Determinaţi cu ajutorul sistemului de asistenţă dimensiunile segmentului date, ale stivei și 

ale heap-ului. Cum pot fi  modifi cate dimensiunile stivei și ale heap-ului?
Calculaţi necesarul de memorie pentru variabilele din următoarele declaraţii:

a) var A : array[1..n, 1..n] of integer;
    B : string;
    C : array [1..n, 1..n, 1..n] of boolean;

b) type Vector  = array[1..n] of real;
     Matrice = array[1..n] of Vector;
var A, B, C : Matrice;
    D : Vector;

c) type Elev = record
               Nume : string;
               Prenume : string;
               NotaMedie : real
            end; 
     ListaElevi = array[1..n] of Elev;
var A, B, C : ListaElevi;

d) type Angajat = record
                  NumePrenume : string;
                  ZileLucrate : 1..31;
                  PlataPeZi : real;
                  PlataPeLuna : real
               end;

     ListaDePlata = array[1..n] of Angajat;
var L1, L2 : ListaDePlata;

e) type Elev = record
               Nume : string;
               Prenume : string;
               NotaMedie : real
            end; 
     FisierElevi = fi le of Elev;
var FE : FisierElevi;
    E : Elev;
    str : string;
    i, n : integer;;

Evaluaţi necesarul de memorie pentru programul ce urmează. Compilaţi acest pro-
gram pentru valorile 50, 60, 70, 80 și 100 ale constantei n. Explicaţi mesajele afi șate pe 
ecran.

Program P146;
   { Dimensiunile segmentului date }
const n = 50;
type Matrice = array[1..n, 1..n] of real;
var A, B : Matrice;
begin
  {...introducerea datelor...}
  {...prelucrarea matricelor A şi B...}
 writeln(’Sfîrşit’);
 readln;
end.

Se consideră următorul program:

Program P147;
 { Dimensiunile stivei }
var n : integer;

function S(n:integer):real;
begin
 if n=0 then S:=0
        else S:=S(n-1)+n;
end; { S }

begin
 write(’n=’); readln(n);
 writeln(’s=’, S(n));
 readln;
end.

 În acest program suma
S(n) = 0 + 1 + 2 + ... + n

 este calculată cu ajutorul funcţiei recursive

 Estimaţi necesarul de memorie al programului P147. Determinaţi valoarea maximală a 
lui n pentru care programul P147 derulează fără erori.

Determinaţi necesarul de memorie al programului ce urmează. Pentru care valori ale lui 
n programul va derula fără erori?

Program P148;
 { Dimensiunile heap-ului }
type Vector = array[1..100] of real;
var p : ^Vector;
    i, n : integer;
begin
 write(’n=’); readln(n);
 for i:=1 to n do new(p);
 writeln(’Sfîrşit’);
 readln;
end.

4.3. Măsurarea timpului de execuţie
În cazul programelor deja elaborate timpul T(n) cerut de un algoritm poate fi  afl at 

prin măsurări directe. Vom folosi în acest scop unitatea de program U7:

Unit U7;
 { Masurarea timpului }
interface
function TimpulCurent : real;
implementation
uses Dos;
var ore : word;
    minute : word;
    secunde : word;
    sutimi : word;
function TimpulCurent;
begin
GetTime(ore, minute, secunde, sutimi);
TimpulCurent:=3600.0*ore+60.0*minute+
              1.0*secunde+0.01*sutimi;
end; { TimpulCurent }
end.

Unitatea de program U7 oferă programatorului funcţia TimpulCurent, care 
returnează o valoare de tip real  timpul în secunde. Indicaţiile ceasului de sis-
tem în ore, minute, secunde şi sutimi de secundă se citesc cu ajutorul proce-
durii GetTime din unitatea de program DOS a mediului de programare Turbo 
PASCAL 7.0.

Pentru exemplifi care prezentăm programul P149 în care se măsoară timpul de 
execuţie a procedurii Sortare:

Program P149;
 { Timpul de execuţie a procedurii Sortare }
uses U7;
type Vector = array[1..10000] of real;
var   A  : Vector;
    i, n : integer;
  T1, T2 : real; { timpul in secunde }

procedure Sortare(var A:Vector; n:integer);
  { Sortarea elementelor vectorului A }
var i, j : integer;
       r : real;
begin
  for i:=1 to n do
     for j:=1 to n-1 do
        if A[j]>A[j+1] then
           begin
             r:=A[j];
             A[j]:=A[j+1];
             A[j+1]:=r;
           end;
end; { Sortare }

begin
 write(’Daţi numarul de elemente n=’);
 readln(n);

 { atribuim lui A valoarea (n, n-1, ..., 3, 2, 1) }
 for i:=1 to n do A[i]:=n-i+1;

 T1:=TimpulCurent;
 Sortare(A, n);
 T2:=TimpulCurent;

 writeln(’Durata de execuţie’, (T2-T1):7:2, ’  secunde’);
 readln;
end.

Procedura Sortare ordonează elementele vectorului A prin metoda bulelor. În 
această metodă vectorul A este parcurs de n ori, la fi ecare parcurgere efectuîndu-se 
n-1 comparări ale elementelor vecine A[j] şi A[j+1]. Dacă A[j]>A[j+1], elemen-
tele vecine îşi schimbă locul.

Pentru a evita introducerea de la tastatură a unui număr mare de date, în progra-
mul P149 vectorului A i se atribuie valoarea iniţială

A = (n, n-1, n-2, ..., 3, 2, 1).

De exemplu, pentru n=4, vectorul iniţial va fi 
A=(4, 3, 2, 1).

În procesul sortării avem:
i = 1,  A = (3, 2, 1, 4);
i = 2,  A = (2, 1, 3, 4);
i = 3,  A = (1, 2, 3, 4);
i = 4,  A = (1, 2, 3, 4).

Timpul de execuţie a procedurii Sortare în cazul unui calculator Pentium cu frecvenţa 
ceasului de sistem 500 MHz este prezentat în tabelul 4.3, iar grafi cul respectiv  în fi gura 4.3.

Tabelul 4.3
Timpul de execuţie a procedurii Sortare

n 1000 2000 3000 4000 5000 6000 7000 8000 9000 10000

T(n), s 0,27 1,10 2,47 4,50 7,03 10,16 13,84 18,02 22,85 28,18

Întrebări şi exerciţii
Cum credeţi, ce legătură există între timpul necesar execuţiei unui program PASCAL, 

frecvenţa ceasului de sistem și capacitatea de prelucrare a calculatorului?
Măsuraţi timpul de execuţie a procedurii Sortare (vezi programul P149) în cazul cal-

culatorului cu care lucraţi dvs. Construiţi un grafi c similar celui din fi gura 4.3.
Reprezentaţi grafi c pe un singur desen timpul de execuţie a procedurilor ce urmează.

a) procedure N2(n : integer);
var i, j, k : integer;
          r : real;
begin
   for i:=1 to n do
     for j:=1 to n do
        for k:=1 to 300 do
           r:=1.0;
end; { N2 }

b) procedure N3(n : integer);
var i, j, k : integer;
          r : real;

Fig. 4.3. Timpul de execuţie a procedurii Sortare

begin
   for i:=1 to n do
     for j:=1 to n do
        for k:=1 to n do
           r:=1.0;
end; { N3 }

c) procedure N4(n : integer);
var i, j, k, m : integer;
             r : real;
begin
   for i:=1 to n do
     for j:=1 to n do
        for k:=1 to n do
           for m:=1 to n do
              r:=1.0;
end; { N4 }

 Pentru exemplifi care, în fi gura 4.4 sînt prezentate grafi cele respective în cazul unui cal-
culator Pentium, frecvenţa ceasului de sistem 500 MHz.

Care este precizia de măsurare a timpului cu ajutorul funcţiei TimpulCurent? 
Argumentaţi răspunsul dvs.

Fig. 4.4. Timpul de execuţie a procedurilor N2, N3 și N4

4.4. Estimarea timpului cerut de algoritm
În procesul implementării practice a oricărui algoritm apare necesitatea estimării 

timpului de execuţie T(n) pînă la testarea şi depanarea defi nitivă a programului ce-l 
realizează. Cu regret, prin metode teoretice este foarte greu de determinat o expresie 
exactă pentru T(n). Din aceste motive se caută o limită superioară a timpului cerut de 
algoritm, analizîndu-se doar cazurile cele mai defavorabile.

Presupunem, în scopuri didactice, că execuţia oricărui operator PASCAL (+, -, 
or, *, /, div, and, <, <=, not etc.) necesită cel mult  unităţi de timp. Acelaşi timp 
 este necesar pentru indexarea componentelor unui tablou [ ], pentru atribuirea := 
şi instrucţiunea goto. Valoarea concretă a mărimii  depinde de mediul de progra-
mare, capacitatea de prelucrare a calculatorului utilizat şi este de ordinul 10 -9 ... 10 -7 
secunde. În continuare vom estima timpul T(n) în forma:

T(n) = Q(n)  ,
unde Q(n) este numărul de operaţii elementare  adunarea, scăderea, înmulţirea, 
compararea etc.  necesare pentru soluţionarea unei probleme.

Admitem că într-o expresie E apar m operatori PASCAL şi k apeluri ale funcţiei 
F. Evident, numărul QE de operaţii elementare necesare pentru calcularea expresiei 
E se determină ca

QE = m + k QF ,
unde QF este numărul de operaţii elementare necesare pentru calcularea funcţiei F.

Exemple:
Expresia E Numărul de operaţii elementare QE

a) a*b+c 2

b) (a<b)or(c>d) 3

c) sin(x)+cos(y) 1 + Qsin + Qcos

d) a+M[i] 2

e) sin(x+y)+sin(x-y) 3 + 2Qsin

Numărul de operaţii elementare QI necesare pentru execuţia unei instrucţiuni I a 
limbajului PASCAL se estimează conform formulelor din tabelul 4.4.

Tabelul 4.4 
Numărul de operaţii elementare necesare pentru execuţia                                                 

unei instrucţiuni PASCAL
Nr. crt. Instrucţiunea PASCAL Numărul de operaţii elementare

1 Atribuirea  := E QE + 1

2 Apelul procedurii P QP + 1

3 Selecţie
if E then I1 else I2

QE + max{QI1
, QI2

} + 1

4 Selecţie multiplă
case E of I1; I2; ...; Ik end

QE + max{QI1
, QI2

, ..., QIk
} + k + 1

5 Ciclu cu contor
for := E1 to/downto E2 do I QE1

 + QE2
 + mQI + m + 1

6 Ciclu cu test iniţial
while E do I (m + 1)QE + mQI + 1

7 Ciclu cu test fi nal
repeat I until E mQI + mQE + 1

8 Instrucţiunea compusă
begin I1; I2; ...; Ik end

QI1
 + QI2

 + ... + QIk
 + 1

9 Instrucţiunea
with  do I QI + 1

10 Saltul goto 1

Formulele din tabelul 4.4 pot fi  deduse urmînd modul de execuţie a fi ecărei instruc-
ţiuni PASCAL. În acest tabel  reprezintă o variabilă sau un nume de funcţie, E  o 
expresie, iar I  o instrucţiune. Numărul de execuţii ale instrucţiunii I din cadrul unui 
ciclu for, while sau repeat este notat prin m. Menţionăm că ciclurile unui program 
PASCAL pot fi  organizate şi cu ajutorul instrucţiunilor if şi goto, însă o astfel de 
utilizare a acestor instrucţiuni contravine regulilor programării structurate.



106

Pentru exemplifi care vom estima numărul de operaţii elementare Q(n) necesare 
ordonării elementelor unui vector prin metoda bulelor:

procedure Sortare(var A:Vector; n:integer);
var i, j : integer;
       r : real;
{1}  begin
{2}  for i:=1 to n do
{3}     for j:=1 to n-1 do
{4}        if A[j]>A[j+1] then
{5}            begin 
{6}              r:=A[j];
{7}              A[j]:=A[j+1];
{8}              A[j+1]:=r;
               end;
end; { Sortare }

Instrucţiunile I1, I2, ... , I8 ale procedurii Sortare vor fi  referite cu ajutorul co-
mentariilor {1}, {2}, ..., {8} din partea stîngă a liniilor de program. Prin Qj vom nota 
numărul de operaţii elementare necesare pentru executarea instrucţiunii Ij :

Q6 = 2;
Q7 = 4;
Q8 = 3;
Q5 = Q6 + Q7 + Q8 + 1 = 10;
Q4 = 4 + Q5 + 1 = 15;
Q3 = 0 + 1 + (n–1)Q4 + (n–1) + 1 = 16n – 14;
Q2 = 0 + 0 + nQ3 + n + 1 = 16n2 – 13n + 1;
Q1 = Q2 +1 = 16n2 – 13n + 2.

Prin urmare, numărul de operaţii elementare
Q(n) = 16n2 – 13n + 2,

iar timpul cerut de procedura Sortare

T(n) = (16n2 – 13n + 2) .

Din exemplul studiat mai sus se observă că ordinea parcurgerii instrucţiunilor 
este impusă de structura programelor PASCAL. Evident, mai întîi se analizează in-
strucţiunile simple, iar apoi cele structurate. În cazul instrucţiunilor imbricate, mai 
întîi se analizează instrucţiunile din interior, apoi cele care le cuprind.

Expresiile analitice T(n) obţinute în urma analizei programelor PASCAL pot fi  
folosite pentru determinarea experimentală a timpului  necesar efectuării unei ope-
raţii elementare. De exemplu, pentru procedura Sortare (vezi tabelul 4.3) n = 10000 
şi T(n) = 28,18 s. Din ecuaţia 

(16n2 – 13n + 2) = 28,18
obţinem   1,8  10 -8 secunde.

Evident, această valoare este valabilă numai pentru mediul de programare Turbo 
PASCAL 7.0 şi calculatorul Pentium cu frecvenţa ceasului de sistem 500 MHz, uti-
lizate în procesul de măsurare a timpului de execuţie a procedurii Sortare. De 
exemplu, în cazul unui calculator Pentium cu frecvenţa ceasului de sistem 150 MHz 
se obţine valoarea   6,0  10 -8 secunde.

Întrebări şi exerciţii
Determinaţi cu ajutorul programului P149 valoarea  pentru mediul de programare și 

calculatorul cu care lucraţi dvs.
Determinaţi numărul de operaţii elementare QI necesare pentru execuţia următoarelor 

instrucţiuni PASCAL:

a) x:=2*a-6*(y+z);

b) p:=not(a=b)and(c>d);

c) p:=(a in R)and(b in P);

d) if a>b then x:=0 else x:=a+b;

e) case i of
   1: x:=0;
   2: x:=a+b;
   3: x:=a+b+c;
end;

f ) for i:=1 to n do A[i]:=2*A[i];

g) for i:=1 to n do A[i]:=B[i+1]-C[i-2];

h) i:=0; while i<n do begin i:=i+1 end;

i) i:=n; repeat i:=i-1 until i=0;

j) begin i:=0; s:=0; r:=0 end;

k) with A do begin x:=0; y:=0 end.

Estimaţi numărul operaţiilor elementare Q(n) din procedurile ce urmează:

a) procedure N2(n : integer);
var i, j, k : integer;
          r : real;
begin
   for i:=1 to n do
     for j:=1 to n do
        for k:=1 to 300 do
           r:=1.0;
end; { N2 }

b) procedure N3(n : integer);
var i, j, k : integer;
          r : real;
begin
   for i:=1 to n do
     for j:=1 to n do
        for k:=1 to n do
           r:=1.0;
end; { N3 }

c) procedure N4(n : integer);
var i, j, k, m : integer;
             r : real;
begin
   for i:=1 to n do
     for j:=1 to n do
        for k:=1 to n do
           for m:=1 to n do
              r:=1.0;
end; { N4 }

 În cazul procedurii Sortare pentru un calculator Pentium cu frecvenţa ceasului de sis-
tem f1 = 500 MHz s-a obţinut 1  1,8  10 -8 s. Pentru același tip de calculator, însă cu 
frecvenţa ceasului de sistem f2 = 150 MHz, s-a obţinut 2  6,0  10 -8 s. Se observă că

       
și

         .
 Cum credeţi, prin ce se explică acest fapt?
În procesul compilării, instrucţiunile limbajului PASCAL sînt translatate în una sau mai 

multe instrucţiuni din limbajul cod-mașină. Numărul concret de instrucţiuni depinde de 
mediul de programare și tipul calculatorului utilizat. Elaboraţi planul unui experiment 
care ne-ar permite să estimăm numărul de instrucţiuni cod-mașină în care este transla-
tată fi ecare instrucţiune PASCAL.

E cunoscut faptul că timpul de execuţie a instrucţiunilor din limbajul cod-mașină de-
pinde de tipul lor. De exemplu, o instrucţiune care adună două numere întregi este mai 
rapidă decît instrucţiunea care adună două numere reale. În consecinţă, valorile , de-
terminate prin măsurarea timpului de execuţie a unui program PASCAL, depind de tipul 
datelor utilizate. Verifi caţi experimental această afi rmaţie în cazul calculatorului cu care 
lucraţi dvs.

Capacitatea de prelucrare a unui calculator se măsoară în Mips  Megainstrucţiuni pe 
secundă. De exemplu, calculatoarele personale au capacitatea de prelucrare 500800 
Mips. Pentru a măsura această caracteristică, producătorii de calculatoare utilizează 
instrucţiunile limbajului cod-mașină. Evident, pentru un programator PASCAL ar pre-
zenta interes și capacitatea de prelucrare exprimată în instrucţiuni PASCAL pe secundă. 
Elaboraţi planul unui experiment care ar permite estimarea acestei caracteristici pentru 
calculatorul cu care lucraţi dvs.

4.5. Complexitatea temporală a algoritmilor
În informatică complexitatea temporală a algoritmilor se caracterizează prin tim-

pul de execuţie T(n) sau numărul de operaţii elementare Q(n). Întrucît calculatoarele 
moderne au o viteză de calcul foarte mare  108 ... 1010 instrucţiuni pe secundă, pro-
blema timpului de execuţie se pune numai pentru valorile mari ale lui n. În conse-
cinţă, în formulele ce exprimă numărul de operaţii elementare Q(n) prezintă interes 
numai termenul dominant, adică acel care tinde cît mai repede la infi nit. Importanţa 
termenului dominant faţă de ceilalţi este pusă în evidenţă în tabelul 4.5.

Tabelul 4.5
Valorile termenilor dominanţi

n log2 n n2 n3 n4 2n

2 1 4 8 16 4
4 2 16 64 256 16
8 3 64 512 4096 256
16 4 256 4096 65536 65536
32 5 1024 32768 1048576 4294967296

De exemplu, numărul de operaţii elementare ale procedurii Sortare se exprimă 
prin formula:

Q(n) = 16n2 – 13n + 2.
Termenul dominant din această formulă este 16n2. Evident, pentru valorile mari 

ale lui n numărul de operaţii elementare
Q(n)  16n2,

iar timpul de execuţie
T(n)  16n2.

În funcţie de complexitatea temporală, algoritmii se clasifi că în:
 algoritmi polinomiali;
 algoritmi exponenţiali;
 algoritmi nederminist polinomiali.
Un algoritm se numeşte polinomial dacă termenul dominant are forma Cnk, adică

Q(n)  Cnk ;   T(n)  Cnk,
unde:

n este caracteristica datelor de intrare;
C  o constantă pozitivă;
k  un număr natural.
Complexitatea temporală a algoritmilor polinomiali este redată prin notaţia 

O(nk) care se citeşte „algoritm cu timpul de execuţie de ordinul nk” sau, mai pe 
scurt, „algoritm de ordinul nk”. Evident, există algoritmi polinomiali de ordinul 
n, n2, n3 etc.

De exemplu, algoritmul de sortare a elementelor unui vector prin metoda bulelor 
este un algoritm polinomial de ordinul n2. Acest lucru se observă şi din grafi cul tim-
pului de execuţie T(n) a procedurii Sortare, grafi c prezentat în fi gura 4.3.

Un algoritm se numeşte exponenţial dacă termenul dominant are forma Ckn, adică
Q(n)  Ckn;    T(n)  Ckn,

unde k>1. Complexitatea temporală a algoritmilor exponenţiali este redată prin no-
taţia O(kn).

Menţionăm faptul că tot exponenţiali se consideră şi algoritmii de complexitatea 
nlog n, cu toate că această funcţie nu este exponenţială în sensul strict matematic al 
acestui cuvînt.

Din comparaţia vitezei de creştere a funcţiilor exponenţială şi polinomială (vezi 
tabelul 4.5) rezultă că algoritmii exponenţiali devin inutilizabili chiar pentru valori 
nu prea mari ale lui n. Pentru exemplifi care amintim tabelul 4.1 în care este prezentat 
timpul de execuţie T1(n) a unui algoritm polinomial de ordinul O(n3) şi timpul de 
execuţie T2(n) a unui algoritm exponenţial de ordinul O(2n).

Algoritmii nederminist polinomiali se studiază în cursurile avansate de infor-
matică.

În funcţie de complexitatea temporală se consideră că o problemă este uşor re-
zolvabilă dacă pentru soluţionarea ei există un algoritm polinomial. O problemă 
pentru care nu există un algoritm polinomial se numeşte difi cilă. Accentuăm faptul 
că această clasifi care se referă doar la analiza asimptotică a algoritmilor, adică la 
comportarea lor pentru valori mari ale lui n. Pentru valorile mici ale lui n situaţia 
poate fi  diferită.

De exemplu, presupunem că pentru soluţionarea unei probleme există doi algo-
ritmi, unul polinomial cu timpul de execuţie T1(n) şi altul exponenţial cu timpul de 
execuţie T2(n):

T1(n) = 1000n2 ;
T2(n) = 2n.

Prin calcule directe se poate verifi ca că pentru n = 1, 2, 3, ..., 18 timpul T2(n) < T1(n). 
Prin urmare, în situaţia n  18 vom prefera algoritmul exponenţial.

În practică, elaborarea programelor de calculator presupune parcurgerea urmă-
toarelor etape:

– formularea exactă a problemei;
– determinarea complexităţii temporale a problemei propuse  problemă uşor 

rezolvabilă sau problemă difi cilă;
– elaborarea algoritmului respectiv şi implementarea lui pe un sistem de calcul.
Evident, în cazul unor probleme uşor rezolvabile, programatorul va depune toate 

eforturile pentru a inventa algoritmi polinomiali, adică algoritmi de ordinul nk, astfel 
încît parametrul k să ia valori cît mai mici. În cazul problemelor difi cile se va da prio-
ritate algoritmilor care minimizează timpul de execuţie cel puţin pentru datele de in-
trare frecvent utilizate în aplicaţiile practice. Metodele de clasifi care a problemelor în 
cele uşor şi cele difi cil rezolvabile se studiază în cursurile avansate de informatică.

Întrebări şi exerciţii
Indicaţi termenii dominanţi:

a) 12n + 5;

b) 6n2 + 100n + 18;

c) 15n3 + 1000n2 – 25n + 6000;

d) 2000n3 + 2n + 13;

e) nlog2n + n5 + 300n2 + 6;

f ) 3n + 2n + 14n3 + 21;

g) n5 + 10n4 + 200n3 + 300n2 + 1000n.

Cum se clasifi că algoritmii în funcţie de complexitatea temporală?
Determinaţi tipul algoritmilor, cunoscînd complexitatea temporală:

a) Q(n) = 200n + 15;

b) Q(n) = 2n + 25n2 + 1000;

c) Q(n) = n3 + 3n + 6000n2 + 106;

d) Q(n) = 3n + 2n + n10 + 4000

e) Q(n) = nlog2n + n3 + n2 + 1500;

f ) Q(n) = 100n2 + 15n3 + 8n + 900.

 Cum credeţi, prin ce se explică importanţa termenului dominant în analiza complexităţii 
temporale a algoritmilor?

 Se consideră procedurile N2, N3 și N4 din paragraful precedent. În urma estimării numă-
rului de operaţii elementare s-a obţinut:

QN2(n) = 602n2 + 2n + 2;
QN3(n) = 2n3 + 2n2 + 2n + 2;
QN4(n) = 2n4 + 2n3 + 2n2 + 2n + 2.

 Determinaţi ordinul de complexitate temporală a algoritmilor descriși cu ajutorul aces-
tor proceduri. Comparaţi viteza de creștere a timpilor de execuţie TN2(n), TN3(n) și TN4(n), 
folosind în acest scop grafi cele din fi gura 4.4.

Se consideră un algoritm format din k cicluri imbricate:

for i1:=1 to n do
   for i2:=1 to n do
       ...
           for ik:=1 to n do P

 Numărul de operaţii elementare QP efectuate în procedura P este o mărime constantă. 
Estimaţi complexitatea temporală a algoritmului.

Schiţaţi un algoritm pentru rezolvarea următoarei probleme:
 Se consideră mulţimea A formată din n numere întregi. Determinaţi dacă există cel puţin 

o submulţime B, BA, suma elementelor căreia este egală cu m. De exemplu, pentru 
A={-3, 1, 5, 9} și m=7, o astfel de submulţime există, și anume, B={-3, 1, 9}.

 Estimaţi complexitatea temporală a algoritmului elaborat.

Capitolul 5

TEHNICI  DE  ELABORARE 
A  ALGORITMILOR

5.1. Iterativitate sau recursivitate

Pe parcursul dezvoltării informaticii s-a stabilit că multe probleme de o reală im-
portanţă practică pot fi  rezolvate cu ajutorul unor metode standard, denumite teh-
nici de programare: recursia, trierea, metoda reluării, metodele euristice ş.a.

Una din cele mai răspîndite tehnici de programare este recursia. Amintim că re-
cursia se defi neşte ca o situaţie în care un subprogram se autoapelează fi e direct, fi e 
prin intermediul altui subprogram. Tehnicile în studiu se numesc respectiv recur-
sia directă şi recursia indirectă şi au fost studiate în cadrul temei „Funcţii şi proce-
duri”.

În general, elaborarea unui program recursiv este posibilă numai atunci cînd se 
respectă următoarea regulă de consistenţă: soluţia problemei trebuie să fi e direct 
calculabilă ori calculabilă cu ajutorul unor valori direct calculabile. Cu alte cuvinte, 
defi nirea corectă a unui algoritm recursiv presupune că în procesul derulării calcu-
lelor trebuie să existe:

– cazuri elementare, care se rezolvă direct;
– cazuri care nu se rezolvă direct, însă procesul de calcul în mod obligatoriu pro-

gresează spre un caz elementar.
De exemplu, în defi niţia recursivă a funcţiei factorial fact: N→N,

deosebim:
− Cazul elementar n = 0.  În acest  caz valoarea fact(0) este direct calculabilă şi 

anume fact(0)=1.
− Cazurile neelementare n > 0. În astfel de cazuri valorile fact(n) nu sînt direct 

calculabile, însă procesul de calcul progresează către cazul elementar fact(0).
De exemplu, pentru n = 3 obţinem:

fact(3) = 3 · fact(2) = 3 · 2 · fact(1) = 3 · 2 · 1  fact(0) = 3 · 2 · 1 ·1 = 6.

Prin urmare, defi niţia recursivă a funcţiei fact(n) este o defi niţie consistentă. 
Amintim că funcţia fact(n) poate fi  exprimată în PASCAL, urmînd direct defi niţia, 
în forma:

function Fact(n:Natural):Natural;
begin
  if n=0 then Fact:=1
  else Fact:=nFact(n-1)
end;

Modul de gestionare a stivei în cazul apelului Fact(3) este prezentat în fi gura 5.1.

Fact = 1 Fact = 1
Eliberarea stivei

Evoluţia în stivă

Fact = 2 Fact = 6

Fact(3) Fact(2) Fact(1) Fact(0)

Fig. 5.1. Gestionarea stivei în cazul apelului Fact(3):
AR – adresa de revenire; n – valoarea curentă a parametrului actual;
*** – spaţiu pentru memorarea valorii f returnate de funcţia Fact

Într-un mod similar, în defi niţia recursivă a funcţiei  incons: N→N,

deosebim cazul elementar n=0 şi cazurile neelementare n0. Însă, spre deosebire de 
funcţia fact(n), pentru n  0 valorile incons(n) nu pot fi  calculate, întrucît procesul de 
calcul progresează către incons().

De exemplu, pentru n=3 obţinem:
incons(3) = 3 · incons(4) = 3 · 4 · incons(5) = 3 · 4 · 5 · incons(6) = ···.

Prin urmare, defi niţia recursivă a funcţiei incons(n) este o defi niţie inconsistentă 
şi teoretic procesul de calcul va dura la nesfîrşit. În practică calculele vor fi  întrerupte 
de sistemul de operare în momentul depăşirii capacităţii de memorare a stivei sau în 
cazul depăşirii capacităţii dispozitivului aritmetic.

Accentuăm faptul că mediile actuale de programare nu asigură verifi carea consis-
tenţei algoritmilor recursivi, acest lucru revenindu-i programatorului.

După cum s-a văzut în capitolele precedente, orice algoritm recursiv poate fi  tran-
scris într-un algoritm iterativ şi invers. Alegerea tehnicii de programare – iterativi-
tate sau recursivitate – ţine, de asemenea, de competenţa programatorului. Evident, 
această alegere trebuie făcută luînd în considerare avantajele şi neajunsurile fi ecărei 
metode, care variază de la caz la caz. Pentru exemplifi care, în tabelul 5.1. sînt prezen-
tate rezultatele unui studiu comparativ al ambelor tehnici de programare în cazul 
prelucrării automate a textelor.

Tabelul 5.1 
Studiul comparativ al iterativităţii şi recursivităţii                                                            

(prelucrarea automată a textelor)
Nr. crt. Caracteristici Iterativitate Recursivitate

1. Necesarul de memorie mic mare
2. Timpul de execuţie acelaşi
3. Structura programului complicată simplă
4. Volumul de muncă necesar 

pentru scrierea programului mare mic

5. Testarea şi depanarea progra-
melor simplă complicată

Propunem cititorului în calitate de exerciţiu efectuarea unui studiu comparativ al 
iterativităţii şi recursivităţii în cazul algoritmilor destinaţi creării şi preluării structu-
rilor dinamice de date din Capitolul 2.

În general, algoritmii recursivi sînt recomandaţi în special pentru problemele ale 
căror rezultate sînt defi nite prin relaţii de recurenţă: analiza sintactică a textelor, pre-
lucrarea structurilor dinamice de date, procesarea imaginilor ş.a. Un exemplu tipic 
de astfel de probleme este analiza gramaticală a programelor PASCAL, sintaxa căro-
ra, după cum se ştie, este defi nită prin relaţii de recurenţă.

Întrebări şi exerciţii
Explicaţi termenul tehnici de programare.
Care este diferenţa dintre recursia directă și recursia indirectă? Daţi exemple.
Ce condiţii trebuie respectate pentru ca defi niţia unui algoritm recursiv să fi e corectă?
Care este diferenţa dintre defi niţiile recursive consistente și defi niţiile recursive inconsis-

tente?

Se consideră următoarele defi niţii recursive de funcţii. Care din aceste defi niţii sînt con-
sistente? Argumentaţi răspunsul.

 a)  f : N→N,  
dacă

dacă

 b)  f : N→N,  
dacă

dacă

 c)  f : Z→Z,   
dacă

dacă

 d)  f : N→N,  
dacă

dacă

      g : N→N,  
dacă

dacă

 e)  f : N→N,  
dacă

dacădiv

Lansaţi în execuţie programul ce urmează. Explicaţi mesajele afi șate la ecran.

Program P150;
 { Depăşirea capacităţii de memorare a stivei }
type Natural = 0..Maxint;

function Incons(n:Natural):Natural;
 { Defi niţie recursivă inconsistentă }
begin
  writeln(’Apel recursiv n=’, n);
  if n=0 then Incons:=1
         else Incons:=n*Incons(n+1);
end; { Incons }

begin
  writeln(Incons(3));
  readln;
end.

 Reprezentaţi pe un desen similar celui din fi gura 5.1 modul de gestionare a stivei în cazul 
apelului Incons(3).

Indicaţi cazurile elementare și cele neelementare pentru următorii algoritmi recursivi:
 a) funcţia Arb și procedura Afi sArb din programul P130;
 b) procedurile Preordine, Inordine și Postordine din programul P131;
 c) procedura InAdincime din programul P133.
Suma primelor n numere naturale S(n) poate fi  calculată cu ajutorul funcţiei iterative

 sau al funcţiei recursive
dacă

dacă

 Utilizînd ca model tabelul 5.1, efectuaţi un studiu comparativ al algoritmilor destinaţi 
calculării sumei S(n).

Se consideră următoarele formule metalingvistice:
Cifră ::= 0123456789
Număr ::= Cifră  Cifră
Semn ::= +-*/
Expresie ::= Număr (Expresie)ExpresieSemnExpresie

 Scrieţi o funcţie recursivă care returnează valoarea true dacă șirul de caractere S este 
conform defi niţiei unităţii lexicale Expresie și false în caz contrar.

Efectuaţi un studiu comparativ al algoritmilor iterativi și algoritmilor recursivi destinaţi 
creări și prelucrării următoarelor structuri dinamice de date:

 a) liste unidirecţionale;
 b) stiva;
 c) cozi;
 d) arbori binari;
 e) arbori de ordinul m.

 Scrieţi o funcţie iterativă care returnează valoarea true dacă șirul de caractere S este 
conform defi niţiei unităţii lexicale Expresie din exerciţiul 9 și false în caz contrar. 
Utilizînd ca model tabelul 5.1, efectuaţi un studiu comparativ al algoritmului iterativ și 
algoritmului recursiv.

 Elaboraţi un subprogram recursiv care calculează suma cifrelor unui număr natural n. 
Examinaţi cazurile  n  MaxInt   și   n  10250.

 Imaginile în alb-negru (fig. 5.2) pot fi codificate cu ajutorul unei matrice binare B = 
= ||bij||nm, 1 n, m  30. Elementul bij indică culoarea microzonei respective: neagră (bij 
=1) sau albă (bij =0).

a) b)

Fig. 5.2. Colorarea unei suprafeţe închise:
a – imaginea iniţială; b – imaginea fi nală

 Elaboraţi o procedură recursivă pentru colorarea unei suprafeţe închise, specifi cate prin 
coordonatele (i, j) ale oricărei microzone A din interiorul ei.

 Elaboraţi o procedură care determină cîte obiecte conţine o imagine în alb-negru. 
Imaginea este împărţită în microzone și codifi cată cu ajutorul matricei binare B = ||bij||nm. 
Elementul bij indică culoarea microzonei cu coordonatele (i, j). 

 Efectuaţi un studiu comparativ al algoritmilor iterativi și algoritmilor recursivi destinaţi 
soluţionării problemelor din exerciţiile 13 și 14.

5.2. Metoda trierii
Metoda trierii presupune că soluţia unei probleme poate fi  găsită analizînd con-

secutiv elementele si ale unei mulţimi fi nite
S = {s1, s2, …,  si, …, sk},

denumită mulţimea soluţiilor posibile. În cele mai simple cazuri  elementele si, si  
S, pot fi  reprezentate prin valori aparţinînd unor tipuri ordinale de date: integer, 
boolean, char, enumerare sau subdomeniu. În problemele mai complicate sîntem ne-
voiţi să reprezentăm aceste elemente prin tablouri, articole sau mulţimi. Menţionăm 
că în majoritatea problemelor soluţiile posibile s1, s2, …, sk nu sînt indicate explicit în 
enunţul problemei şi elaborarea algoritmilor pentru calcularea lor cade în sarcina 
programatorului.

Schema generală a unui algoritm bazat pe metoda trierii poate fi  redată cu ajuto-
rul unui ciclu:

for i:= 1 to k do
   if SolutiePosibila(si) then PrelucrareaSolutiei(si)

unde SolutiePosibila este o funcţie booleană care returnează valoarea 
true  dacă elementul si satisface condiţiile  problemei şi false în caz contrar, iar 
PrelucrareaSolutiei este o procedură care efectuează prelucrarea elementului 
selectat. De obicei, în această procedură soluţia si  este afi şată la ecran.

În continuare vom analiza două exemple care pun în evidenţă avantajele şi nea-
junsurile algoritmilor bazaţi pe metoda trierii.

Exemplul 1. Se consideră numerele naturale din mulţimea {0, 1, 2, ..., n}. Elaboraţi 
un program care determină pentru cîte numere K din această mulţime suma cifrelor 
fi ecărui număr este egală cu m. În particular, pentru n = 100 şi m = 2, în mulţimea 
{0, 1, 2, …, 100} există 3 numere care satisfac condiţiile problemei: 2, 11 şi 20. Prin 
urmare, K = 3.

Rezolvare. Evident, mulţimea soluţiilor posibile S = {0, 1, 2, …, n}. În programul ce 
urmează suma cifrelor oricărui număr natural i, iS, se calculează cu ajutorul funcţiei 
SumaCifrelor. Separarea cifrelor zecimale din scrierea numărului natural i se efectuea-
ză de la dreapta la stînga prin împărţirea numărului i şi a cîturilor respective la baza 10.

Program P151;
 { Suma cifrelor unui număr natural }
type Natural=0..MaxInt;
var i, K, m, n : Natural;

function SumaCifrelor(i:Natural):Natural;
var suma : Natural;
begin
  suma:=0;
  repeat
    suma:=suma+(i mod 10);
    i:=i div 10;
  until i=0;
  SumaCifrelor:=suma;
end; { SumaCifrelor }

function SolutiePosibila(i:Natural):boolean;
begin
  if SumaCifrelor(i)=m then SolutiePosibila:=true
                       else SolutiePosibila:=false;
end; { SumaCifrelor }

procedure PrelucrareaSolutiei(i:Natural);
begin
  writeln(’i=’, i);
  K:=K+1;
end; { PrelucrareaSolutiei }

begin
  write(’Daţi n=’); readln(n);
  write(’Daţi m=’); readln(m);
  K:=0;
  for i:=0 to n do
    if SolutiePosibila(i) then PrelucrareaSolutiei(i);
  writeln(’K=’, K);
  readln;
end.

Din analiza programului P151 rezultă că complexitatea temporară a algoritmu-
lui respectiv este O(n).

Exemplul 2. Se consideră mulţimea  P = {P1, P2, …, Pn}  formată din n puncte (2  n 
 30) pe un plan euclidian. Fiecare punct Pj este defi nit prin coordonatele sale xj, yj. 
Elaboraţi un program  care afi şează la ecran coordonatele punctelor Pa, Pb distanţa 
dintre care este maximă.

Rezolvare. Mulţimea soluţiilor posibile S = PP.  Elementele (Pj, Pm) ale produsului 
cartezian PP pot fi  generate  cu ajutorul a două cicluri imbricate:

for j:=1 to n do
  for m:=1 to n do
    if SolutiePosibilă(Pj, Pm) then PrelucrareaSolutiei(Pj, Pm)

Distanţa dintre punctele  Pj, Pm  se calculează cu ajutorul formulei:

.

Program P152;
 { Puncte pe un plan euclidian }
const nmax=30;
type Punct = record
              x, y : real;
             end;
     Indice = 1..nmax;
var P : array[Indice] of Punct;
    j, m, n : Indice;
    dmax : real; { distanţa maxima }
    PA, PB : Punct;

function Distanta(A, B : Punct) : real;
begin
  Distanta:=sqrt(sqr(A.x-B.x)+sqr(A.y-B.y));
end; { Distanta }

function SolutiePosibila(j,m:Indice):boolean;
begin
  if j<>m then SolutiePosibila:=true
          else SolutiePosibila:=false;
end; { SolutiePosibila }

procedure PrelucrareaSolutiei(A, B : Punct);
begin
  if Distanta(A, B)>dmax then
    begin
      PA:=A; PB:=B;
      dmax:=Distanta(A, B);
    end;
end; { PrelucrareaSolutiei }

begin
  write(’Dati n=’); readln(n);
  writeln(’Daţi coordonatele x, y ale punctelor’);
  for j:=1 to n do
    begin
      write(’P[’, j, ’]: ’); readln(P[j].x, P[j].y);
    end;
  dmax:=0;

  for j:=1 to n do
    for m:=1 to n do

       if SolutiePosibila(j, m) then
          PrelucrareaSolutiei(P[j], P[m]);

  writeln(’Soluţia: PA=(’, PA.x:5:2, ’,’, PA.y:5:2, ’)’);
  writeln(’         PB=(’, PB.x:5:2, ’,’, PB.y:5:2, ’)’);
  readln;
end.

Complexitatea temporală a algoritmului descris cu ajutorul programului P152 
este O(n2).

Din exemplele prezentate mai sus se observă că în algoritmii bazaţi pe metoda tri-
erii se calculează, implicit sau explicit, mulţimea soluţiilor posibile S. În problemele 
relativ simple (exemplul 1) elementele din mulţimea soluţiilor posibile pot fi  enume-
rate direct. În problemele mai complicate (exemplul 2) generarea soluţiilor posibile 
necesită elaborarea unor algoritmi speciali. În general, aceşti algoritmi realizează 
operaţiile legate de prelucrarea unor mulţimi:

– reuniunea;
– intersecţia;
– diferenţa;
– generarea tuturor submulţimilor;
– generarea elementelor unui produs  cartezian;
– generarea permutărilor, aranjamentelor sau combinărilor de obiecte etc.
Avantajul principal al algoritmilor bazaţi pe metoda trierii constă în faptul că 

programele respective sînt relativ simple, iar depanarea lor nu necesită teste sofi sti-
cate. Complexitatea temporală a acestor algoritmi este determinată de numărul de 
elemente k din mulţimea soluţiilor posibile S. În majoritatea problemelor de o reală 
importanţă practică metoda trierii conduce la algoritmii exponenţiali. Întrucît algo-
ritmii exponenţiali sînt inacceptabili în cazul datelor  de intrare foarte mari, metoda 
trierii este aplicată numai în scopuri didactice sau pentru elaborarea unor programe 
al căror timp de execuţie nu este critic.

Întrebări şi exerciţii
Explicaţi structura generală a algoritmilor bazaţi pe metoda trierii.
Cum poate fi  realizată trierea soluţiilor posibile cu ajutorul ciclurilor while și repeat?
Estimaţi timpul de execuţie al programelor P151 și P152. Modifi caţi programul P152 

astfel, încît timpul de execuţie să se micșoreze aproximativ de două ori.
Daţi exemple de programe timpul de execuţie al cărora nu este critic.
Care sînt avantajele și dezavantajele algoritmilor bazaţi pe metoda trierii?
Se consideră mulţimea  P = {P1, P2, …, Pn}  formată din n puncte (3  n  30) pe un plan 

euclidian. Fiecare punct Pj este defi nit prin coordonatele sale xj, yj. Elaboraţi  un program 
ce determină trei puncte din mulţimea P pentru care aria triunghiului respectiv este 
maximă. Estimaţi timpul de execuţie a programului elaborat.

Scrieţi o funcţie PASCAL care, primind ca parametru un număr natural n, returnează va-
loarea true dacă n este prim și false în caz contrar. Estimaţi complexitatea temporală 
a funcţiei respective.

În notaţia (a)x litera x reprezintă baza sistemului de numeraţie, iar litera a – un număr 
scris în sistemul respectiv. Elaboraţi un program care calculează, dacă există, cel puţin o 
rădăcină a ecuaţiei

(a)x = b,
 unde a și b sînt numere naturale, iar x este necunoscuta. Fiecare cifră a numărului na-

tural a aparţine mulţimii {0, 1, 2, …, 9}, iar numărul b este scris în sistemul zecimal. De 
exemplu, rădăcina ecuaţiei

(160)x = 122
 este x = 8, iar ecuaţia

(5)x = 10
 nu are soluţii. Estimaţi complexitatea temporală a programului elaborat.
Într-o pușculiţă se afl ă N monede de diferite valori cu greutatea totală G grame. Greutatea 

fi ecărei monede de o anumită valoare este dată în tabelul ce urmează.

Valoarea monedei, lei Greutatea monedei, grame
1 1
5 2
10 3
25 4
50 5

 Elaboraţi un program care determină suma minimă S care ar putea să fi e în pușculiţă.
Elaboraţi un program care determină cîte puncte cu coordonate întregi se conţin într-o 

sferă de raza R cu centrul în originea sistemului de coordonate. Se consideră că R este 
un număr natural, 1R30. Distanţa d dintre un punct cu coordonatele (x, y, z) și originea 
sistemului de coordonate se determină după formula .

5.3. Tehnica Greedy
Această metodă presupune că problemele pe care trebuie să le rezolvăm au ur-

mătoarea structură:
 se dă o mulţime A={a1, a2, ..., an} formată din n elemente;
 se cere să determinăm o submulţime B, BA, care îndeplineşte anumite condiţii 

pentru a fi  acceptată ca soluţie.
În principiu, problemele de acest tip pot fi  rezolvate prin metoda trierii, generînd 

consecutiv cele 2n submulţimi Ai ale mulţimii A. Dezavantajul metodei trierii constă 
în faptul că timpul cerut de algoritmii respectivi este foarte mare.

Pentru a evita trierea tuturor submulţimilor Ai, AiA, în metoda Greedy se uti-
lizează un criteriu (o regulă) care asigură alegerea directă a elementelor necesare 
din mulţimea A. De obicei, criteriile sau regulile de selecţie nu sînt indicate explicit 
în enunţul problemei şi formularea lor cade în sarcina programatorului. Evident, în 
absenţa unor astfel de criterii metoda Greedy nu poate fi  aplicată.

Schema generală a unui algoritm bazat pe metoda Greedy poate fi  redată cu aju-
torul unui ciclu:

while ExistaElemente do
begin
  AlegeUnElement(x);
  IncludeElementul(x);
end

Funcţia ExistaElemente returnează valoarea true dacă în mulţimea A exis-
tă elemente care satisfac criteriul (regula) de selecţie. Procedura AlegeUnElement 
extrage din mulţimea A un astfel de element x, iar procedura IncludeElementul 
înscrie elementul selectat în submulţimea B. Iniţial B este o mulţime vidă.

După cum se vede, în metoda Greedy soluţia problemei se caută prin testarea conse-
cutivă a elementelor din mulţimea A şi prin includerea unora din ele în submulţimea 
B. Într-un limbaj plastic, submulţimea B încearcă să „înghită” elementele „gustoase” 
din mulţimea A, de unde provine şi denumirea metodei (greedy  lacom, hrăpăreţ).

Exemplu. Se consideră mulţimea A={a1, a2, ..., ai, ..., an} elementele căreia sînt nume-
re reale, iar cel puţin unul din ele satisface condiţia ai>0. Elaboraţi un program care 
determină o submulţime B, BA, astfel încît suma elementelor din B să fi e maximă. 
De exemplu, pentru A={21,5; -3,4; 0; -12,3; 83,6} avem B={21,5; 83,6}.

Rezolvare. Se observă că dacă o submulţime B, BA, conţine un element b0, atunci 
suma elementelor submulţimii B \{b} este mai mare sau egală cu cea a elementelor 
din B. Prin urmare, regula de selecţie este foarte simplă: la fi ecare pas în submulţi-
mea B se include un element pozitiv arbitrar din mulţimea A.

În programul ce urmează mulţimile A şi B sînt reprezentate prin vectorii (tablo-
urile unidimensionale) A şi B, iar numărul de elemente ale fi ecărei mulţimi  prin 
variabilele întregi, respectiv n şi m. Iniţial B este o mulţime vidă, respectiv m=0.

Program P153;
 { Tehnica Greedy }
const nmax=1000;
var A : array [1..nmax] of real;
    n : 1..nmax;
    B : array [1..nmax] of real;
    m : 0..nmax;
    x : real;
    i : 1..nmax;

Function ExistaElemente : boolean;
var i : integer;
begin
  ExistaElemente:=false;
  for i:=1 to n do
    if A[i]>0 then ExistaElemente:=true;
  end; { ExistaElemente }

procedure AlegeUnElement(var x : real);
var i : integer;

begin
  i:=1;
  while A[i]<=0 do i:=i+1;
  x:=A[i];
  A[i]:=0;
end; { AlegeUnElement }

procedure IncludeElementul(x : real);
begin
  m:=m+1;
  B[m]:=x;
end; { IncludeElementul }

begin
  write(’Daţi n=’); readln(n);
  writeln(’Daţi elementele mulţimii A:’);
  for i:=1 to n do read(A[i]);
  writeln;
  m:=0;
  while ExistaElemente do
    begin
      AlegeUnElement(x);
      IncludeElementul(x);
    end;
  writeln(’Elementele mulţimii B:’);
  for i:=1 to m do writeln(B[i]);
  readln;
end.

Menţionăm că procedura AlegeUnElement zerografi ază componenta vectoru-
lui A ce conţinea elementul x, extras din mulţimea respectivă.

Complexitatea temporală a algoritmilor bazaţi pe metoda Greedy poate fi  evalu-
ată urmînd schema generală de calcul prezentată mai sus. De obicei, timpul cerut 
de procedurile ExistaElemente, AlegeUnElement şi IncludeElementul este 
de ordinul n. În componenţa ciclului while aceste proceduri se execută cel mult de 
n ori. Prin urmare, algoritmii bazaţi pe metoda Greedy sînt algoritmi polinomiali. 
Pentru comparare, amintim că algoritmii bazaţi pe trierea tuturor submulţimilor Ai, 
Ai A, sînt algoritmi de ordinul O(2n), deci exponenţiali. Cu regret, metoda Greedy 
poate fi  aplicată numai atunci cînd din enunţul problemei poate fi  dedusă regula 
care asigură selecţia directă a elementelor necesare din mulţimea A.

Întrebări şi exerciţii
Explicaţi structura generală a algoritmilor bazaţi pe metoda Greedy.
Care sînt avantajele și neajunsurile algoritmilor bazaţi pe tehnica Greedy?
Estimaţi timpul de execuţie al programului P153.

Schiţaţi un algoritm care determină submulţimea B din exemplul de mai sus prin meto-
da trierii. Estimaţi complexitatea temporală a algoritmului elaborat.

Memorarea fi șierelor pe benzi magnetice. Se consideră n fi șiere f1, f2, ..., fn care trebu-
ie memorate pe o bandă magnetică. Elaboraţi un program care determină ordinea de 
amplasare a fi șierelor pe bandă astfel încît timpul mediu de acces să fi e minim. Se pre-
supune că frecvenţa de citire a fi șierelor este aceeași, iar pentru citirea fi șierului fi (i=1, 2, 
..., n) sînt necesare ti secunde.

Problema continuă a rucsacului. Se consideră n obiecte. Pentru fi ecare obiect i (i=1, 
2, ..., n) se cunoaște greutatea gi și cîștigul ci care se obţine în urma transportului său 
la destinaţie. O persoană are un rucsac cu care pot fi  transportate unul sau mai multe 
obiecte greutatea sumară a cărora nu depășește valoarea Gmax. Elaboraţi un program 
care determină ce obiecte trebuie să transporte persoana în așa fel încît cîștigul să fi e 
maxim. În caz de necesitate, unele obiecte pot fi  tăiate în fragmente mai mici.

Hrubele de la Cricova. După o vizită la renumitele hrube  * de la Cricova un informatician 
a construit un robot care funcţionează într-un cîmp divizat în pătrăţele (fi g. 5.3). Robotul 
poate executa doar instrucţiunile SUS, JOS, DREAPTA, STINGA, conform cărora se de-
plasează în unul din pătrăţelele vecine. Dacă în acest pătrat este un obstacol, de exem-
plu, un perete sau un butoi, are loc un accident și robotul iese din funcţiune.

Robot

Intrare Ieşire

Fig. 5.3. Cîmpul de acţiune al robotului

 Elaboraţi un program care, cunoscînd planul hrubelor, deplasează robotul prin încăperi-
le subterane, de la intrarea în hrube la ieșire. Colecţia de vinuri fi ind foarte bogată, nu se 
cere vizitarea obligatorie a tuturor încăperilor subterane.

 Datele de intrare. Planul hrubelor este desenat pe o foaie de hîrtie liniată în pătrăţe-
le. Pătrăţelele hașurate reprezintă obstacolele, iar cele nehașurate – spaţiile libere. 
Pătrăţelele de pe perimetrul planului, cu excepţia celor de intrare sau ieșire, sînt hașu-
rate prin defi niţie. În formă numerică planul hrubelor este redat prin tabloul A cu m linii 
și n coloane. Elementele A[i, j]  ale acestui tablou au următoarea semnifi caţie: 0 – spaţiu 

* Hrubă – încăpere sau galerie subterană care serveşte la depozitarea produselor alimentare. 
În hrubele de la Cricova pe parcursul mai multor decenii au fost depozitate cele mai bune 
vinuri din Republica Moldova.

liber; 1 – obstacol; 2 – intrarea în hrube; 3 – ieșirea din hrube. Iniţial, robotul se afl ă în 
pătrăţelul pentru care A[i, j]=2.

 Fișierul HRUBE.IN conţine pe prima linie numerele m, n separate prin spaţiu. Pe urmă-
toarele m linii se conţin cîte n numere A[i, j] separate prin spaţiu.

 Datele de ieșire. Fișierul HRUBE.OUT va conţine pe fi ecare linie cîte una din instrucţiuni-
le SUS, JOS, DREAPTA, STINGA scrise în ordinea executării lor de către robot.

 Restricţii. 5  m, n  100 . Timpul de execuţie nu va depăși 3 secunde.
 Exemplu:

HRUBE.IN HRUBE.OUT

7 9
1 1 1 1 1 1 1 1 1
1 0 0 1 0 0 0 0 1
1 0 1 0 0 1 0 1 1
1 0 0 0 1 0 0 0 1
1 0 1 0 1 0 1 0 1
1 0 0 0 0 0 1 0 1
1 2 1 1 1 1 1 3 1

SUS
DREAPTA
DREAPTA
DREAPTA
DREAPTA
SUS
SUS
DREAPTA
DREAPTA
JOS
JOS
JOS

 Indicaţii. La fi ecare pas selectaţi din mulţimea {SUS, JOS, DREAPTA, STINGA} cîte o 
instrucţiune în așa fel încît robotul să se deplaseze numai de-a lungul unui perete.

5.4. Metoda reluării
În metoda reluării se presupune că soluţia problemei pe care trebuie să o rezol-

văm poate fi  reprezentată printr-un vector:
X=(x1, x2 , ..., xk , ... , xn).

Fiecare componentă xk a vectorului X poate lua valori dintr-o anumită mulţime 
Ak, k = 1, 2, ..., n. Se consideră că cele mk elemente ale fi ecărei mulţimi Ak sînt ordonate 
conform unui criteriu bine stabilit, de exemplu, în ordinea apariţiei lor în memoria 
calculatorului.

În principiu, astfel de probleme pot fi  soluţionate prin metoda trierii, tratînd vec-
torul X ca un element al produsului cartezian S = A1A2 ... An. Dezavantajul me-
todei trierii constă în faptul că timpul cerut de algoritmul respectiv este foarte mare. 
De exemplu, pentru mulţimile A1, A2, ..., An formate numai din cîte două elemente, 
timpul necesar este O(2n), deci exponenţial.

Pentru evitarea trierii tuturor elementelor produsului cartezian A1A2 ... An, în 
metoda reluării componentele vectorului X primesc valori pe rînd, în sensul că lui xk 
i se atribuie o valoare numai dacă au fost deja atribuite valori lui x1, x2 , ..., xk-1 . Mai 
mult, odată o valoare pentru xk stabilită, nu se trece direct la atribuirea de valori lui 
xk+1, ci se verifi că anumite condiţii de continuare referitoare la x1, x2 , ..., xk. Aceste 

condiţii stabilesc situaţiile în care are sens să trecem la calculul lui xk+1. Dacă aceste 
condiţii nu sînt satisfăcute, va trebui să facem o altă alegere pentru xk sau, dacă ele-
mentele din mulţimea Ak s-au epuizat, să micşorăm pe k cu o unitate încercînd să 
facem o nouă alegere pentru xk-1.

Menţionăm faptul că anume micşorarea lui k dă nume metodei studiate, cuvîntul 
„reluare” semnifi cînd revenirea la alte variante de alegere a variabilelor x1, x2 , ..., xk-1. 
Evident, aceeaşi semnifi caţie o are şi denumirea engleză a metodei în studiu  back-
tracking (back  înapoi, track  urmă).

Pentru exemplifi care, în fi gura 5.4 este prezentată ordinea în care sînt examinate 
elementele mulţimilor A1 = {a11, a12}, A2 = {a21, a22} şi A3 = {a31, a32, a33}. În scopuri di-
dactice se consideră că mulţimile A1 şi A2 au cîte două elemente (m1 = 2, m2 = 2), iar 
mulţimea A3  trei elemente (m3 = 3). Elementele akj ale mulţimilor respective sînt 
simbolizate prin cerculeţe. Rezultatele verifi cării condiţiilor de continuare sînt repre-
zentate prin valorile binare 0 (false) şi 1 (true).

A1

A2

A3

k:=1

k:=k+1

0
0

k:=k+1 k:=k–1

k:=k+1

0

0
0

Fig. 5.4. Căutarea soluţiei prin metoda reluării

Din fi gura 5.4 se observă că primul element a11 din mulţimea A1 nu satisface con-
diţiile de continuare şi, în consecinţă, se trece la elementul al doilea a12 din aceeaşi 
mulţime. Mai departe în vectorul X se include primul element a21 din mulţimea A2, 
element care satisface condiţiile de continuare, şi se trece la examinarea elementelor 
din mulţimea A3.

Întrucît niciunul din elementele a31 , a32 , a33 nu satisface condiţiile de continuare, 
se revine la mulţimea A2 din care se alege elementul al doilea, şi anume, a22. După 
aceasta se testează din nou elementele mulţimii A3, elementul a32 satisfăcînd condiţi-
ile de continuare.

Schema generală a unui algoritm recursiv bazat pe metoda reluării este redată cu 
ajutorul procedurii ce urmează:

procedure Reluare(k:integer);
begin
  if k<=n then 

    begin
      X[k]:=PrimulElement(k);
      if Continuare(k) then Reluare(k+1);
      while ExistaSuccesor(k) do
        begin
          X[k]:=Succesor(k);
          if Continuare(k) then Reluare(k+1)
        end; { while }
    end { then }
  else PrelucrareaSolutiei;
end; {Reluare}

Procedura Reluare comunică cu programul apelant şi subprogramele ape-
late prin variabilele globale ce reprezintă vectorul X şi mulţimile A1, A2, ..., An. 
Subprogramele apelate execută următoarele operaţii:

PrimulElement(k)  returnează primul element din mulţimea Ak;
Continuare(k)  returnează valoarea true dacă elementele înscrise în primele 

k componente ale vectorului X satisfac condiţiile de continuare şi false în caz con-
trar;

ExistaSuccesor(k)  returnează valoarea true dacă elementul memorat în 
componenta xk are un succesor în mulţimea Ak şi false în caz contrar;

Succesor(k)  returnează succesorul elementului memorat în componenta xk;
PrelucrareaSolutiei  de obicei, în această procedură soluţia reţinută în 

vectorul X este afi şată la ecran.
Pentru o înţelegere mai clară a procesului recursiv, vom urmări execuţia procedu-

rii Reluare în cazul mulţimilor A1, A2, şi A3 din fi gura 5.3.
La execuţia apelului Reluare(1) în componenta x1 a vectorului X se înscrie pri-

mul element al mulţimii A1:
X=(a11).

Întrucît în cazul acestui vector apelul Continuare(1) returnează valoarea fal-
se, se trece la execuţia instrucţiunii while. Apelul ExistaSuccesor(1) returnea-
ză valoarea true şi, prin urmare, în componenta x1 a vectorului X se înscrie succe-
sorul elementului a11:

X=(a12).
De data aceasta apelul Continuare(1) returnează valoarea true şi, prin urma-

re, se trece la execuţia recursivă a apelului Reluare(2). Evident, în componenta x2 
a vectorului X va fi  înscris primul element din mulţimea A2:

X=(a12, a21).
Pentru acest vector funcţia Continuare returnează valoarea true, fapt ce de-

clanşează execuţia recursivă a apelului Reluare(3).
În cazul apelului Reluare(3) componenta x3 a vectorului X ia recursiv valorile 

a31, a32 şi a33:
X=(a12, a21, a31);
X=(a12, a21, a32);

X=(a12, a21, a33),
însă nici unul din aceşti vectori nu satisface condiţiile de continuare. După exami-
narea ultimului element din mulţimea A3, funcţia ExistaSuccesor returnează va-
loarea false şi, în consecinţă, se revine în contextul apelului Reluare(2). În acest 
context în componenta x2 a vectorului X se înscrie succesorul elementului a21:

X=(a12, a22).
Întrucît şi pentru acest vector funcţia Continuare returnează valoarea true, 

din nou se execută apelul recursiv Reluare(3). Însă, spre deosebire de apelul pre-
cedent, în acest caz vectorul

X=(a12, a22, a32)
satisface condiţiile de continuare, fapt ce declanşează execuţia apelului recursiv 
Reluare(4). În acest apel k>n şi, prin urmare, procedura PrelucrareaSolutiei 
va afi şa coordonatele vectorului X la ecran.

Exemplu. Se consideră mulţimile A1, A2, ..., An, fi ecare mulţime fi ind formată din 
mk numere naturale. Selectaţi din fi ecare mulţime cîte un număr în aşa mod încît 
suma lor să fi e egală cu q.

Rezolvare. Vom reprezenta mulţimile A1, A2, ..., An prin liniile tabloului bidimensi-
onal (matricei) A = ||akj||. Numărul de elemente mk al fi ecărei mulţimi Ak va fi  reţinut 
în componenta respectivă a tabloului unidimensional (vectorului) M = ||mk||.

Din enunţul problemei rezultă că toate condiţiile de continuare sînt respectate 
numai atunci cînd suma elementelor referite de primele k componente ale vectorului 
X nu depăşeşte valoarea q pentru k<n şi este egală cu q pentru k = n. Pentru a simpli-
fi ca scrierea programului, vom memora în vectorul X numai indicii j ai elementelor 
aij selectate din mulţimile A1, A2, ..., An.

Program P154;
 { Program bazat pe metoda reluării }
const mmax=50; { numărul maximal de mulţimi }
      nmax=50; { numărul maximal de elemente }
type Natural = 0..MaxInt;
     Multime = array [1..nmax] of Natural;

var  A : array[1..nmax] of Multime;
     n : 1..nmax;                   { numărul de mulţimi }
     M : array[1..nmax] of 1..mmax; { cardinalul mulţimii S[k] }

     X : array[1..nmax] of 1..mmax; { indicii elementelor 
selectate }

     q : Natural;
     k, j : integer;
     Indicator : boolean;

function PrimulElement(k : integer) : Natural;
begin
  PrimulElement:=1;
end; {PrimulElement }

function Continuare(k : integer) : boolean;
var j : integer;
    suma : Natural;
begin
  suma:=0;
  for j:=1 to k do suma:=suma+A[j, X[j]];
  if ((k<n) and (suma<q)) or ((k=n) and (suma=q))
    then Continuare:=true
    else Continuare:=false;
end; { Continuare }

function ExistaSuccesor(k : integer) : boolean;
begin
  ExistaSuccesor:=(X[k]<M[k]);
end; { ExistaSuccesor }

function Succesor(k : integer) : integer;
begin
  Succesor:=X[k]+1;
end; { Succesor }

procedure PrelucrareaSolutiei;
var k : integer;
begin
  write(’Soluţia: ’);
  for k:=1 to n do write(A[k, X[k]], ’ ’);
  writeln;
  Indicator:=true;
end; { PrelucrareaSolutiei }

procedure Reluare(k : integer);
 { Metoda reluarii - varianta recursiva }
begin
  if k<=n then
       begin
         X[k]:=PrimulElement(k);
         if Continuare(k) then Reluare(k+1);
         while ExistaSuccesor(k) do
           begin
             X[k]:=Succesor(k);
             if Continuare(k) then Reluare(k+1);
           end { while }
       end { then }
     else PrelucrareaSolutiei;
end; { Reluare }

    begin
      write(’Daţi cardinalul A[’, k, ’]=’); readln(M[k]);
      write(’Daţi elementele mulţimii A[’, k, ’]: ’);
      for j:=1 to M[k] do read(A[k, j]);
      writeln;
    end;
  Write(’Daţi q=’); readln(q);
  Indicator:=false;
  Reluare(1);
  if Indicator=false then writeln(’Nu există soluţii’);
  readln;
end.

Din analiza procedurii Reluare şi a ordinii de parcurgere a elementelor din mul-
ţimile A1, A2, ..., An (fi g. 5.4) rezultă că în cazul cel mai nefavorabil vor fi  generaţi toţi 
vectorii X ai produsului cartezian A1A2...An. Prin urmare, complexitatea tempo-
rală a algoritmilor bazaţi pe metoda reluării este, ca şi în cazul trierii, O(mn), unde 
m=max(m1, m2, ..., mn).

Timpul real de execuţie, cerut de un algoritm, depinde în mare măsură de natura 
problemei, mai exact, de esenţa condiţiilor de continuare şi ordinea în care apar ele-
mentele akj în mulţimile A1, A2, ..., An. O bună alegere a condiţiilor de continuare are 
ca efect o importantă reducere a numărului de calcule. În particular, în cazul cel mai 
favorabil, din fi ecare mulţime A1, A2, ..., An va fi  testat numai cîte un singur element 
şi, evident, timpul de execuţie va fi  proporţional cu n.

Cu regret, în prezent nu există o regulă universală care ar permite formularea 
unor condiţii „bune” de continuare pentru orice problemă întîlnită în practică. 
Identifi carea acestor condiţii şi, în consecinţă, reducerea timpului de calcul depinde 
de iscusinţa programatorului. Menţionăm că în majoritatea problemelor mulţimile 
A1, A2, ..., An nu sînt cunoscute apriori, iar componentele vectorului X pot aparţine şi 
ele unor tipuri structurate de date  tablouri, mulţimi sau articole.

Întrebări şi exerciţii
Explicaţi structura generală a algoritmilor bazaţi pe metoda reluării.
Indicaţi cazurile elementare și cele neelementare în procedura recursivă Reluare.
Elaboraţi o variantă iterativă a procedurii Reluare.
Elaboraţi un studiu comparativ al algoritmilor iterativi și algoritmilor recursivi bazaţi pe 

metoda reluării.
Reprezentaţi pe un desen similar celui din fi gura 5.4 ordinea în care sînt examinate ele-

mentele mulţimilor A1, A2, ..., An din programul P154:
a) A1={1},  A2={2},  A3={3},   q=4;
b) A1={1},  A2 ={2, 3},  A3={4, 5},   q=10; 
c) A1={1, 2, 3},  A2={4, 5},  A3={6},   q=14;
d) A1={1, 2},  A2={3, 4},  A3={5, 6},  A4={7, 8},   q=20.

Precizaţi condiţiile de continuare pentru programele în care se dorește generarea tutu-
ror elementelor produsului cartezian A1A2...An. Pentru a verifi ca răspunsul scrieţi și 
depanaţi programul respectiv.

Indicaţi pe desenul din fi gura 5.4 ordinea de parcurgere a elementelor din mulţimile A1, 
A2, A3 în cazurile cele mai favorabile și cele mai nefavorabile.

Se consideră mulţimea B={b1, b2, ..., bn} formată în primele n litere ale alfabetului latin. 
Elaboraţi un program bazat pe metoda reluării care generează toate submulţimile 
Bi, BiB, formate exact din q litere.

 Indicaţie. Fiecare submulţime Bi poate fi  reprezentată prin vectorul caracteristic X = ||xk||n, 
unde

 

dacă
în caz contrar.

 Evident, submulţimea Bi satisface condiţiile problemei dacă x1+x2+...+xn=q.
Colorarea hărţilor. Pe o hartă sunt reprezentate n ţări, n30. În memoria calculatorului 

harta este redată prin matricea A=||aij||nm, unde

dacă ţările i, j sînt vecine;
în caz contrar.

 Determinaţi numărul minim de culori m necesare pentru a colora harta. Evident, se cere 
ca oricare două ţări vecine să fi e colorate diferit.

Labirintul. Se consideră planul unui labirint desenat pe o foaie de hîrtie liniată în pătră-
ţele (fi g. 5.5). Pătrăţelele hașurate reprezintă obstacolele, iar cele nehașurate  camerele 
și coridoarele labirintului.

Fig. 5.5. Planul unui labirint

 În memoria calculatorului planul labirintului este redat prin matricea A=||aij||nm, 
1n,m30, unde

dacă pătrăţelul (i, j) este haşurat;
în caz contrar.

 Călătorul se poate deplasa dintr-un pătrăţel nehașurat în alt pătrăţel nehașurat numai 
atunci cînd ele au o latură comună. Elaboraţi un program, care găsește, dacă există, un 
drum din pătrăţelul B în pătrăţelul C.

 Se consideră n (n30) săculeţe, numerotate prin 1, 2, 3, ..., n. Săculeţul i conţine mi mo-
nede de aceeași valoare Vi. Elaboraţi un program care afi șează la ecran modul de plată a 
unei sume S cu exact p monede din săculeţe.

 Elaboraţi un program care afi șează la ecran toate modurile de a descompune un număr 
natural în sumă de k numere naturale distincte. De exemplu, pentru n=9 și k=3 soluţiile 
sînt: 1+2+6, 2+3+4, 1+3+5.

 Efectuaţi un studiu comparativ al algoritmilor bazaţi pe metoda trierii și algoritmilor 
bazaţi pe metoda reluării în cazul problemelor din exerciţiile 8 și 12.

5.5. Metoda desparte şi stăpîneşte
Metoda desparte şi stăpîneşte (în latină divide et impera) este o metodă generală de 

elaborare a algoritmilor, care presupune:
1) împărţirea repetată a unei probleme de dimensiuni mari în două sau mai multe 

subprobleme de acelaşi tip, dar de dimensiuni mai mici;
2) rezolvarea subproblemelor în mod direct, dacă dimensiunea lor permite aceas-

ta, sau împărţirea lor în alte subprobleme de dimensiuni şi mai mici;
3) combinarea soluţiilor subproblemelor rezolvate pentru a obţine soluţia proble-

mei iniţiale.
În limbaj matematic, admitem că la un anumit pas al algoritmului se dă o mulţi-

me ordonată
A = (ai, ai+1, ..., aj)

şi că trebuie efectuată o prelucrare oarecare asupra elementelor sale. Pentru a îm-
părţi problema curentă în două subprobleme de aproximativ aceleaşi dimensiuni, 
stabilim

m = (j – i) div 2
şi împărţim mulţimea A în două submulţimi care vor fi  prelucrate separat:

A1 = (ai, ai+1, ..., ai+m);  A2 = (ai+m+1, ai+m+2, ..., aj).
În continuare, mulţimile A1 şi A2 se împart din nou în cîte două submulţimi, res-

pectiv A1-1, A1-2 şi A2-1, A2-2. Acest proces va continua pînă cînd soluţiile ce corespund 
submulţimilor curente vor putea fi  calculate în mod direct.

Pentru exemplifi care, în fi gura 5.6 este prezentat modul de împărţire a mulţimii 
A=(a1, a2, ..., a7) în cazul divizării problemelor curente în cîte două subprobleme de 
acelaşi tip.

Schema generală a unui algoritm bazat pe metoda desparte şi stăpîneşte poate fi  
redată cu ajutorul unei proceduri recursive:

procedure DesparteSiStapineste(i, j : integer; var x : tip);
var m : integer;
    x1, x2 : tip;
begin
   if SolutieDirecta(i, j) then Prelucrare(i, j, x)

      else
         begin
           m:=(j-i) div 2;
           DesparteSiStapineste(i, i+m, x1);
           DesparteSiStapineste(i+m+1, j, x2);
           Combina(x1, x2, x);
         end;
end;

A=(a1, a2, a3, a4, a5, a6, a7)

A1=(a1, a2, a3, a4) A2=(a5, a6, a7)

A1-1=(a1, a2) A1-2=(a3, a4) A2-1=(a5, a6) A2-2=(a7)

Fig. 5.6. Descompunerea mulţimii A în submulţimi

Parametrii i şi j din antetul procedurii defi nesc submulţimea curentă (ai, ai+1, 
..., aj) supusă prelucrării. Funcţia SolutieDirecta returnează valoarea true dacă 
subproblema curentă admite o rezolvare directă şi false în caz contrar. Procedura 
Prelucrare returnează prin parametrul x soluţia subproblemei curente, calculată 
în mod direct. Dacă calculul direct este imposibil, se execută două apeluri recursive 
– unul pentru submulţimea (ai, ai+1, ..., am) şi altul pentru submulţimea (ai+m+1, ai+m+2, ..., 
aj). Procedura Combina prelucrează soluţiile x1 şi x2 ale subproblemelor respective 
şi returnează soluţia x a problemei curente.

Exemplul 1. Se consideră mulţimea A={a1, a2, ..., an} formată din n numere reale. 
Elaboraţi un program care determină numărul maximal din această mulţime.

Rezolvare. În programul ce urmează mulţimea A este reprezentată printr-un ta-
blou unidimensional cu n componente. Se consideră că soluţia unei subprobleme 
poate fi  calculată direct numai atunci cînd mulţimea (ai, ..., aj) este formată din unul 
sau două numere. Evident, în astfel de cazuri x = ai sau x = max (ai, aj).

Program P155;
 { Gasirea elementuli maximal prin metoda desparte şi 
stăpîneşte }
const nmax=100;

var A : array[1..nmax] of real;
 i, n : 1..nmax;
    x : real;

function SolutieDirecta(i, j : integer) : boolean;
begin
  SolutieDirecta:=false;
  if (j-i<2) then SolutieDirecta:=true;
end; { SolutieDirecta }

procedure Prelucrare(i, j : integer; var x : real);
begin
  x:=A[i];
  if A[i]<A[j] then x:=A[j];
end; { Prelucrare }

procedure Combina(x1, x2 : real; var x : real);
begin
  x:=x1;
  if x1<x2 then x:=x2;
end; { Combina }

procedure DesparteSiStapineste(i, j : integer; var x : real);
var m : integer;
    x1, x2 : real;
begin
  if SolutieDirecta(i, j) then Prelucrare(i, j, x)
     else
       begin
         m:=(j-i) div 2;
         DesparteSiStapineste(i, i+m, x1);
         DesparteSiStapineste(i+m+1, j, x2);
         Combina(x1, x2, x);
       end;
end; { DesparteSiStapineste }

begin
  write(’Daţi n=’); readln(n);
  writeln(’Daţi ’, n, ’ numere reale’);
  for i:=1 to n do read(A[i]);
  writeln;
  DesparteSiStapineste(1, n, x);
  writeln(’Numărul maximal x=’, x);
  readln;
  readln;
end.

În programul P155 la fi ecare apel al procedurii DesparteSiStapineste pro-
blema curentă este rezolvată direct sau împărţită în două subprobleme de dimensi-
uni aproximativ egale. Evident, metoda desparte şi stăpîneşte admite împărţirea pro-
blemelor curente într-un număr arbitrar de subprobleme, nu neapărat în două. În 

exemplul ce urmează problema curentă  tăierea unei placi de arie maximă  este 
împărţită în patru subprobleme de acelaşi tip, dar de dimensiuni mai mici.

Exemplul 2. Se consideră o placă dreptunghiulară de dimensiunile LH. Placa are 
n găuri punctiforme, fi ecare gaură fi ind defi nită prin coordonatele (xi, yi). Elaboraţi 
un program care decupează din placă o bucată de arie maximă, dreptunghiulară şi 
fără găuri. Sînt admise doar tăieturi de la o margine la alta pe direcţii paralele cu 
laturile plăcii  verticale sau orizontale (fi g. 5.7).

a)

b) c)

Fig. 5.7. Tăierea unei plăci de arie maximă:
a – placa iniţială; b – tăierea pe verticală; c – tăierea pe orizontală

Rezolvare. Vom defi ni placa curentă prin vectorul P=(a, b, c, d), unde a şi b sînt co-
ordonatele colţului stînga-jos, iar c şi d  coordonatele colţului dreapta-sus. Evident, 
placa iniţială se defi neşte prin (0, 0, L, H). Metoda desparte şi stăpîneşte poate fi  reali-
zată după cum urmează:

 iniţial stabilim aria maximă Smax=0;
 dacă placa curentă nu are găuri, problema poate fi  soluţionată direct, compa-

rînd aria curentă cu valoarea Smax;
 în caz contrar alegem o gaură arbitrară (xi, yi) prin care tăiem placa curentă în 

plăci mai mici, arătate în fi gura 5.7:
P1=(a, b, xi, d), P2=( xi, b, c, d)   sau   P3=(a, yi, c, d), P4=(a, b, c, yi );

 în continuare examinăm în acelaşi mod fi ecare din plăcile obţinute în urma tăie-
rii, memorînd consecutiv în variabila Smax aria plăcii de suprafaţă maximă.

Program P156;
 { Tăierea unei placi prin metoda desparte şi stăpîneşte }
const nmax=100;

var L, H : real;
    n : 1..nmax;
    X,Y : array[1..nmax] of real;
    Smax, amax, bmax, cmax, dmax : real;
    i : integer;

function SolutieDirecta(a, b, c, d : real;
                        var i : integer) : boolean;
label 1;
var j : integer;
begin
  SolutieDirecta:=true;
  for j:=1 to n do
    if (X[j]>a) and (X[j]<c) and (Y[j]>b) and (Y[j]<d) then
      begin
        SolutieDirecta:=false;
        i:=j;
        goto 1;
      end;
1:end; { SolutieDirecta }

procedure PrelucrareaSolutiei(a, b, c, d : real);
var S : real;
begin
  S:=(c-a)*(d-b);
  if S>=Smax then
    begin
      Smax:=S;
      amax:=a; bmax:=b; cmax:=c; dmax:=d;
    end;
end; { PrelucrareaSolutiei }

procedure DesparteSiStapineste(a, b, c, d : real);
var i : integer;
begin
  writeln(’Examinăm placa (’, a:5:1,’ ’, b:5:1, ’ ’,
                              c:5:1, ’ ’, d:5:1, ’)’);
  readln;
 if SolutieDirecta(a, b, c, d, i)
     then PrelucrareaSolutiei(a, b, c, d)

     else begin
            DesparteSiStapineste(a, b, X[i], d);
            DesparteSiStapineste(X[i], b, c, d);
            DesparteSiStapineste(a, Y[i], c, d);
            DesparteSiStapineste(a, b, c, Y[i]);
          end;
end; { DesparteSiStapineste }

begin
  writeln(’Daţi dimensiunile L, H’); readln(L, H);
  write(’Daţi n=’); readln(n);
  writeln(’Daţi coordonatele X[i], Y[i]’);
  for i:=1 to n do read(X[i], Y[i]);
  writeln;
  Smax:=0;
  DesparteSiStapineste(0, 0, L, H);
  writeln(’Placa de arie maximă (’,
           amax:5:1, ’ ’, bmax:5:1, ’ ’,
           cmax:5:1, ’ ’, dmax:5:1, ’)’);
  writeln(’Smax=’, Smax:5:2);
  readln;
end.

Funcţia SolutieDirecta din programul P156 returnează valoarea true dacă 
placa (a, b, c d) nu are găuri şi false în caz contrar. În cazul valorii false, funcţia 
returnează suplimentar numărul de ordine i al uneia din găurile plăcii. Această gau-
ră este depistată verifi cînd relaţiile a<xi<c şi b<yi<d.

Procedura PrelucrareaSolutiei compară aria plăcii curente S=(c-a)(d-b) cu valoa-
rea Smax. Dacă SSmax, procedura memorează placa curentă în vectorul (amax, bmax, cmax, dmax).

Complexitatea temporală a algoritmilor bazaţi pe metoda desparte şi stăpîneşte 
depinde de numărul de subprobleme k în care este divizată problema curentă şi de 
complexitatea algoritmilor destinaţi rezolvării directe a subproblemelor respective. 
S-a demonstrat că în majoritatea cazurilor timpul cerut de un algoritm bazat pe me-
toda desparte şi stăpîneşte este de ordinul n log2 n sau n2 log2 n, deci polinomial.

Programele elaborate în baza metodei desparte şi stăpîneşte sînt simple, iar timpul de 
execuţie este relativ mic. Cu regret, această metodă nu este universală, întrucît ea poate 
fi  aplicată numai atunci cînd prelucrarea cerută admite divizarea problemei curente în 
subprobleme de dimensiuni mai mici. De obicei, această proprietate nu este indicată ex-
plicit în enunţul problemei şi găsirea ei, dacă există, cade în sarcina programatorului.

Întrebări şi exerciţii
Explicaţi schema de calcul a algoritmilor bazaţi pe metoda desparte și stăpînește.
Care sînt avantajele și neajunsurile metodei desparte și stăpînește?
Utilizînd metoda desparte și stăpînește, elaboraţi un program care determină suma ele-

mentelor mulţimii A={a1, a2, ..., an} formată din n numere reale.

Indicaţi pe un  desen similar celui din fi gura 5.7 ordinea examinării plăcilor curente pe 
parcursul executării programului P156:

  dimensiunea plăcii iniţiale: 3  4;
  numărul de găuri ale plăcii iniţiale: 3;
  coordonatele găurilor: (1, 1); (1, 2); (2, 2).
Estimaţi necesarul de memorie și complexitatea temporală a programelor P155 și P156.
Schiţaţi un algoritm care rezolvă problema tăierii unei plăci de arie maximă prin metoda 

trierii. Estimaţi complexitatea temporală și necesarul de memorie al algoritmului elabo-
rat. Efectuaţi un studiu comparativ al algoritmilor bazaţi pe metoda trierii și algoritmilor 
bazaţi pe metoda desparte și stăpînește.

Căutarea binară. Se consideră mulţimea A={a1, a2, ..., an}, elementele căreia sînt numere 
întregi sortate în ordine crescătoare. Elaboraţi un program care determină dacă mulţimea 
A conţine numărul dat p. Estimaţi complexitatea temporală a programului elaborat.

Utilizînd metoda desparte și stăpînește, elaboraţi un program care determină cel mai 
mare divizor comun al numerelor naturale a1, a2, ..., an.

Sortarea prin interclasare. Elaboraţi un program care sortează elementele șirului (a1, 
a2, ..., an) în ordine crescătoare după cum urmează:

 – divizăm șirul curent în două subșiruri de aproximativ aceeași lungime;
 – dacă subșirul conţine numai două elemente, aranjăm elementele respective în ordine 

crescătoare;
 – avînd două subșiruri sortate, le interclasăm pentru a obţine șirul curent sortat.
 Se consideră că elementele șirului care trebuie sortat sînt numere întregi. De exemplu, în urma 

interclasării subșirurilor sortate (-3, 4, 18) și (-2, -1, 15) obţinem șirul ( -3, -2, -1, 4, 15, 18).
Problema turnurilor din Hanoi.* Se consideră trei tije notate prin 1, 2 și 3 și n discuri 

perforate de dimensiuni diferite (fi g. 5.8). Iniţial toate discurile sînt pe tija 1, aranjate în 
ordinea descrescătoare a diametrelor, considerînd sensul de la bază la vîrf. Elaboraţi un 
program care mută toate discurile pe tija 2 folosind tija 3 ca tijă de manevră și respectînd 
următoarele reguli:

  la fi ecare pas se mută un singur disc;
  orice disc poate fi  așezat doar peste un disc cu diametrul mai mare.

Fig. 5.8. Problema turnurilor din Hanoi

* Denumirea problemei provine de la o veche legendă hindusă conform căreia după mutarea 
celor 64 de discuri va veni sfîrșitul lumii.

 Indicaţii. Mutarea unui disc de pe tija i pe tija j poate fi  reprezentată ca o pereche (i, j) cu 
i, j{1, 2, 3}, ij. Prin H(m, i, j) vom nota șirul mutărilor necesare pentru a muta primele m 
discuri (evident, cele situate cel mai sus) de pe tija i pe tija j. De exemplu,

H(1, 1, 2)=(1, 2);
H(2, 1, 2)=(1, 3),  (1, 2),  (3, 2);
H(3, 1, 2)=(1, 2),  (1, 3),  (2, 3),  (1, 2),  (3, 1),  (3, 2),  (1, 2).

 În general,
pentru

pentru

 unde k=6–i–j. Prin urmare, problema celor n discuri se reduce la rezolvarea a două sub-
probleme de același tip pentru (n–1) discuri.

5.6. Programarea dinamică
Programarea dinamică reprezintă o metodă de rezolvare a problemelor soluţia 

cărora poate fi  privită ca rezultatul unui şir de decizii (d1, d2, ..., dp, ..., dq). Fiecare 
decizie dp trebuie să furnizeze o soluţie locală care optimizează un criteriu global 
de calitate, de exemplu, costul unei călătorii, lungimea drumului parcurs, greutatea 
transportată, spaţiul ocupat de un fi şier pe disc etc. Pentru aplicarea acestei metode, 
este necesar ca problema de rezolvat să satisfacă principiul optimalităţii: dacă (d1, 
d2, ..., dp, dp+1, ..., dq) este un şir optim de decizii, atunci şirurile (d1, d2, ..., dp) şi (dp+1, ..., 
dq) trebuie să fi e optimale.

De exemplu, dacă drumul cel mai scurt dintre Chişinău şi Bălţi trece prin Orhei, 
atunci ambele porţiuni din acest drum – Chişinău–Orhei şi Orhei–Bălţi – de asemenea 
vor fi  cele mai scurte. Prin urmare, problemele în care se cere determinarea celui mai 
scurt drum satisfac principiul optimalităţii.

În PASCAL metoda programării dinamice poate fi  realizată utilizînd tablouri 
componentele cărora se calculează cu ajutorul unor relaţii de recurenţă. În general, 
relaţiile de recurenţă sînt de următoarele două tipuri:

1) Fiecare decizie dp depinde de deciziile dp+1, ..., dq. Spunem în acest caz că se apli-
că metoda înainte. Deciziile vor fi  luate în ordinea dq, dq-1, ..., d1.

2) Fiecare decizie dp depinde de deciziile d1, ..., dp-1. În acest caz spunem că se aplică 
metoda înapoi. Deciziile vor fi  luate în ordinea d1, d2, ..., dq.

Evident, pentru fi ecare problemă propusă, programatorul va verifi ca mai întîi 
respectarea principiului optimalităţii şi în cazul unui răspuns pozitiv va deduce re-
laţiile respective de recurenţă. În caz contrar, problema nu poate fi  rezolvată prin 
metoda programării dinamice.

Exemplu. Planul unui teren aurifer de formă dreptunghiulară cu dimensiunile 
nm este format din zone pătrate cu latura 1 (fi g. 5.9). În zona din colţul nord-vest se 
afl ă un robot. Din zona cu coordonatele (i, j), 1 ≤ i ≤ n, 1 ≤ j ≤ m, robotul poate extrage 
cel mult aij grame de aur. Din considerente tehnologice pe teren există restricţii de 
deplasare: la fi ecare pas robotul se poate mişca din zona curentă numai în una din 



141

zonele vecine – cea din est sau cea din sud. Elaboraţi un program care determină 
cantitatea maximă de aur Cmax care poate fi  extrasă de robot.

Robot

Fig. 5.9. Zonele unui teren aurifer

Rezolvare. În calculator informaţia despre un teren aurifer poate fi  reprezentată cu 
ajutorul tabloului bidimensional A = ||aij||nm, unde aij reprezintă cantitatea de aur ce 
poate fi  extrasă din zona cu coordonatele (i, j). Pentru exemplifi care, în continuare 
vom examina un teren aurifer descris cu ajutorul tabloului ce urmează:

 1  2 3 4 5
1  1 2 3 4 5

2 3 4 6 4 6

3 6 2 7 7 5

4 7 2 3 4 5

A =

Pentru a sistematiza calculele, introducem în studiu tabloul bidimensional C = 
=||cij||nm, unde componenta cij reprezintă cantitatea maximă de aur pe care o poate 
extrage robotul deplasîndu-se din zona iniţială (1,1) în zona cu coordonatele (i, j). 
Evident, Cmax = cnm.

Conform condiţiilor problemei, în oricare zonă (i, j) se poate intra numai prin una 
din zonele vecine: prin cea din vest cu coordonatele (i, j-1) sau prin cea din nord cu 
coordonatele (i-1, j). Prin urmare, cantitatea maximă de aur pe care o poate extrage 
robotul ajungînd în zona (i, j) este dată de formula recurentă

cij = aij + max(ci,j–1, ci–1,j).
Ordinea de calcul a componentelor cij ale tabloului C este impusă de modul de 

deplasare a robotului, şi anume:
pasul 1:  c11;
pasul 2:  c21, c12;
pasul 3:  c31, c22, c13;
...
pasul n+m-1:  cnm.

Se observă că la pasul k vor fi  calculate numai acele elemente cij ale tabloului C 
pentru care se respectă egalitatea i + j – 1 = k. Ordinea în care sînt calculate elemen-
tele cij în cazul tabloului A de mai sus este prezentată în fi gura 5.10.

k = 1 k = 2 k = 3 k = 4

k = 5 k = 6 k = 8 k = 8

Fig. 5.10. Calcularea elementelor tabloului C

Pentru a evita verifi cările legate de calcularea indicilor i şi j, în programul ce ur-
mează elementele cij din prima linie şi din prima coloană sînt calculate separat de 
celelalte componente ale tabloului C.

Program P157;
 { Deplasarea robotului pe un teren aurifer }
var A , C : array [1..50, 1..50] of real;
    m, n, i, j, k : integer;

function Max(a, b : real) : real;
begin
  if a>b then Max:=a else Max:=b;
end; { Max }

begin
  write(’Daţi valorile n, m:  ’); readln(n, m);
  writeln(’Daţi componentele tabloului A’);
  for i:=1 to n do
    for j:=1 to m do read(A[i,j]);
  writeln;

  C[1,1]:=A[1,1];
  for i:=2 to n do C[i,1]:=A[i,1]+C[i-1,1];
  for j:=2 to m do C[1,j]:=A[1,j]+C[1,j-1];

  for k:=2 to n+m-1 do
    for i:=2 to n do
      for j:=2 to m do
        if (i+j-1)=k then
          C[i,j]:=A[i,j]+Max(C[i,j-1], C[i-1,j]);
  writeln(’Cmax=’, C[n,m]);
  readln;
end.

Complexitatea temporală a algoritmilor bazaţi pe metoda programării dinamice 
depinde de natura relaţiilor de recurenţă şi în majoritatea cazurilor este polinomială. 
De exemplu, programul P157 are complexitatea O(n3).

Întrebări şi exerciţii
Explicaţi esenţa principiului optimalităţii.
În ce ordine pot fi  luate deciziile în procesul soluţionării unei probleme prin metoda 

programării dinamice?
Explicaţi ordinea de luare a deciziilor în problema deplasării robotului pe un teren aurifer.
Demonstraţi că problema deplasării robotului pe un teren aurifer satisface principiul 

optimalităţii.
Estimaţi necesarul de memorie și timpul de execuţie al programului P157.
Cum credeţi, care sînt avantajele și neajunsurile algoritmilor bazaţi pe metoda progra-

mării dinamice?
Problema discretă a rucsacului. Se consideră n obiecte. Pentru fi ecare obiect i (i=1, 

2, ..., n) se cunoaște greutatea gi și cîștigul ci care se obţine în urma transportului său 
la destinaţie. O persoană are un rucsac cu care pot fi  transportate unul sau mai multe 
obiecte greutatea sumară a cărora nu depășește valoarea Gmax. Elaboraţi un program 
care determină ce obiecte trebuie să transporte persoana în așa fel încît cîștigul să fi e 
maxim. Obiectele respective nu pot fi  tăiate în fragmente mai mici.

Drumul de cost minim. Se consideră o reţea formată din n orașe. Între unele orașe 
există curse directe de autobuz. Costul unui bilet la o cursă directă din orașul i în orașul 
j este de cij lei. Evident, costul unei curse directe întotdeauna este mai mic decît costul 
unei călătorii cu transbordări: cij < cik + ckj. Elaboraţi un program care determină costul 
minim al unei călătorii din orașul i în orașul j.

Arhivarea fi șierelor. E cunoscut faptul că pe suporturile de memorie externă orice fi șier 
este memorat ca o secvenţă de cifre binare. Pentru a economisi spaţiul de memorare, 
programele de arhivare descompun fi șierul iniţial într-o succesiune de cuvinte binare ce 
aparţin unui dicţionar și înscriu pe disc numai numerele de ordine ale cuvintelor respec-
tive. De exemplu, în cazul unui dicţionar format din 5 cuvinte binare:

 1) 0;
 2) 1;
 3) 000;
 4) 110;
 5) 1101101,

 fi șierul iniţial 10001101101110 va fi  memorat pe disc în forma 2354. Elaboraţi un 
program care descompune orice secvenţă binară într-un număr minim de cuvinte apar-
ţinînd dicţionarului propus.

Triangularea polinoamelor. Procesarea imaginilor cu ajutorul calculatoarelor presu-
pune descompunerea poligoanelor în fi guri geometrice mai simple, și anume, în triun-
ghiuri. Admitem că poligonul convex P1P2...Pn este defi nit prin coordonatele (xi, yi) ale 
vîrfurilor Pi, i=1, 2, ..., n. Elaboraţi un program care descompune poligonul P1P2...Pn  în 
triunghiuri trasînd în acest scop un set de coarde PjPm, jm,  j, m{1, 2, ..., n}, care nu se 
intersectează. Se cere ca lungimea totală a coardelor respective să fi e minimă.

5.7. Metoda ramifi că şi mărgineşte
În metoda ramifi că şi mărgineşte (în engleză branch and bound) căutarea soluţiei 

unei probleme se efectuează prin parcurgerea unui arbore de ordinul m, denumit 
arborele soluţiilor. Nodurile acestui arbore reprezintă mulţimea stărilor 
posibile S = {s1, s2, ..., sn} în care se poate afl a un sistem, de exemplu, confi guraţiile 
pieselor pe o tablă de şah, poziţia unui automobil pe harta drumurilor naţionale, 
poziţiile capului de scriere–citire ale unei unităţi de disc magnetic etc. Legăturile sisj 
de tipul tată–fi u indică faptul că din starea si sistemul poate trece direct în starea sj în 
urma unor transformări sau operaţii relativ simple (fi g. 5.11). Pentru exemplifi care 
amintim deplasarea pieselor făcută alternativ de adversari în cursul unei partide 
de şah, trecerea automobilului dintr-o localitate în alta, executarea unei comenzi de 
scriere–citire etc. Evident, în astfel de cazuri soluţia problemei poate fi  reprezentată 
ca un drum ce leagă nodul rădăcină cu unul din nodurile terminale şi care optimi-
zează un criteriu de calitate, de exemplu, numărul de mutări într-o partidă de şah, 
distanţa parcursă de automobil, timpul de scriere–citire a unui fi şier etc.

Formal, astfel de probleme pot fi  rezolvate prin metoda trierii, efectuînd în acest 
scop următoarele operaţii:

– parcurgem arborele soluţiilor în adîncime sau în lăţime şi găsim stările fi nale 
ale sistemului;

– construim mulţimea tuturor drumurilor posibile ce leagă starea iniţială cu stă-
rile fi nale;

– din mulţimea drumurilor posibile îl alegem pe cel ce optimizează criteriul de 
calitate.

Întrucît parcurgerea completă a arborelui necesită foarte mult timp, pentru a 
micşora numărul de noduri vizitate, în metoda ramifi că şi mărgineşte ordinea de 
parcurgere este impusă de valorile unei funcţii f : SR, denumită funcţie de cost. 
Această funcţie este defi nită pe mulţimea de noduri ale arborelui soluţiilor, valoarea 
f(si) caracterizînd „cheltuielile” necesare pentru a trece din starea curentă si în una 
din stările fi nale ale subarborelui de rădăcină si.

Vizitarea nodurilor unui arbore în ordinea impusă de valorile funcţiei de cost se 
numeşte parcurgere de cost minim. Această parcurgere se realizează memorînd no-
durile active într-o listă din care la fi ecare pas se extrage nodul de cost minim. Pentru 
comparare amintim că la parcurgerea arborilor în lăţime sau în adîncime nodurile 
active se memorizează într-o coadă sau, respectiv, într-o stivă.

finală

Starea
finală

Fig. 5.11. Arborele soluţiilor

Fiind defi nită o funcţie de cost, metoda ramifi că şi mărgineşte presupune efectua-
rea următoarelor operaţii:

1. Etapa iniţială: se creează lista nodurilor active. La început această listă conţine 
un singur element – rădăcina arborelui soluţiilor.

2. Etapa ramifi că: din lista nodurilor active se extrage nodul de cost minim şi se 
generează toţi descendenţii acestuia.

3. Etapa mărgineşte: pentru fi ecare descendent se calculează valoarea funcţiei de 
cost. Descendenţii costul cărora este inaceptabil (de obicei, subarborii respectivi nu 
conţin stări fi nale) sînt excluşi din studiu. Descendenţii rămaşi se includ în lista no-
durilor active.

Punctele 2 şi 3 se repetă pînă cînd se ajunge la una din stările fi nale sau lista nodu-
rilor active devine vidă. Evident, în ultimul caz soluţia problemei nu a fost găsită.

Pentru exemplifi care, în tabelul 5.2 este reprezentată ordinea în care sînt vizitate 
nodurile arborelui din fi gura 5.11. Valorile funcţiei de cost sînt indicate direct pe de-
sen lîngă nodurile respective. Se consideră că descendenţii costul cărora depăşeşte 
valoarea 9 sînt inacceptabili.

În cazul arborelui din fi gura 5.11 parcurgerea de cost minim se termină cînd în lis-
ta nodurilor active apare starea fi nală s10. Întrucît orice drum ce leagă o stare fi nală de 
starea iniţială poate fi  construit urmînd legăturile de tipul fi u–tată, soluţia problemei 
pentru exemplul în studiu este (s1, s3, s5, s10).

Tabelul 5.2
Parcurgerea de cost minim

Nr. 
crt. Etapa Lista nodurilor 

active
Nodul de cost 

minim
Descendenţii nodului 

de cost minim
1. iniţială (s1) - -
2. ramifi că  s1 s2, s3, s4

3. mărgineşte (s3, s4) - -
4. ramifi că (s3) s4 s8, s9

5. mărgineşte (s3, s8) - -
6. ramifi că (s8) s3 s5, s6, s7

7. mărgineşte (s8, s5, s6, s7) - -
8. ramifi că (s8, s6, s7) s5 s10, s11

9. mărgineşte (s8, s6, s7, s10, s11) - -
10. stop (s8, s6, s7, s11) s10 -

Schema de calcul a metodei ramifi că şi mărgineşte poate fi  redată cu ajutorul urmă-
toarei secvenţe de instrucţiuni PASCAL:

IniţializareaListei;
repeat
  Ramifi ca;
  Margineste;
until Gasit or ListaEsteVida

Complexitatea algoritmilor bazaţi pe metoda ramifi că şi mărgineşte depinde de 
faptul cum este defi nită funcţia de cost f : SR. În cazul unor defi niţii nereuşite vor 
fi  vizitate toate nodurile arborelui soluţiilor, iar complexitatea temporală a algorit-
mului respectiv va fi  egală cu cea a algoritmilor bazaţi pe metoda trierii. Din contra, 
o alegere reuşită a funcţiei de cost permite excluderea din studiu a subarborilor care 
nu conţin stările fi nale dorite. În cursurile avansate de informatică se demonstrează 
că o defi niţie „bună” a funcţiei de cost poate fi  reprezentată în forma:

f(si) = niv(si) + h(si),
unde niv(si) este nivelul nodului si, iar funcţia h(si) estimează complexitatea transfor-
mărilor (operaţiilor) necesare pentru a ajunge din starea curentă si în una din stările 
fi nale ale subarborelui de rădăcină si.

De exemplu, în cazul unei partide de şah funcţia h(si) poate să exprime numărul 
de piese grele ale adversarului, în cazul unui automobil  distanţa pînă la punctul 
de destinaţie, iar în cazul unei unităţi de disc magnetic  timpul necesar pentru a citi 
sau a înscrie informaţia dorită.

Accentuăm faptul că aplicarea practică a metodei ramifi că şi mărgineşte este posi-
bilă numai atunci cînd valorile f(si) ale funcţiei de cost pot fi  calculate fără a construi 
subarborele de rădăcină si. Cu regret, în prezent nu sînt cunoscute reguli universale 
care ar asigura defi nirea univocă a funcţiei de cost pentru orice tip de problemă. Prin 



147

urmare, complexitatea algoritmilor bazaţi pe metoda ramifi că şi mărgineşte depinde 
în mare măsură de experienţa şi iscusinţa programatorului.

Întrebări şi exerciţii
Pentru care tip de probleme poate fi  aplicată metoda ramifi că și mărginește?
Cum se construiește arborele soluţiilor? Ce informaţie conţine fi ecare nod al arborelui?
Cum se reprezintă soluţia unei probleme în metoda ramifi că și mărginește?
Care este destinaţia funcţiei de cost ? Cum poate fi  defi nită această funcţie?
Explicaţi schema de calcul a metodei ramifi că și mărginește.
Scrieţi o procedură PASCAL care realizează parcurgerea de cost minim. Se consideră că 

fi ecare nod al arborelui conţine un cîmp în care este indicată valoarea funcţiei de cost. 
Estimaţi necesarul de memorie și complexitatea temporală a procedurii elaborate.

Indicaţi ordinea în care sînt vizitate nodurile arborelui din fi gura 5.11 în următoarele cazuri:
 a) parcurgerea în lăţime;
 b) parcurgerea în adîncime;
 c) parcurgerea de cost minim.
Sînt oare vizitate toate nodurile unui arbore în cazul parcurgerii de cost minim? 

Argumentaţi răspunsul dvs.
Schiţaţi un algoritm care caută soluţia optimă parcurgînd arborele soluţiilor în lăţime 

(în adîncime). Estimaţi necesarul de memorie și complexitatea temporală a algoritmului 
elaborat.

Daţi exemple de probleme ce pot fi  soluţionate prin metoda ramifi că și mărginește.
 Încercaţi să desenaţi nodurile de pe nivelele 0, 1 și 2 ale unui arbore ce reprezintă jocul 

dvs. preferat, de exemplu, lupta maritimă, șah, dame etc.
 Cum credeţi, care sînt avantajele și neajunsurile algoritmilor bazaţi pe metoda ramifi că 

și mărginește?

5.8. Aplicaţiile metodei ramifi că şi mărgineşte
Metoda ramifi că şi mărgineşte poate fi  aplicată numai problemelor din enunţul 

cărora rezultă regulile sau operaţiile necesare pentru generarea directă a arborelui 
soluţiilor. De obicei, în astfel de probleme se simulează comportamentul unor par-
teneri ce au scopuri sau tendinţe contrare: aplicaţii militare, scenarii de dezvoltare 
economică în condiţii de concurenţă, jocuri de şah sau de dame etc. Menţionăm că 
existenţa unor reguli de generare directă a descendenţilor face inutilă construirea 
completă a arborelui soluţiilor, micşorîndu-se astfel volumul de memorie internă 
cerut de algoritm.

Implementarea practică a metodei ramifi că şi mărgineşte presupune rezolvarea ur-
mătoarelor probleme:

– determinarea mulţimii de stări care, în funcţie de natura problemei de rezolvat, 
poate fi  fi nită sau infi nită;

– stabilirea transformărilor (operaţiilor) elementare în urma cărora sistemul trece 
dintr-o stare în alta;

– defi nirea funcţiei de cost;
– defi nirea structurilor de date necesare pentru a reprezenta stările sistemului, 

arborele soluţiilor şi lista nodurilor active;
– elaborarea subprogramelor necesare pentru generarea descendenţilor, calcula-

rea funcţiei de cost, selectarea nodurilor de cost minim etc.
Pentru exemplifi care vom analiza modul de implementare a metodei ramifi că şi 

mărgineşte în cazul jocului Perspico, cunoscut şi sub denumirea jocul 15. În acest joc 
sînt 15 plăcuţe pătrate numerotate de la 1 la 15 (fi g. 5.12).

a) b)

Fig. 5.12. Jocul Perspico:
a – starea iniţială; b – starea fi nală

Plăcuţele sînt încadrate într-o ramă pătrată de dimensiunile 44, o poziţie din 
interiorul ramei fi ind liberă. Orice plăcuţă vecină cu poziţia liberă poate fi  mutată pe 
locul respectiv. Se cere a trece, folosind mutările permise, din starea iniţială în starea 
fi nală.

Starea curentă a jocului Perspico poate fi  exprimată printr-o distribuţie a celor 15 
plăcuţe în cele 16 poziţii libere. Numerotînd placa liberă prin 0, putem scrie:

type Stare = array [1..4, 1..4] of 0..15;

Menţionăm că numărul de stări posibile este dat de numărul aranjamentelor 16 
din 16 şi constituie:

,
pe cînd capacitatea memoriei interne a calculatoarelor personale este de ordinul 108 
octeţi.

Trecerea dintr-o stare în alta poate fi  efectuată prin „deplasarea” plăcuţei lipsă 
spre stînga, sus, dreapta, jos (bineînţeles cînd acest lucru este posibil). În PASCAL 
această „deplasare” se exprimă prin modifi carea componentelor respective ale vari-
abilelor de tip Stare.

Cunoscînd stările sistemului şi regulile de trecere dintr-o stare în alta, putem con-
strui arborele soluţiilor, o parte din care este prezentată în fi gura 5.13.



149

Starea
iniţială

Starea
finală

dreapta

jos

sus jos stînga dreapta

Fig. 5.13. Arborele soluţiilor în jocul Perspico

Întrucît plăcuţa liberă poate fi  deplasată pe trasee ciclice, arborele soluţiilor este 
infi nit. Pentru a evita revenirea la stările deja examinate, vom include în arbore nu-
mai stările curente noi. De asemenea, luînd în considerare faptul că memoria internă 
a calculatoarelor personale nu permite stocarea tuturor stărilor posibile, în continu-
are vom examina numai primele 1015 nivele ale arborelui soluţiilor. Evident, în 
prezenţa acestei restricţii nu se mai garantează găsirea, chiar dacă există, a şirului de 
mutări permise care transformă starea iniţială în stare fi nală.

Funcţia de cost f : S  R poate fi  defi nită în forma:
f(si) = niv(si) + h(si),

unde componenta h(si) caracterizează numărul de mutări necesare pentru a trece 
din starea curentă si în stare fi nală. Întrucît acest număr nu poate fi  calculat fără a 

construi mai întîi subarborele de rădăcină si, vom folosi o aproximare prin lipsă a 
numărului necesar de mutări:

h(si) = numărul de plăcuţe care nu se afl ă la locul lor.
De exemplu, pentru arborele din fi gura 5.13 avem:

f(s1) = niv(s1) + h(s1) = 0 + 4 = 4;
f(s2) = niv(s2) + h(s2) = 1 + 5 = 6;

f(s3) = niv(s3) + h(s3) = 1 + 3 = 4;

f(s6) = niv(s6) + h(s6) = 2 + 2 = 4;

f(s7) = niv(s7) + h(s7) = 3 + 0 = 3.
Arborele soluţiilor şi lista nodurilor active pot fi  reprezentate în memoria calcula-

torului cu ajutorul următoarelor structuri de date:

type AdresaNod = ^Nod;
     Nod = record
             S : Stare;
             Nivel, Cost : integer;
             Drum : boolen;
             D : array [1..4] of AdresaNod;
             Tata : AdresaNod;
           end;
     AdresaCelula = ^Celula;
     Celula = record
                ReferintaNod : AdresaNod;
                Urm : AdresaCelula;
              end;
var Radacina : AdresaNod;
    BazaListei : AdresaCelula;

Fiecare nod al arborelui soluţiilor conţine cîmpurile informaţionale S, Nivel, 
Cost, Drum şi cîmpurile de legătură D, Tata. Cîmpul Drum se foloseşte pentru a 
marca nodurile ce aparţin drumului care leagă nodul rădăcină cu nodul ce conţine 
starea fi nală. Cîmpul Tata se utilizează pentru a memora adresa nodului tată.

Fiecare celulă a listei nodurilor active conţine cîmpul informaţional ReferintaNod 
în care se indică adresa nodului inclus în listă. Cîmpul de legătură Urm conţine adre-
sa celulei următoare din listă. Prin urmare, lista nodurilor active conţine nu nodurile 
propriu-zise, ci doar adresele lor.

Subprogramele necesare pentru implementarea metodei ramifi că şi mărgineşte în 
cazul jocului Perspico sînt incluse în programul P158.

Program P158;
 { Jocul Perspico - metoda ramifi ca si margineste }
const NivelMaxim=15;
type Stare=array[1..4, 1..4] of 0..15;
     AdresaNod=^Nod;

     Nod=record
           S : Stare;
           Nivel, Cost : integer;
           Drum : boolean;
           D : array [1..4] of AdresaNod; {adresele
                                           descendentilor}
           Tata : AdresaNod;
         end;
     AdresaCelula=^Celula;
     Celula=record
              ReferintaNod : AdresaNod;
              Urm : AdresaCelula;
            end;

var Radacina : AdresaNod;
    BazaListei : AdresaCelula;
    StareaInitiala, StareaFinala : Stare;
    Gasit : boolean;  { true cind este gasit un drum }
    Finput, Foutput : text; { datele de intrare si iesire }
    Str : array[1..4] of Stare; { starile descendente  }
    m : 0..4;                   { numarul curent de stari }
    i, j : integer;

procedure CalculareaCostului(Adresa : AdresaNod);
var i, j, C : integer;
begin
  C:=0;
  for i:=1 to 4 do
    for j:=1 to 4 do
      if Adresa^.S[i,j] <> StareaFinala[i,j] then C:=C+1;
  Adresa^.Cost:=Adresa^.Nivel+C;
end; { CalculareaCostului }

procedure Initializare;
 { Creeaza nodul radacina si lista nodurilor active }
 { Inscrie in lista nodurilor active nodul radacina }
var i : integer;
begin
  Gasit:=false;
  new(Radacina);
  Radacina^.S:=StareaInitiala;
  Radacina^.Nivel:=0;
  CalculareaCostului(Radacina);
  Radacina^.Drum:=false;
  for i:=1 to 4 do Radacina^.D[i]:=nil;
  Radacina^.Tata:=nil;

  new(BazaListei);
  BazaListei^.ReferintaNod:=Radacina;
  BazaListei^.Urm:=nil;
end; { Initializare }

procedure Ramifi ca(Adresa : AdresaNod);
 { Inscrie in Str toate starile care pot fi  obtinute }
 { din starea curenta prin efectuarea unei mutari permise }
label 1;
var   St : Stare;
      i, j, k : integer;
begin
  { cautarea placutei 0 }
  for i:=1 to 4 do
    for j:=1 to 4 do
      if Adresa^.S[i,j]=0 then goto 1;
1: m:=0;
  { Deplasarea placutei de sus }
  if i<>1 then
    begin
      St:=Adresa^.S;
      St[i,j]:=St[i-1, j];
      St[i-1, j]:=0;
      m:=m+1;
      Str[m]:=St;
    end;
  { Deplasarea placutei de jos }
  if i<>4 then
    begin
      St:=Adresa^.S;
      St[i,j]:=St[i+1, j];
      St[i+1, j]:=0;
      m:=m+1;
      Str[m]:=St;
    end;
  { Deplasarea placutei din stinga }
  if j<>1 then
    begin
      St:=Adresa^.S;
      St[i,j]:=St[i, j-1];
      St[i, j-1]:=0;
      m:=m+1;
      Str[m]:=St;
    end;
 { Deplasarea placutei din dreapta }
 if j<>4 then

    begin
      St:=Adresa^.S;
      St[i,j]:=St[i, j+1];
      St[i, j+1]:=0;
      m:=m+1;
      Str[m]:=St;
    end;
end; { Desparte }

procedure IncludeInLista(Adresa : AdresaNod);
 { Include adresa nodului in lista nodurilor active }
var R : AdresaCelula;
begin
  new(R);
  R^.ReferintaNod:=Adresa;
  R^.Urm:=BazaListei;
  BazaListei:=R;
end; { IncludeInLista }

procedure ExtrageDinLista(var Adresa : AdresaNod);
 { Extrage adresa nodului de cost minim din lista }
label 1;
var P, R : AdresaCelula;
    C : integer;  { costul curent }
begin
  if BazaListei=nil then goto 1;
  { cautarea nodului de cost minim }
  C:=MaxInt;
  R:=BazaListei;
  while R<>nil do
    begin
      if R^.ReferintaNod^.Cost < C then
        begin
          C:=R^.ReferintaNod^.Cost;
          Adresa:=R^.ReferintaNod;
          P:=R;
        end; { then }
      R:=R^.Urm;
    end; { while }
  { excludera nodului de cost minim din lista }
  if P=BazaListei then BazaListei:=P^.Urm
    else
      begin
        R:=BazaListei;
        while P<>R^.Urm do R:=R^.Urm;
        R^.Urm:=P^.Urm;
      end;

    dispose(P);
1:end; { ExtrageDinLista }

function StariEgale(S1, S2 : Stare) : boolean;
 { Returneaza TRUE daca starile S1 si S2 coincid }
var i, j : integer;
    Coincid : boolean;
begin
  Coincid:=true;
  for i:=1 to 4 do
    for j:=1 to 4 do
      if S1[i,j]<>S2[i,j] then Coincid:=false;
  StariEgale:=Coincid;
end; { StariEgale }

function StareDejaExaminata(St : Stare) : boolean;
 { Returneaza TRUE daca starea curenta este deja inclusa in 
arbore }
var EsteInArbore : boolean;

procedure InAdincime(R : AdresaNod);
 { Parcurgerea arborelui in adincime }
label 1;
var i : integer;
begin
  if R<>nil then
    begin
      if StariEgale(St, R^.S) then
        begin
          EsteInArbore:=true;
          goto 1;
        end;
      for i:=1 to 4 do InAdincime(R^.D[i]);
    end;
1:end; { InAdincime }

begin
  EsteInArbore:=false;
  InAdincime(Radacina);
  StareDejaExaminata:=EsteInArbore;
end; { StareDejaExaminata }

procedure Margineste(Adresa : AdresaNod);
  { Selectarea descendentilor si includerea lor in lista }
label 1;
var i, k : integer;
    R : AdresaNod;

begin
  k:=0;
  if (Adresa^.Nivel+1) > NivelMaxim then goto 1;
  for i:=1 to m do
    if not StareDejaExaminata(Str[i]) then
      begin
        k:=k+1;
        new(R);
        R^.S:=Str[i];
        R^.Nivel:=Adresa^.Nivel+1;
        CalculareaCostului(R);
        for j:=1 to 4 do R^.D[j]:=nil;
        Adresa^.D[i]:=R;
        R^.Tata:=Adresa;
        R^.Drum:=false;
        if StariEgale(R^.S, StareaFinala) then
             begin
               R^.Drum:=true;
               Gasit:=true;
             end;
        IncludeInLista(R);
      end;
  writeln(Foutput);
  writeln(Foutput, ’In lista au fost inclusi ’, k, ’ 
descendenti’);
  writeln(Foutput);
1:end; { Margineste }

procedure Afi sareaNodului(R : AdresaNod);
var i, j : integer;
begin
  writeln(Foutput, ’Drum=’, R^.Drum, ’   Nivel=’, R^.Nivel,
                ’  Cost=’, R^.Cost);
  for i:=1 to 4 do
    begin
      for j:=1 to 4 do write(Foutput, R^.S[i, j] : 3);
      writeln(Foutput);
    end;
    for i:=1 to 4 do
      if R^.D[i]<>nil then write(Foutput, ’*** ‘)
                      else write(Foutput, ’nil ‘);
    writeln(Foutput); writeln(Foutput);
end; { Afi sareaNodului }

procedure Ramifi caSiMargineste;
var NodulCurent : AdresaNod;

begin
  Initializare;
  repeat
    ExtrageDinLista(NodulCurent);
    writeln(Foutput, ’   NODUL EXTRAS DIN LISTA’);
    writeln(Foutput, ’   ======================’);
    Afi sareaNodului(NodulCurent);
    Ramifi ca(NodulCurent);
    Margineste(NodulCurent);
  until Gasit or (BazaListei=nil);
end; { Ramifi caSiMargineste }

procedure Afi sareaDrumului;
label 1;
var R : AdresaCelula;
    P, Q : AdresaNod;
begin
  if not Gasit then
    begin
      writeln(Foutput, ’DRUMUL NU A FOST GASIT’);
      goto 1;
    end;
  writeln(Foutput, ’     DRUMUL GASIT:’);
  writeln(Foutput, ’     =============’);
  { cautarea in lista a nodului terminal}
  R:=BazaListei;
  while (R<>nil) and (not R^.ReferintaNod^.Drum) do R:=R^.Urm;
  { marcarea nodurilor care formeaza drumul }
  P:=R^.ReferintaNod;
  while P<>nil do
    begin
      P^.Drum:=true;
      P:=P^.Tata;
    end;
  { afi sarea drumului }
  P:=Radacina;
  while P<>nil do
    begin
      Afi sareaNodului(P);
      Q:=nil;
      for i:=1 to 4 do
        if (P^.D[i]<>nil) and P^.D[i]^.Drum then Q:=P^.D[i];
      P:=Q;
    end;
  writeln(Foutput, ’Sfi rsitul drumului’);
1:end; { Afi sareaDrumului }

begin
  { Citirea starii initiale }
  assign(Finput, ’FINPUT.TXT’);
  reset(Finput);
  for i:=1 to 4 do
    for j:=1 to 4 do read(Finput, StareaInitiala[i, j]);
  { Citirea starii fi nale }
  for i:=1 to 4 do
    for j:=1 to 4 do read(Finput, StareaFinala[i, j]);
  close(Finput);
  { Deschiderea fi sierului de iesire }
  assign(Foutput, ’FOUTPUT.TXT’);
  rewrite(Foutput);
  Ramifi caSiMargineste;
  Afi sareaDrumului;
  close(Foutput);
  writeln(’Gasit=’, Gasit);
  readln;
end.

În programul P158 starea iniţială şi starea fi nală sînt citite din fi şierul FINPUT.
TXT, iar nodurile extrase la fi ecare pas din listă şi drumul găsit sînt înscrise în fi şierul 
FOUTPUT.TXT. Conţinutul acestor fi şiere poate fi  vizualizat sau modifi cat cu ajuto-
rul unor editoare simple de text.

Întrebări şi exerciţii
Indicaţi ordinea în care vor fi  vizitate nodurile arborelui din fi gura 5.13 în cazul parcurge-

rii de cost minim. Folosind ca model tabelul 5.2, completaţi un tabel cu datele referitoare 
la acest arbore.

Explicaţi destinaţia fi ecărui subprogram din programul P158.
Lansaţi în execuţie programul P158 pentru datele iniţiale din fi gura 5.12. Comentaţi re-

zultatele înscrise în fi șierul de ieșire.
Lansaţi în execuţie programul P158 pentru următoarele stări iniţiale ale jocului 

Perspico:
a)

1 2 0 4
b)

1 2 3 4

5 6 3 8 5 7 6 8

9 11 7 12 9 0 10 11

13 10 14 15 13 14 15 12

c)
1 3 4 8

d)
0 1 2 3

5 2 6 0 6 7 8 4

9 10 7 11 5 9 10 11

13 14 15 12 13 14 15 12

Comentaţi rezultatele înscrise de program în fi șierul FOUTPUT.TXT.
 E cunoscut faptul că pentru stările iniţiale ce urmează:

a)
1 3 4 4

b)
1 2 3 4

9 5 7 8 5 0 7 8

13 6 10 11 9 6 10 11

0 14 15 12 13 14 15 12

 programul P158 găsește soluţiile respective. Însă, după introducerea modifi cării
 const NivelMaxim = 5; 

 soluţia pentru cazul (a) nu mai este găsită. Cum credeţi, prin ce se explică acest fapt?
Se consideră varianta simplifi cată a jocului Perspico, denumită jocul 9 (fi g. 5.14). Ce mo-

difi cări trebuie introduse în programul P158 pentru a calcula șirul de mutări permise 
destinate trecerii din starea iniţială în starea fi nală?

a) b)

Fig. 5.14. Varianta simplifi cată a jocului Perspico:
a – starea iniţială; b – starea fi nală

Modifi caţi programul P158 utilizînd următoarea funcţie de cost:
F(si) = niv(si) + g(si),

 unde g(si) este suma distanţelor dintre fi ecare plăcuţă și locul ei în starea fi nală, exprimată 
prin numărul respectiv de mutări. De exemplu, pentru starea iniţială din fi gura 5.12 avem:

g(si) = 0 + 0 + 0 + 0 + 0 + 0 + 1 + 0 + 0 + 0 + 1 + 0 + 0 + 0 + 1 + 3 = 6.

 Verifi caţi cum derulează programul modifi cat în cazul stărilor iniţiale din exerciţiile 4 și 5. 
Elaboraţi planul unui experiment care ne-ar permite să stabilim care din funcţiile h(si), 
g(si) micșorează timpul de calcul.

Comparaţi complexitatea programului P158 cu complexitatea unui program bazat pe 
metoda trierii.

Se consideră o tablă de șah pe care se află o singură piesă  un cal de culoare albă. 
Elaboraţi un program care determină dacă piesa respectivă poate trece, utilizînd 
mutările permise, din poziţia (, ) în poziţia (, ). Amintim că ,   {A, B, ..., H}, iar 
,   {1, 2, ..., 8}.

Problema comis-voiajorului*. Se consideră n orașe între care există curse aeriene di-
recte. În orașul i se afl ă un comis-voiajor care trebuie să viziteze celelalte n-1 orașe și să 
revină în localitatea din care s-a pornit. Elaboraţi un program care, cunoscînd distanţele 
dij între orașe, determină ordinea în care ele vor fi  vizitate. Se cere ca distanţa parcursă 
de comis-voiajor să fi e cît mai mică, iar fi ecare oraș să fi e vizitat numai o singură dată.

 Scrieţi un program PASCAL care soluţionează problema comis-voiajorului prin metoda 
trierii. Comparaţi complexitatea programului elaborat cu complexitatea programului 
bazat pe metoda ramifi că și mărginește.

 Indicaţii: Drumul parcurs de comis-voiajor poate fi  reprezentat ca o mulţime ordonată (1, 
i,  j, ..., k, 1) de orașe în care i, j, ..., k  {2, 3, ..., n}. Pentru a genera toate drumurile posibile, 
se calculează permutările mulţimii {2, 3, ..., n}.

 Efectuaţi un studiu comparativ al algoritmilor bazaţi pe metoda ramifi că și mărginește și 
algoritmilor bazaţi pe metoda trierii.

5.9. Algoritmi exacţi şi algoritmi euristici
Algoritmii utilizaţi pentru rezolvarea problemelor cu ajutorul calculatorului se 

împart în două categorii distincte: algoritmi exacţi şi algoritmi euristici.
Un algoritm este exact numai atunci cînd el găseşte soluţiile optime ale proble-

melor pentru rezolvarea cărora a fost conceput. Desigur, acest fapt trebuie confi rmat 
printr-o demonstraţie matematică riguroasă.

Un algoritm este euristic atunci cînd el găseşte soluţii bune, dar nu neapărat op-
time. În astfel de cazuri demonstraţia matematică respectivă nu există sau nu este 
cunoscută. De obicei, algoritmii euristici includ prelucrări care, chiar dacă nu pot fi  
justifi cate matematic, sînt alese în virtutea experienţei sau intuiţiei programatorului.

Evident, algoritmii bazaţi pe metoda trierii sînt exacţi, întrucît în procesul exa-
minării consecutive a soluţiilor posibile neapărat va fi  găsită şi soluţia optimă. 
Confi rmarea exactităţii acestor algoritmi se reduce la demonstrarea faptului că în 
procesul derulării pe calculator vor fi  generate şi analizate toate elementele din mul-
ţimea soluţiilor posibile. În cazul algoritmilor bazaţi pe celelalte tehnici de progra-
mare  tehnica Greedy, metoda reluării, metoda ramifi că şi mărgineşte etc.  un algo-
ritm va fi  exact sau euristic în funcţie de natura condiţiilor care ne permit să evităm 

* Persoană care se ocupă cu mijlocirea vînzărilor de mărfuri, deplasîndu-se în diferite locuri în 
căutarea unor benefi ciari.

trierea tuturor soluţiilor posibile. Dacă aceste condiţii sînt alese nereuşit, soluţiile op-
time pot fi  pierdute şi, în consecinţă, algoritmul respectiv nu va mai fi  exact. Pentru 
exemplifi care vom analiza problema ce urmează.

Problema drumului minim. Se consideră n oraşe legate printr-o reţea de drumuri 
(fi g. 5.15). Cunoscînd distanţele dij dintre oraşele vecine, determinaţi cel mai scurt 
drum din oraşul a în oraşul b.

Fig. 5.15. Reţea de drumuri

Datele iniţiale ale problemei în studiu pot fi  descrise cu ajutorul matricei (tabelu-
lui bidimensional) D = ||dij||nn cu n linii şi n coloane, denumită matricea distanţelor. 
În această matrice componenta dij este egală cu distanţa dintre oraşele i, j atunci cînd 
ele sînt vecine şi 0 în caz contrar. Prin defi niţie, dii = 0, i = 1, 2, ..., n.

De exemplu, pentru oraşele din fi gura 5.15 avem:

 1  2 3 4 5 6
1  0 1 2 6 0 0

2 1 0 0 0 0 0

3 2 0 0 2 3 0

4 6 0 2 0 5 0

5 0 0 3 5 0 2

6 0 0 0 0 2 0

D =

Drumul minim ce leagă oraşele a = 1, b =  6 are lungimea 7 şi include localităţile 
1, 3, 5, 6.

În general, drumul minim poate fi  găsit prin metoda trierii, generînd consecutiv 
toate permutările mulţimilor ordonate

unde q ia valori de la 0 la n-2. Este clar că algoritmii bazaţi pe generarea tuturor 
permutărilor întotdeauna determină drumul minim, însă complexitatea temporală 
O(n!) a acestor algoritmi este inacceptabilă.

Pentru a reduce volumul de calcule, vom încerca să determinăm drumul minim 
prin metoda reluării. În această metodă construirea drumului începe cu oraşul ini-
ţial x1 = a şi continuă cu primul dintre vecinii săi nevizitaţi, fi e acesta x2, trecîndu-se 
la primul dintre vecinii lui x2 nevizitaţi încă ş.a.m.d. Pentru a construi drumuri cît 
mai scurte, vom aplica următoarea regulă intuitivă: la fi ecare pas vom examina, în 
primul rînd, vecinii nevizitaţi care se afl ă cît mai aproape de oraşul curent.

Drumul în construcţie poate fi  reprezentat în forma unui vector:
X = (a, x2, ..., xk-1, xk, ..., b),

în care componenta xk trebuie să fi e unul din vecinii oraşului xk-1. Pentru a sistematiza 
calculele vom memora vecinii încă nevizitaţi ai oraşului i în mulţimea Ai , i = 1, 2, ..., 
n. Evident, conform regulii intuitive formulate mai sus, oraşele din fi ecare mulţime 
Ai trebuie sortate în ordinea creşterii distanţelor dij, j  Ai.

De exemplu, pentru fi gura 5.15 iniţial vom avea:
A1 = (2, 3, 4);
A2 = (1);
A3 = (1, 4, 5);
A4 = (3, 5, 6);
A5 = (6, 3, 4);
A6 = (5).

Condiţiile de continuare în metoda reluării rezultă direct din enunţul problemei: 
oraşul xk poate fi  adăugat la porţiunea de drum deja construită (a, x2, ..., xk-1) numai 
atunci cînd:

1) xk este un vecin încă nevizitat al oraşului xk-1, deci xk  Ak-1;
2) oraşul xk anterior nu a fost inclus în drumul în curs de construcţie, deci xk  a, 

xk  x2, ..., xk  xk-1.
De exemplu, pentru fi gura 5.15 vectorul X va lua consecutiv următoarele valori:

X = (1);
X = (1, 2);
X = (1);
X = (1, 3);
X = (1, 3, 4); 
X = (1, 3, 4, 5);
X = (1, 3, 4, 5, 6).

Drumul (1, 3, 4, 5, 6) construit prin metoda reluării are lungimea 11 şi, evident, 
nu este un drum minim. Totuşi acest drum este mai bun ca drumul (1, 4, 5, 6) care 
are lungimea 13.

În programul ce urmează elementele mulţimilor A1, A2, ..., An sînt plasate la în-
ceputul liniilor A[1], A[2], ..., A[n] ale tabelului bidimensional A, restul poziţiilor 
avînd valoarea zero.

Program P159;
 { Problema drumului minim - metoda reluării }
const nmax=50;
var n : integer; { numărul de oraşe }
    D : array[1..nmax, 1..nmax] of real; { matricea distanţelor }
    a, b : 1..nmax;
    X : array [1..nmax] of integer;      { drumul construit }
    V : array[1..nmax, 1..nmax] of integer; { vecinii }
    Finput : text;

procedure InitializareVecini;
 { Înscrie în V[k] vecinii oraşului k }
var k, i, j, p, q, r : integer;
begin
  for k:=1 to n do
    begin
      { iniţial mulţimea V[k] este vidă }
      for i:=1 to n do V[k,i]:=0;
      { calculăm elementele mulţimii V[k] }
      i:=0;
      for j:=1 to n do
        if D[k,j]<>0 then
          begin
            i:=i+1;
            V[k,i]:=j;
          end; { then }
      { sortarea elementelor mulţimii V[k] prin metoda bulelor }
      for j:=1 to i do
        for p:=1 to i-1 do
          if D[k, V[k,p]]>D[k, V[k, p+1]] then
            begin
              q:=V[k,p];
              V[k,p]:=V[k, p+1];
              V[k, p+1]:=q;
            end; { then }
    end; { for }
end; { InitializareVecini }

procedure Initializare;
var i, j : integer;
begin
  assign(Finput, ’DRUM.IN’);
  reset(Finput);
  readln(Finput, n);
  readln(Finput, a, b);
  writeln(’n=’, n, ’   a=’, a, ’  b=’, b);
    for i:=1 to n do
      for j:=1 to n do read(Finput, D[i,j]);



163

  close(Finput);
  InitializareVecini;
end; { Initializare }

function MultimeVida(k : integer) : boolean;
begin
  MultimeVida:=(V[X[k-1], 1]=0);
end; { MultimeVida }

function PrimulElement(k : integer) : integer;
var i : integer;
begin
  PrimulElement:=V[X[k-1], 1];
  for i:=1 to n-1 do V[X[k-1],i]:=V[X[k-1], i+1];
end; { PrimulElement }

function ExistaSuccesor(k : integer) : boolean;
begin
  ExistaSuccesor:=(V[X[k-1], 1]<>0);
end; { ExistaSuccesor }

function Succesor(k : Integer) : integer;
var i : integer;
begin
  Succesor:=V[X[k-1], 1];
  for i:=1 to n-1 do V[X[k-1], i]:=V[X[k-1], i+1];
end; { Succesor }

function Continuare(k : integer) : boolean;
var i : integer;
Indicator : boolean;
begin
  Continuare:=true;
  for i:=1 to k-1 do
   if X[i]=X[k] then Continuare:=false;
end; { Continuare }

procedure PrelucrareaSolutiei(k : integer);
var i : integer;
    s : real;
begin
  writeln(’Drumul gasit:’);
  for i:=1 to k-1 do write(X[i] : 3);
 writeln;
  s:=0;
  for i:=1 to k-1 do s:=s+D[X[i], X[i+1]];
  writeln(’Lungimea drumului ’, s : 10:2);

  readln;
  halt;
end; { PrelucrareaSolutiei }

procedure Reluare(k : integer);
label 1;
var i : integer;
begin
  if MultimeVida(k) then goto 1;
  if X[k-1]<>b then
    begin
      X[k]:=PrimulElement(k);
      if Continuare(k) then Reluare(k+1);
      while ExistaSuccesor(k) do
        begin
          X[k]:=Succesor(k);
          if Continuare(k) then Reluare(k+1);
        end; { while }
    end { then}
  else PrelucrareaSolutiei(k);
1:end; { Reluare }

begin
  Initializare;
  X[1]:=a;
  Reluare(2);
end.

Din analiza programului P159 se observă că algoritmul bazat pe metoda reluă-
rii are complexitatea O(mn), deci este exponenţial. În această formulă m reprezintă 
numărul maxim de vecini pe care îi poate avea fi ecare oraş. Menţionăm că numă-
rul concret de operaţii efectuate de programul P159 depinde în mare măsură de 
confi guraţia reţelei de drumuri şi distanţele respective. Evident, „plata” pentru 
reducerea numărului de operaţii în algoritmul euristic bazat pe metoda reluării în 
comparaţie cu algoritmul exact bazat pe metoda trierii constă în pierderea soluţi-
ilor optime.

În scopul soluţionării efi ciente a problemelor de o reală importanţă practică, în 
informatică se depun eforturi considerabile pentru elaborarea unor algoritmi exacţi 
care ar avea o complexitate polinomială. În cazul problemei drumului minim un ast-
fel de algoritm poate fi  elaborat utilizînd metoda programării dinamice.

Amintim că metoda programării dinamice poate fi  aplicată doar problemelor care 
satisfac principiul optimalităţii, principiul care se respectă în cazul drumurilor mini-
me. Pentru a formula relaţiile respective de recurenţă, introducem în studiu matri-
cea costurilor C = ||cij||nn, elementele căreia indică lungimea drumului minim între 
oraşele i, j. Componentele cij se calculează după cum urmează:

1) mai întîi se examinează drumurile minime care leagă oraşele vecine:
dacă i = j;
dacă oraşele i, j sînt vecine;
în caz contrar;

2) în continuare se examinează drumurile minime între oraşele i, j formate prin 
folosirea localităţii k drept punct de conexiune:

cij = min(cik, cik+cjk),   i, j  {1, 2, ..., n},   i  j,   i  k,   j  k;
3) punctul 2 se repetă pentru k = 1, 2, ..., n.
Valorile curente ale matricei costurilor pentru oraşele din fi gura 5.15 sînt prezen-

tate în fi gura 5.16.

Matricea iniţială k = 1 k = 2

k = 3 k = 4 k = 5

k = 6

Fig. 5.16. Valorile curente ale matricei costurilor

Întrucît matricea costurilor C nu include drumurile minime propriu-zise, ci nu-
mai lungimile lor, vom construi drumul minim ce leagă oraşele a, b cu ajutorul teh-
nicii Greedy. Conform acestei tehnici, construcţia drumului minim X = (x1, ..., xk-1, xk, 
...) începe cu oraşul iniţial x1 = a. În continuare, la fi ecare pas k se alege acel oraş xk, xk 
 Ak-1, care satisface principiul optimalităţii:

C[a, xk] + C[xk, b] = C[a, b].
De exemplu, pentru oraşele a = 1, b = 6 din fi gura 5.15 avem:

X = (1);
X = (1, 3);
X = (1, 3, 5);
X = (1, 3, 5, 6).

Algoritmul de determinare a drumurilor minime bazat pe metoda programării 
dinamice este cunoscut în literatura de specialitate sub denumirea Roy-Floyd. În ca-
litate de exerciţiu vă propunem să demonstraţi că acest algoritm este exact, adică 
întotdeauna construieşte numai drumuri minime.

În programul ce urmează algoritmul exact este implementat cu ajutorul procedu-
rii RoyFloyd.

Program P160;
 { Problema drumului minim - metoda programării dinamice }
const nmax=50;
      Infi nit=1.0E+35;
var n : integer; { numarul de orase }
    D : array[1..nmax, 1..nmax] of real; { matricea distanţelor }
    a, b : 1..nmAx;
    X : array [0..nmax+1] of integer;      { drumul construit }
    V : array[1..nmax, 1..nmax] of integer; { vecinii }
    C : array[1..nmax, 1..nmax] of real; { matricea costurilor }
    Finput : text;

procedure InitializareVecini;
 { Înscrie in V[k] vecinii oraşului k }
var k, i, j, p, q, r : integer;
begin
  for k:=1 to n do
    begin
      { iniţial multimea V[k] este vidă }
      for i:=1 to n do V[k,i]:=0;
      { calculăm elementele mulţimii V[k] }
      i:=0;
      for j:=1 to n do
       if D[k,j]<>0 then
          begin
            i:=i+1;
            V[k,i]:=j;



167

          end; { then }
    end; { for }
end; { InitializareVecini }

procedure Initializare;
var i, j : integer;
begin
  assign(Finput, ’DRUM.IN’);
  reset(Finput);
  readln(Finput, n);
  readln(Finput, a, b);
  writeln(’n=’, n, ’    a=’,  a, ’  b=’, b);
    for i:=1 to n do
      for j:=1 to n do read(Finput, D[i,j]);
  close(Finput);
  InitializareVecini;
end; { Initializare }

procedure Afi sareaDrumului;
var k : integer;
begin
  write(’Drumul găsit: ’);
  k:=1;
  repeat
    write(X[k] : 3);
    k:=k+1;
  until X[k]=0;
  writeln;
  writeln(’Lungimea drumului ’, C[a, b] : 5);
  readln;
end; { PrelucrareaSolutiei }

function Min(p, q : real) : real;
 { Returnează minimul din p si q }
var s : real;
begin
  if p<q then s:=p else s:=q;
  if s>Infi nit then s:=Infi nit;
  Min:=s;
end; { Min }

procedure RoyFloyd;
var i, j, k : integer;
    s : real;
    ors : integer; { oraşul candidat la includerea în drumul
                     minim }
    cnd : boolean; { condiţiile de includere a oraşului în drum }

begin
  { Iniţializarea matricei costurilor }
  for i:=1 to n do
    for j:=1 to n do
    if (D[i,j]=0) and (i<>j) then C[i,j]:=Infi nit
                             else C[i,j]:=D[i,j];
  { Calcularea matricei costurilor }
  for k:=1 to n do
    for i:=1 to n do
      if i<>k then
        for j:=1 to n do
          if j<>k then C[i,j]:=Min(C[i,j], C[i,k]+C[j,k]);
  { Trasarea drumului - tehnica Greedy}
  for k:=1 to n do X[k]:=0;
  k:=1; X[1]:=a;
  while X[k]<>b do
    begin
      i:=1;
      while V[X[k], i]<>0 do
        begin
          ors:=V[X[k], i];
          cnd:=true;
          for j:=1 to k do if ors=X[j] then cnd:=false;
          if cnd and (C[A, ors]+C[ors, B]=C[a,b])
            then X[k+1]:=ors;
          i:=i+1;
        end; { while }
      k:=k+1
    end; { while }
end; { RoyFloyd }

begin
  Initializare;
  RoyFloyd;
  Afi sareaDrumului;
end.

Din analiza procedurii RoyFloyd se observă că complexitatea temporală a algo-
ritmului exact este O(n3), deci polinomială.

În practică, pentru soluţionarea unei probleme, mai întîi se încearcă elaborarea 
unui algoritm exact de complexitate polinomială. Dacă această tentativă eşuează, se 
elaborează un algoritm euristic. Pentru a sistematiza acest proces este convenabil să 
se pună în evidenţă toate condiţiile pe care le satisface o soluţie optimă. Condiţiile 
respective pot fi  împărţite în două clase:

1) condiţii necesare în sensul că neîndeplinirea lor împiedică obţinerea unei solu-
ţii posibile pentru problemă;

2) condiţii pentru care se poate accepta un compromis, în sensul că ele pot fi  înlo-
cuite cu alte condiţii ce permit apropierea de o soluţie optimală.

De exemplu, în cazul drumului minim X = (a, x2, ..., xk-1, xk, ..., b) faptul că xk trebuie 
să fi e un vecin al oraşului xk-1 este o condiţie necesară, iar respectarea principiului op-
timalităţii este o condiţie de compromis. În metoda reluării condiţia de compromis a 
fost înlocuită cu una mai simplă, şi anume, oraşul xk trebuie să fi e cît mai aproape de 
oraşul xk-1. Evident, formularea condiţiilor care acceptă un compromis şi înlocuirea 
lor cu altele mai simple cade în sarcina programatorului.

Întrebări şi exerciţii
Care este diferenţa dintre algoritmii exacţi și cei euristici?
În care cazuri algoritmii euristici sînt „mai buni” decît cei exacţi?
Scrieţi un program care determină cel mai scurt drum prin metoda trierii. Estimaţi com-

plexitatea temporală a programului elaborat.
Efectuaţi un studiu comparativ al algoritmilor exacţi și algoritmilor euristici destinaţi 

soluţionării problemei drumului minim.
Calculaţi matricea distanţelor pentru reţeaua de drumuri auto ce leagă centrele raionale. 

Determinaţi cele mai scurte drumuri între centrele raionale date cu ajutorul algoritmilor 
exacţi și algoritmilor euristici.

Cum credeţi, poate fi  oare aplicată metoda trierii pentru determinarea celui mai scurt 
drum între oricare două localităţi din ţară ? Argumentaţi răspunsul dvs.

Formulaţi condiţiile necesare și condiţiile pentru care se poate accepta un compromis în 
cazul robotului ce explorează un teren aurifer (vezi paragraful 5.6).

Estimaţi complexitatea algoritmilor exacţi și a algoritmilor euristici destinaţi soluţionării 
problemelor ce urmează:

 a) memorarea fi șierelor pe benzi magnetice (exerciţiul 5, paragraful 5.3);
 b) problema continuă a rucsacului (exerciţiul 6, paragraful 5.3);
 c) colorarea unei hărţi (exerciţiul 9, paragraful 5.4);
 d) problema discretă a rucsacului (exerciţiul 7, paragraful 5.6);
 e) arhivarea fi șierelor (exerciţiul 9, paragraful 5.6);
 f ) triangularea polinoamelor (exerciţiul 10, paragraful 5.6);
 g) jocul Perspico (paragraful 5.8);
 h) problema comis-voiajorului (exerciţiul 10, paragraful 5.8).

Capitolul 6

ALGORlTMI  DE  REZOLVARE  A  UNOR 
PROBLEME  MATEMATICE

6.1. Operaţii cu mulţimi
Generarea soluţiilor posibile ale unei probleme presupune prelucrarea elemente-

lor ce aparţin diferitor mulţimi.
Fie A o mulţime oarecare cu n elemente:

A = {a1, a2, …, aj, …, an}.
Întrucît limbajul PASCAL nu permite accesul direct la elementele mulţimilor de-

scrise cu ajutorul tipului de date set, vom memora elementele mulţimii A într-un 
vector (tablou unidimensional) cu n componente A=(a1, a2, …, aj, …, an). Evident, o 
astfel de reprezentare stabileşte implicit o ordine a elementelor mulţimii corespun-
zătoare ordinii în care apar componentele în vectorul A. 

Fie Ai o submulţime a lui A. În PASCAL această submulţime poate fi  reprezentată 
prin vectorul caracteristic al submulţimii:

Bi =  (b1, b2, …, bj, …, bn),
unde

dacă
în caz contrar.

Între submulţimile Ai ale mulţimii A şi vectorii caracteristici Bi există o corespon-
denţă biunivocă:

A1 =                      B1 = (0, 0, …, 0);
A2 = {a1}                   B2 = (1, 0, …, 0);
A3 = {a2}                   B3 = (0, 1, …, 0);
A4 = {a1, a2}              B3 = (1, 1, …, 0);
...                                     ...
Ak = {a1, a2, …, an}    Bk = (1, 1, …, 1);

Evident, numărul tuturor submulţimilor unei mulţimi este k = 2n.
Reprezentarea submulţimilor cu ajutorul vectorilor caracteristici presupune ela-

borarea unei unităţi de program ce conţine cîte o procedură pentru fi ecare din opera-
ţiile frecvent utilizate în calculul cu mulţimi:  – reuniunea,  – intersecţia, \ – dife-
renţa, ¯ – complementarea. Scrierea procedurilor respective este propusă cititorului 
în calitate de exerciţiu.

O altă operaţie frecvent utilizată în algoritmii bazaţi pe metoda trierii este gene-
rarea tuturor submulţimilor unei mulţimi. Realizarea acestei operaţii în programele 
PASCAL este prezentată cu ajutorul exemplului  ce urmează.

Exemplul 1. Se consideră mulţimea A={a1, a2, …, an} formată din n numere întregi. 
Determinaţi dacă există cel puţin o submulţime Ai,  Ai  A,  suma elementelor căreia 
este egală cu  m.

Rezolvare. Soluţiile posibile  , {a1}, {a2}, {a1, a2} ş.a.m.d. pot fi  generate formînd 
consecutiv vectorii binari B1, B2, ..., Bk.

Program P161;
 { Generarea tuturor submulţimilor unei mulţimi }
const nmax=50;
type Multime = array [1..nmax] of integer;
     CifraBinara = 0..1;
     VectorCaracteristic = array[1..nmax] of CifraBinara;
var A : Multime;
    B : VectorCaracteristic;
    n, m, j : integer;

function SolutiePosibila : boolean;
var j, suma : integer;
begin
  suma:=0;
  for j:=1 to n do
    if B[j]=1 then suma:=suma+A[j];
  if suma=m then SolutiePosibila:=true
            else SolutiePosibila:=false;
end; { SolutiePosibila }

procedure PrelucrareaSolutiei;
var j : integer;
begin
  write(’Submulţimea: ’);
  for j:=1 to n do
    if B[j]=1 then write(A[j], ’ ’);
  writeln;
end; { PrelucrareaSolutiei }

procedure GenerareSubmultimi(var t:CifraBinara);
var j : integer;
begin
  t:=1; { transportul }
  for j:=1 to n do
     if t=1 then
       if B[j]=1 then B[j]:=0
                 else begin B[j]:=1; t:=0 end;
end; { GenerareSubmultimi }

procedure CautareSubmultimi;
var i : integer;
    t : CifraBinara;
begin
  for j:=1 to n do B[j]:=0;
  { începem cu vectorul caracteristic B=(0, 0, ..., 0) }
  repeat
    if SolutiePosibila then PrelucrareaSolutiei;
    GenerareSubmultimi(t);
  until t=1;
end; { CautareSubmultimi }

begin
  write(’Daţi n=’); readln(n);
  writeln(’Daţi ’, n, ’ numere intregi:’);
  for j:=1 to n do read(A[j]);
  write(’Daţi m=’); readln(m);
  CautareSubmultimi;
  writeln(’Sfîrşit’);
  readln;
end.

În programul P161 trierea consecutivă a soluţiilor posibile este organizată cu aju-
torul ciclului  repeat...until  din componenţa procedurii CautareSubmultimi. 
Vectorii caracteristici ai submulţimilor respective sînt formaţi cu ajutorul procedurii 
GenerareSubmultimi. În această procedură vectorul caracteristic B este tratat ca 
un număr binar valoarea căruia  trebuie mărită cu o unitate la fi ecare apel.

Instrucţiunile if din componenţa procedurii GenerareSubmultimi simulează 
funcţionarea unui semisumator care adună cifrele binare bj şi t, variabila t reprezen-
tînd transportul. Valoarea  t = 1  a transportului din rangul n indică faptul că de la 
vectorul fi nal  Bk = (1, 1, …, 1)  se trece la vectorul iniţial B1 = (0, 0, …, 0).

Complexitatea temporală a algoritmilor bazaţi pe generarea tuturor submulţimi-
lor unei mulţimi este O(2n).

În unele probleme mulţimea soluţiilor posibile S poate fi  calculată ca produsul 
cartezian al altor mulţimi.

Exemplul 2. Se consideră n mulţimi A1, A2, …, An, fi ecare mulţime Aj fi ind formată 
din mj numere întregi. Selectaţi din fi ecare mulţime Aj cîte un număr aj în aşa mod, 
încît produsul a1 a2  ...  an să fi e maxim.

Rezolvare. Mulţimea soluţiilor posibile S=A1A2...An. Evident, numărul soluţiilor 
posibile k = m1 m2 … mn. Fiecare element  si  al produsului cartezian  A1A2...An  
poate fi  reprezentat prin vectorul indiciilor:

Ci = (c1, c2, …, cj, …, cn),
unde cj este indicele elementului respectiv  din mulţimea Aj. De exemplu, pentru 

                                               
A1 = (-6,  2,  1);    A2 = (4,  9);    A3 = (-8,  3,  5),

obţinem:
s1  = (-6, 4, -8)         C1 = (, , );
s2  = ( 2, 4, -8)         C2 = (, , );
s3  = ( 1, 4, -8)         C3 = (, , );
s4  = (-6, 9, -8)         C4 = (, , );
s5  = ( 2, 9, -8)         C5 = (, , );
…
s18= ( 1, 9,  5)         C18 = (, , ),

unde ,  şi  sînt indicii elementelor din mulţimile respective.
Vectorii C1, C2, …, Ck pot fi  generaţi în ordine lexicografi că pornind de la vectorul 

iniţial C1=(1, 1, …, 1).

Program P162;
 { Generarea elementelor unui produs cartezian }
const nmax=50; { numărul maximal de mulţimi }
      mmax=50; { numărul maximal de elemente }
type Multime = array [1..mmax] of integer;
     VectorIndicii = array[1..nmax] of 1..mmax;

var A : array[1..nmax] of Multime;
    n : 1..nmax;                   { numărul de mulţimi }
    M : array[1..nmax] of 1..mmax; { cardinalul mulţimii A[i] }
    Pmax : integer;                { produsul maximal }
    C, Cmax : VectorIndicii;
    i, j : integer;

procedure PrelucrareaSolutieiPosibile;
var j, p : integer;
begin
  p:=1;
  for j:=1 to n do p:=p*A[j, C[j]];
  if p > Pmax then begin Pmax:=p; Cmax:=C end;
end; { PrelucrareaSolutieiPosibile }

procedure GenerareProdusCartezian(var t:integer);
var j : integer;
begin
  t:=1; { transportul }
  for j:=1 to n do
     begin
       C[j]:=C[j]+t;
       if C[j]<=M[j] then t:=0 else C[j]:=1;
     end; { for }
end; { GenerareProdusCartezian }

procedure CautareaProdusuluiMaximal;
var j : integer;
    t : integer;
begin
  Pmax:=-MaxInt;
  writeln(’Pmax=’, Pmax);
  for j:=1 to n do C[j]:=1;
  { începem cu vectorul indiciilor C=(1, 1, ..., 1) }
  repeat
    PrelucrareaSolutieiPosibile;
    write(’Produsul cartezian: ’);
    for j:=1 to n do write(A[j, C[j]], ’ ’); writeln;
    GenerareProdusCartezian(t);
  until t=1;
end; { CautareaProdusuluiMaximal }

begin
  write(’Daţi numărul de mulţimi n=’); readln(n);
  for i:=1 to n do
    begin
      write(’Daţi cardinalul M[’, i, ’]=’); readln(M[i]);
      write(’Daţi elementele multimii A[’, i, ’]: ’);
      for j:=1 to M[i] do read(A[i, j]);
      writeln;
    end;

  CautareaProdusuluiMaximal;

  writeln(’Pmax=’, Pmax);
  write(’Elementele selectate: ’);
  for j:=1 to n do write(A[j, Cmax[j]], ’ ’);
  writeln;
  readln;
  readln;
end.

În procedura GenerareProdusCartezian vectorul indiciilor C este tratat ca 
un număr natural scris într-un sistem mixt de numeraţie. În acest sistem cifra c1 
este scrisă în baza m1, cifra c2 – în baza m2, cifra c3 – în baza m3 ş.a.m.d. La fi ecare 
apel al procedurii GenerareProdusCartezian valoarea numărului înscris în 
vectorul C este mărită cu o unitate. Valoarea  t = 1  a transportului din rangul n 
indică faptul că de la vectorul fi nal Ck = (m1, m2, …, mn) se trece la vectorul iniţial 
C1 = (1, 1, …, 1).

Menţionăm că dacă numărul de mulţimi n este cunoscut pînă la scrierea progra-
mului, generarea elementelor produsului cartezian A1A2…An poate fi  făcută mult 
mai simplu cu ajutorul a n cicluri imbricate:

for j1:=1 to m1 do
  for j2:=1 to m2 do

      ...

      for jn:=1 to mn do
              if SolutiePosibila(aj1, aj2 ,..., ajn)
                then PrelucrareaSoluţiei(aj1, aj2, ..., ajn)

Complexitatea temporală a algoritmilor bazaţi pe generarea tuturor elementelor 
unui produs cartezian este O(mn), unde  m = max(m1, m2, …, mn).

Întrebări şi exerciţii
Submulţimile Ai, Aj  ale mulţimii A sînt reprezentate prin vectori caracteristici. Elaboraţi 

procedurile necesare  pentru efectuarea următoarelor operaţii: Ai  Aj, Ai  Aj, Ai \ Aj, Ai.
Orice submulţime Ai, Ai  A, poate fi  reprezentată printr-un vector cu n componente, 

unde elementele submulţimii Ai sînt plasate la începutul vectorului, iar restul poziţiilor 
au o valoare ce nu aparţine mulţimii A. Elaboraţi procedurile necesare pentru efectuarea 
operaţiilor frecvent întîlnite în calculul cu mulţimi: , , \ , ¯. Cum credeţi, care repre-
zentare a submulţimilor este mai comodă: prin vectorii caracteristici sau prin vectorii ce 
conţin chiar elementele submulţimii?

Estimaţi timpul de execuţie a procedurii CautareSubmultimi din programul P161. 
Verifi caţi aceste estimări prin măsurări directe ale timpului de execuţie pentru diferite 
valori ale lui n.

Se consideră mulţimea A formată din primele n caractere ale alfabetului latin. Elaboraţi 
un program care afi șează la ecran toate submulţimile acestei mulţimi.

Se consideră numărul natural n = 32*35*17*, format din 9 cifre zecimale. Determinaţi 
cifrele care trebuie înscrise în poziţiile marcate cu simbolul * pentru ca numărul obţinut 
să se împartă fără rest la m.

Estimaţi timpul de execuţie a procedurii CautareaProdusuluiMaximal din pro-
gramul P162. Verifi caţi aceste estimări prin măsurări directe ale timpului de execuţie 
pentru diferite valori ale lui n și m1, m2, …, mn.

Într-un coș sînt m mere și p pere. Să se genereze toate posibilităţile de a alege f fructe 
dintre care k să fi e mere.

Elaboraţi o procedură recursivă care generează toate elementele unui produs cartezian.
Fie  A = (a1, a2, …, aj, …, an) o mulţime ordonată de caractere numită alfabet. Numim cu-

vînt de lungime p orice succesiune de p caractere din alfabetul A. Elaboraţi o procedură 
care generează toate cuvintele de lungimea p.

6.2. Analiza combinatorie
În rezolvarea multor probleme implementarea algoritmilor bazaţi pe analiza con-

secutivă a soluţiilor posibile presupune generarea permutărilor, aranjamentelor sau 
combinărilor unei mulţimi.

Generarea permutărilor. E cunoscut faptul că numărul de permutări posibile ale 
unei mulţimi A={a1, a2, ..., an} cu n elemente se determină ca Pn = n!. Acest număr poa-
te fi  calculat cu ajutorul funcţiei factorial, exprimată în formă iterativă:

Pn = 1· 2· 3· ...· n
sau recursivă:

dacă
dacă

De exemplu, pentru A={a1, a2} cele P2 = 2! = 2 permutări sînt (a1, a2) şi (a2, a1). Pentru 
A={a1, a2, a3} cele P3 = 3! = 6 permutări sînt:

(a1, a2, a3);      (a1, a3, a2);      (a2, a1, a3);
(a2, a3, a1);      (a3, a1, a2);      (a3, a2, a1).

Întrucît între permutările mulţimii A={a1, a2, ..., an} şi permutările mulţimii I={1, 
2, ..., n} există o corespondenţă biunivocă, problema generării permutărilor oricărei 
mulţimi A cu n elemente se reduce la generarea permutărilor mulţimii {1, 2, ..., n}, 
denumite permutări de grad n.

Există mai multe metode ingenioase de generare a permutărilor de grad n, cea 
mai răspîndită fi ind metoda lexicografi că. În această metodă se pleacă de la per-
mutarea cea mai mică în ordine lexicografi că, şi anume de la permutarea identică 
(1, 2, ..., n).

Avînd construită o permutare p = (p1, ...,  pi-1, pi,  pi+1, ..., pn), pentru determinarea 
următoarei permutări p´ care îi urmează în ordine lexicografi că se caută acel indice i 
care satisface relaţiile:

pi < pi+1; pi+1 > pi+2 > ... > pn.
În continuare, elementul pi este înlocuit cu cel mai mic dintre elementele pi+1, ...,  pn 

care este mai mare decît pi, fi e el pk:
(p1, ...,  pi-1,  pk,  pi+1, ...,  pk-1, pi, pk+1, ..., pn).

Permutarea căutată p´ se obţine prin inversarea ordinii ultimilor (n – i) elemente 
din acest vector, astfel încît ele să apară în ordine crescătoare.

Dacă nu există niciun indice i ca mai sus, înseamnă că s-a ajuns la permutarea cea 
mai mare în ordine lexicografi că, adică la (n, (n – 1), ..., 1) şi algoritmul se termină.

De exemplu, pentru n=3 se obţin permutările:
(1, 2, 3);      (1, 3, 2);      (2, 1, 3);
(2, 3, 1);      (3, 1, 2);      (3, 2, 1).

În programul ce urmează metoda lexicografi că este realizată cu ajutorul procedu-
rii GenerarePermutari.

Program P163;
 { Generarea permutărilor }
const nmax=100;

type Permutare=array[1..nmax] of 1..nmax;

var P : Permutare;
    n : 2..nmax;
    Indicator : boolean;
    i : integer;

procedure GenerarePermutari(var Indicator : boolean);
label 1;
var i, j, k, aux : integer;
begin
  { permutarea identică }
  if not Indicator then
     begin
       for i:=1 to n do P[i]:=i;
       Indicator:=true;
       goto 1;
     end;

  { căutarea indicelui i }
  i:=n-1;
  while P[i]>P[i+1] do
    begin
      i:=i-1;
      if i=0 then
        begin
          { un astfel de indice nu mai există }
          Indicator:=false;
          goto 1;
        end; { then }
    end; {while }

  { căutarea indicelui k }
  k:=n;
  while P[i]>P[k] do k:=k-1;

  { interschimbarea P[i] - P[k] }
  aux:=P[i]; P[i]:=P[k]; P[k]:=aux;

  { ordonarea ultimilor (n-i) elemente }
  for j:=1 to (n-i) div 2 do
    begin
      aux:=P[i+j];
      P[i+j]:=P[n-j+1];
      P[n-j+1]:=aux;
    end; { for }
  Indicator:=true;
1:end; { GenerarePermutari }

begin
  write(’Daţi n=’); readln(n);
  Indicator:=false;
  repeat
    GenerarePermutari(Indicator);
    if Indicator then
      for i:=1 to n do write(P[i] : 3);
      writeln;
  until not Indicator;
  readln;
end.

Pentru a porni de la permutarea iniţială, înainte de primul apel al procedurii 
GenerarePermutari, parametrului Indicator i se atribuie valoarea false. La 
fi ecare apel procedura înscrie în vectorul P permutarea ce urmează în ordine lexico-
grafi că şi atribuie parametrului Indicator valoarea true. După generarea tuturor 
permutărilor, procedura GenerarePermutari va atribui parametrului Indicator 
valoarea false.

Din păcate, indiferent de metoda folosită, timpul necesar pentru generarea tu-
turor permutărilor este cel puţin O(n!). În consecinţă, algoritmii bazaţi pe căutarea 
soluţiilor prin generarea tuturor permutărilor posibile pot fi  aplicaţi numai pentru 
valori mici ale lui n.

Generarea aranjamentelor. Numărul aranjamentelor de m elemente ale unei 
mulţimi A={a1, a2, ..., an} cu n elemente este dat de formula:

sau, pentru a evita folosirea funcţiei factorial,
Am

n = n· (n – 1)· (n – 2)· ...· (n – m + 2)· (n – m + 1).
Ca şi în cazul permutărilor, putem reduce problema generării aranjamentelor unei 

mulţimi arbitrare A la generarea aranjamentelor mulţimii I={1, 2, ..., n}. De exemplu, 
pentru I={1, 2, 3} şi m=2 cele A2

3 = 6 aranjamente sînt:
(1, 2);     (2, 1);     (1, 3);
(3, 1);     (2, 3);     (3, 2).

Generarea aranjamentelor poate fi  făcută în ordine lexicografi că, pornind de la 
aranjamentul cel mai mic a=(1, 2, ..., m).

Fie a=(a1, a2, ..., ai, ..., am) un aranjament oarecare. Pentru a determina succesorul 
a´ al aranjamentului a, se caută mai întîi cel mai mare indice i cu proprietatea că ai 
poate fi  mărit. Un element ai poate fi  mărit dacă cel puţin una din valorile ai+1, ai+2, 
..., n cu care ar putea fi  înlocuit ai este disponibilă. Pentru a putea efectua mai uşor 
aceste verifi cări, se utilizează vectorul D=(d1, d2, ..., di, ..., dn), unde di este 0 sau 1 în 
funcţie dacă valoarea i apare sau nu în aranjamentul curent a. În momentul în care a 
fost determinat indicele i, elementele ai, ai+1, ..., am vor primi în ordine crescătoare cele 
mai mici numere disponibile.



179

Dacă nu există un indice i cu proprietatea menţionată, înseamnă că s-a ajuns 
la aranjamentul (n–m+1, n–m+2, ..., n), deci procesul generării s-a încheiat. Pentru 
a semnala acest lucru, în programul ce urmează se utilizează variabila booleană 
Indicator.

Program P164;
 { Generarea aranjamentelor }
const nmax=100;
      mmax=100;

type Aranjament=array[1..mmax] of 1..nmax;

var A : Aranjament;
    D : array[1..nmax] of 0..1;
    n : 1..nmax;
    m : 1..nmax;
    i : integer;
    Indicator : boolean;

procedure GenerareAranjamente(var Indicator : boolean);
label 1;
var i, j, k, l : integer;
begin
  { aranjamentul iniţial }
  if not Indicator then
    begin
      for i:=1 to m do
        begin
          A[i]:=i; D[i]:=1;
        end;
     for i:=m+1 to n do D[i]:=0;
      Indicator:=true;
      goto 1;
    end;
  { succesorul aranjamentului curent }
  for i:=m downto 1 do
    begin
      D[A[i]]:=0;
      for j:=A[i]+1 to n do
        if D[j]=0 then
          begin
            A[i]:=j; D[j]:=1; k:=0;
            for l:=i+1 to m do
              begin
                repeat k:=k+1 until D[k]=0;
               A[l]:=k; D[k]:=1;
              end; { for }

            goto 1;
          end; { if }
    end; { for }
  Indicator:=false;
1:end; { GenerareAranjamente }

begin
  write(’Daţi n=’); readln(n);
  write(’Daţi m=’); readln(m);
  Indicator:=false;
  repeat
    GenerareAranjamente(Indicator);
    if Indicator then
      for i:=1 to m do write(A[i] : 3);
    writeln;
  until not Indicator;
  readln;
end.

Aşadar, complexitatea temporală a algoritmilor bazaţi pe generarea tuturor aran-
jamentelor este cel mult O(n!).

Generarea combinărilor. Numărul de combinări de n elemente luate cîte m (mn) 
se calculează ca

De exemplu, pentru I={1, 2, 3} şi m=2 avem C2
3 = 3 combinări:

{1, 2};     {1, 3};     {2, 3}.
Generarea combinărilor poate fi  făcută în ordine lexicografi că, pornind de la com-

binarea iniţială {1, 2, ..., m}.
Fie dată o combinare c={c1, c2, ..., ci, ..., cm}. Combinarea c care îi urmează imediat 

în ordine lexicografi că se determină astfel:
 se stabileşte indicele i care satisface relaţiile ci<n–m+1, ci+1=n–m+i+1, cm-1=n–1, 

cm=n;
 se trece la combinarea c={c1, ..., ci-1, ci+1, ci+2, ..., ci+n–i+1}.
Dacă nu există un indice i care satisface condiţiile de mai sus, înseamnă că au fost 

generate toate combinările.
În programul ce urmează combinările mulţimii I={1, 2, ..., n} se generează conse-

cutiv în vectorul (tabloul unidimensional) C.

Program P165;
  { Generarea combinarilor }
const nmax=100;
      mmax=100;

type Combinare=array[1..mmax] of 1..nmax;

var C : Combinare;
    n : 1..nmax;
    m : 1..mmax;
    i : integer;
    Indicator : boolean;

procedure GenerareCombinari(var Indicator : boolean);
label 1;
var i, j : integer;
begin
  { combinarea initiala }
  if not Indicator then
    begin
      for i:=1 to m do C[i]:=i;
      Indicator:=true;
      goto 1;
    end;
  { succesorul combinarii curente }
  for i:=m downto 1 do
     if C[i]<(n-m+i) then
        begin
          C[i]:=C[i]+1;
          for j:=i+1 to m do C[j]:=C[j-1]+1;
          goto 1;
        end; { then }
  Indicator:=false;
1:end; { GenerareCombinari }

begin
  write(’Daţi n=’); readln(n);
  write(’Daţi m=’); readln(m);
  Indicator:=false;
  repeat
    GenerareCombinari(Indicator);
    if Indicator then
       for i:=1 to m do write(C[i] :3);
    writeln;
  until not Indicator;
  readln;
end.

Se poate demonstra că timpul cerut de un algoritm bazat pe generarea tuturor 
combinărilor este de ordinul O(nk), unde k=min(m, n–m+1), deci polinomial.

Întrebări şi exerciţii
Scrieţi un program PASCAL care afi șează la ecran numărul permutărilor Pn, numărul 

aranjamentelor Am
n și numărul combinărilor Cm

n . Valorile n și m se citesc de la tastatură.
Elaboraţi o procedură recursivă pentru generarea tuturor permutărilor posibile ale mul-

ţimii I={1, 2, ..., n}.
Utilizînd metoda reluării, schiţaţi trei algoritmi pentru generarea, respectiv, a permută-

rilor, aranjamentelor și combinărilor unei mulţimi formate din n elemente distincte.
Se consideră un tablou bidimensional T[1..n, 1..n] format din numere întregi. 

Elaboraţi un program care determină o permutare a coloanelor tabloului astfel încît 
suma componentelor de pe diagonala principală să fi e minimă.

Elaboraţi un program care afi șează la ecran toate șirurile posibile formate din caractere-
le A, b, C, d, E. Fiecare caracter apare în șir numai o singură dată.

Dintr-o listă ce conţine n candidaţi trebuie alese m persoane care vor fi  incluse în echipa 
de fotbal a unui raion. Elaboraţi un program care afi șează la ecran toate modalităţile de 
selecţie a celor m persoane.

Se consideră mulţimea numerelor întregi A={a1, a2, ..., an}. Elaboraţi un program care 
determină o submulţime ce conţine exact m elemente ale mulţimii A astfel încît suma 
lor să fi e maximă.

Cum credeţi, care sînt avantajele și neajunsurile algoritmilor bazaţi pe generarea tuturor 
permutărilor, aranjamentelor și combinărilor posibile?

Există oare tehnici de programare care ar permite evitarea unei analize exhaustive a 
tuturor permutărilor, aranjamentelor sau combinărilor posibile?

Se consideră mulţimea numerelor întregi A={a1, a2, ..., an}. Elaboraţi un program care de-
termină o descompunere a mulţimii A în două submulţimi nevide B și C astfel încît suma 
elementelor din submulţimea B să fi e egală cu suma elementelor din submulţimea C. De 
exemplu, pentru A={–4, –1, 0, 1, 2, 3, 9} avem B={–4, 0, 9} și C={–1, 1, 2, 3}.

Capitolul 7

PROBLEME  RECAPITULATIVE

Problemele ce urmează au fost propuse la diverse concursuri de informatică. 
Rezolvarea lor necesită cunoaşterea profundă a tehnicilor de programare şi a meto-
delor de estimare a complexităţii algoritmilor.

1. Cercuri. Un pătrat cu latura a cm conţine n cercuri (n ≤ 100). Fiecare cerc i este 
defi nit prin coordonatele centrului (xi, yi) şi raza ri. Elaboraţi un program care în cel 
mult t secunde calculează cît mai exact aria obţinută prin reuniunea celor n cercuri.

2. Puncte. Se dau n puncte în plan, n ≤ 100. Să se calculeze numărul maxim de 
puncte coliniare.

3. Poduri. Se consideră n insule legate prin m poduri. E cunoscut faptul că pe 
podul ce leagă insulele i, j pot circula vehicule greutatea cărora nu depăşeşte gij tone. 
Determinaţi greutatea maximă Gab a vehicolului care poate ajunge de pe insula a pe 
insula b.

4. Texte. Se consideră n texte care trebuie tipărite pe foi de hîrtie. Textul i este 
format din ri linii. Pe o foaie pot fi  tipărite cel mult m linii care pot forma unul, două 
sau mai multe texte. Pentru a le separa, între textele de pe aceeaşi foaie se inserează o 
linie vidă. Fragmentarea textelor este interzisă, adică toate liniile oricărui text trebuie 
imprimate pe aceeaşi foaie. Elaboraţi un program care determină numărul minim de 
foi necesare pentru a tipări toate textele.

5. Circumferinţe. Se consideră n puncte pe un plan cartezian. Fiecare punct i este 
defi nit prin coordonatele sale xi, yi. Elaboraţi un program care verifi că dacă punctele 
în studiu aparţin unei circumferinţe.

6. Reţeaua telefonică. Se consideră n oraşe, n  100, care trebuie legate prin cabluri 
într-o reţea telefonică. Pentru fi ecare oraş i sînt cunoscute coordonatele carteziene xi, yi. 
Cablurile telefonice ce leagă oricare două oraşe nu pot avea ramifi caţii. Abonaţii reţelei 
telefonice comunică între ei direct sau prin intermediul staţiilor telefonice din alte oraşe. 
Lungimea cablului care leagă oraşele i, j este egală cu distanţa dintre ele. Determinaţi 
lungimea sumară minimă a cablurilor necesare pentru a conecta toate oraşele.

7. Evaluarea expresiilor. Se consideră expresiile aritmetice formate din numere 
întregi, parantezele (, ) şi operatorii binari +, -, *, mod, div. Scrieţi un program care 
evaluează expresiile aritmetice în studiu.

8. Psihologie. Se consideră n angajaţi, n  100, care trebuie repartizaţi în m echipe. 
Fiecare echipă este formată din k angajaţi, k  m = n. Relaţia de compatibilitate între 
angajaţii i, j se caracterizează prin coefi cientul rij care poate lua valorile 0 (incompa-
tibili), 1, 2, ..., 10 (compatibilitate excelentă). Compatibilitatea întregului colectiv C 

se calculează însumînd coefi cienţii rij pentru toate perechile posibile (i, j) din cadrul 
fi ecărei echipe. Determinaţi compatibilitatea maximă Cmax care poate fi  asigurată prin 
formarea corespunzătoare a echipelor.

9. Compasul. Se consideră n puncte pe un plan cartezian. Fiecare punct i este spe-
cifi cat prin coordonatele sale xi, yi. Elaboraţi un program care verifi că dacă se poate 
desena o circumferinţă cu centrul în unul din punctele în studiu şi care ar trece prin 
toate celelalte puncte.

10. Intersecţia dreptunghiurilor. Se dau n dreptunghiuri (n  10) care au laturile 
paralele cu axele de coordonate, iar coordonatele vîrfurilor sînt numere naturale din 
mulţimea {0, 1, 2, ..., 20}. Elaboraţi un program care calculează aria fi gurii obţinute 
prin intersecţia celor n dreptunghiuri.

11. Reuniunea dreptunghiurilor. Se dau n dreptunghiuri (n  10) care au latu-
rile paralele cu axele de coordonate, iar coordonatele vîrfurilor sînt numere reale. 
Elaboraţi un program care calculează aria fi gurii obţinute prin reuniunea celor n 
dreptunghiuri.

12. Numere prime. Calculaţi toate numerele prime formate din 4 cifre inversul că-
rora este la fel un număr prim, iar suma cifrelor este de asemenea un număr prim.

13. Vizibilitate. Se consideră linia frîntă închisă P1P2 ... PnP1, n  20, care nu se 
autointersectează. În punctul A din interiorul liniei este situat un observator. Pentru 
observator unele segmente ale liniei frînte pot fi  invizibile. Elaboraţi un program 
care calculează numărul segmentelor invizibile.

14. Felinare. Un parc de formă dreptunghiulară este împărţit în pătrate de ace-
leaşi dimensiuni. În fi ecare pătrat al parcului poate fi  instalat cîte un felinar. În ge-
neral, un felinar asigură iluminarea nu numai a pătratului în care el se afl ă, dar şi 
a celor opt pătrate vecine. Elaboraţi un program care determină numărul minim de 
felinare necesare pentru iluminarea parcului.

15. Laserul. Se consideră o placă dreptunghiulară cu dimensiunile mn, unde m 
şi n sînt numere naturale. Această placă trebuie tăiată în mn plăci mai mici, fi ecare 
bucată avînd forma unui pătrat cu dimensiunile 11. Întrucît placa este neomogenă, 
pentru fi ecare bucată se indică densitatea dxy, unde x, y sînt coordonatele colţului 
stînga-jos al pătratului respectiv.

Pentru operaţiile de tăiere se foloseşte un strung cu laser. Fiecare operaţie de tă-
iere include:

 fi xarea unei plăci pe masa de tăiere;
 stabilirea puterii laserului în funcţie de densitatea materialului de tăiat;
 o singură deplasare a laserului de-a lungul oricărei drepte paralele cu una din 

axele de coordonate;
 scoaterea celor două plăci de pe masa de tăiere.
Costul unei operaţii de tăiere se determină după formula c = dmax, unde dmax 

este densitatea maximă a bucăţilor 11 peste marginile cărora trece raza laserului. 
Evident, costul total T poate fi  determinat adunînd costurile individuale c ale tuturor 
operaţiilor de tăiere necesare pentru obţinerea bucăţilor 11. Scrieţi un program care 
calculează costul minim T.

16. Judeţe. Teritoriul unei ţări este împărţit în judeţe (fi g. 7.1). Pe hartă, frontiera 
ţării şi graniţele administrative ale fi ecărui judeţ reprezintă cîte un poligon defi nit 
prin coordonatele (xi, yi) ale vîrfurilor sale. Se presupune că vîrfurile de poligoane 
sînt numerotate direct prin 1, 2, 3, ..., n, iar coordonatele lor sînt numere întregi. În 
interiorul oricărui judeţ nu există alte judeţe.

Fig. 7.1. Judeţele unei ţări

Un virus de calculator a distrus parţial informaţia despre graniţele administrati-
ve, lăsînd intacte următoarele date:

 numărul n şi coordonatele (xi, yi) ale tuturor vîrfurilor de poligoane;
 numărul de segmente m care formează laturi de poligoane şi informaţia despre 

extremităţile fi ecărui segment.
Scrieţi un program care determină numărul de judeţe d şi vîrfurile fi ecărui poli-

gon ce reprezintă o graniţă administrativă de judeţ.
17. Turnuri. Fie n plăci dreptunghiulare numerotate de la 1 la n. Despre placa i se 

ştie că are grosimea hi şi lungimile laturilor xi, yi. Elaboraţi un program care determi-
nă înălţimea maximă a unui turn ce poate fi  construit din plăcile în studiu. Pentru a 
asigura stabilitatea turnului, se vor respecta următoarele reguli (fi g. 7.2):

 plăcile sînt puse una peste alta orizontal, nu pe muchii sau în alt mod;
 muchiile omoloage ale plăcilor suprapuse sînt paralele;
 orice placă din componenţa turnului se va sprijini în întregime pe placa de mai 

jos (evident, placa de la baza tunului se sprijină pe sol);
 nu se cere să folosim toate plăcile.
Date de intrare. Fişierul text TURNURI.IN conţine pe prima linie numărul n. Pe 

fi ecare din următoarele n linii se conţin cîte trei numere întregi pozitive xi, yi, hi sepa-
rate prin spaţiu.

Date de ieşire. Fişierul text TURNURI.OUT va conţine pe o singură linie un număr 
întreg  înălţimea maximă a turnului.

Fig. 7.2. Turn construit din plăci dreptunghiulare

Exemplu:

TURNURI.IN TURNURI.OUT

5
3 4 2
3 4 3
4 3 2
1 5 4
2 2 1

8

Restricţii: n  1000; xi, yi, hi < 1000. Timpul de execuţie nu va depăşi 10 sec.
18. Speologie. Speologul este un specialist care se ocupă cu explorarea şi studie-

rea peşterilor. Antrenarea speologilor presupune parcurgerea unor labirinturi sub-
terane. Un astfel de labirint constă din n, n ≤ 100, peşteri şi coridoare (fi g. 7.3). Fiecare 
peşteră are o denumire individuală formată din cel mult 10 caractere alfanumerice, 
fără spaţii, scrisă pe unul din pereţii ei. Peştera de intrare are denumirea INTRARE, 
iar cea de ieşire – denumirea IESIRE. La intrarea în fi ecare coridor este scrisă denu-
mirea peşterii spre care ea duce.

Speologul nu cunoaşte planul labirintului. În schimb, el este echipat cu un caiet, 
un creion şi o lanternă, fapt ce îi permite să poată citi denumirile de peşteri sau de 
coridoare şi să facă notiţe. Vom numi drum o succesiune de peşteri cu proprietatea 
că între oricare două peşteri consecutive din succesiune există un coridor. Prin lun-
gimea drumului înţelegem numărul de peşteri ce-l formează. De exemplu, drumul 
INTRARE, STALAGMITE, LILIECI, IZVOARE, IESIRE are lungimea 5.

Elaboraţi un program care găseşte unul dintre cele mai scurte drumuri de la peş-
tera INTRARE la peştera IESIRE.

Date de intrare. Fişier de intrare nu există. Totuşi caracteristica peşterii curente 
poate fi  afl ată prin apelul funcţiei predefi nite UndeMaAfl u de tip string. Funcţia 
returnează un şir de caractere ce conţine denumirea peşterii în care în prezent se afl ă 
speologul, două puncte şi denumirile de intrări de galerii, separate prin spaţiu. De 
exemplu, dacă speologul se afl ă în peştera LILIECI, funcţia va întoarce valoarea:

LILIECI: STALAGMITE IZVOARE LILIECI LILIECI

Fig. 7.3. Planul unei peșteri

Trecerea speologului din peştera curentă în peştera spre care duce galeria c se 
realizează prin apelul procedurii predefi nite TreciCoridorul(c), unde c este o 
expresie de tip string. Dacă se indică un coridor inexistent, speologul rămîne pe 
loc. Pentru ca aceste subprograme să fi e accesibile, includeţi în partea declarativă a 
programului de elaborat linia

uses LABIRINT;

Date de ieşire. Fişierul text SPEOLOG.OUT va conţine pe prima linie un număr 
întreg  lungimea celui mai scurt drum. Pe următoarele linii se va scrie drumul. 
Fiecare denumire de peşteră ocupă o linie separată. Dacă un astfel de drum nu exis-
tă, fi şierul va conţine o singură linie cu cuvîntul FUNDAC.

Exemplu. Pentru labirintul din fi gura 7.3 avem:

SPEOLOG.OUT

4
INTRARE
STALACTITE
IZVOARE
IESIRE

Timpul de execuţie nu va depăşi 20 sec.
19. Livada. Planul unei livezi de formă dreptunghiulară cu dimensiunile nm este 

format din zone pătrate cu latura 1. În fi ecare zonă creşte o singură specie de arbori. Orice 
specie de arbori poate ocupa una sau mai multe zone, nu neapărat vecine. Elaboraţi un 
program care găseşte un sector dreptunghiular de arie minimă ce conţine cel puţin k 
specii de arbori. Laturile sectorului vor coincide cu laturile zonelor din plan.

20. Discoteca. La o discotecă participă mai multe persoane, numerotate de la 1 
la n. Iniţial, numai un singur participant, cel cu numărul i, cunoaşte o ştire foarte 
importantă pe care el o comunică prietenilor săi. În continuare, orice participant  
j, care deja cunoaşte această ştire, o comunică, de asemenea, numai prietenilor săi. 
Elaboraţi un program care determină numărul de participanţi p care vor afl a ştirea 
respectivă. Relaţia de prietenie este defi nită prin m perechi distincte de tipul {j, k} 
cu semnificaţia „participanţii j, k sînt prieteni”. Se consideră că 3  n  1000 şi 
2  m  30000.

21. Genistul. O porţiune de drum este împărţită în n segmente. Pe fi ecare seg-
ment poate fi  plantată cel mult o mină. Genistul a memorat informaţia despre minele 
plantate într-un tablou unidimensional M = ||mi||, în care mi = 1 dacă segmentul i 
conţine o mină şi mi = 0 în caz contrar. Pentru orice eventualitate, genistul a cifrat in-
formaţia din tabloul M într-un alt tablou C = ||ci||n componentele căruia se determină 
după cum urmează:

pentru

pentru
pentru

În condiţiile de luptă datele din tabloul M au fost pierdute. Elaboraţi un program 
care calculează tabloul M avînd ca date de intrare tabloul C. Se consideră că proble-
ma are cel puţin o soluţe şi 3  n  10000.

22. Cutii. Se consideră n cutii în formă de paralelipiped dreptunghiular. Pentru 
fi ecare cutie sînt cunoscute cele trei dimensiuni xi, yi, zi. În funcţie de dimensiunile 
cutiilor, unele din ele pot fi  puse una în alta. În general, cutiile pot fi  rotite. Cutia i 
poate fi  pusă în cutia  j  numai atunci cînd în urma tuturor rotirilor posibile se va găsi 
o astfel de poziţie pentru care dimensiunile cutiei i sînt cu stricteţe mai mici decît 
dimensiunile corespunzătoare ale cutiei j. Pentru a asigura o împachetare estetică, se 
cere ca feţele omoloage ale cutiilor să fi e paralele (fi g. 7.4). Mai mult decît atît, unele 
cutii pot fi  imbricate sau, cu alte cuvinte, poate fi  format un şir de numere de cutii 
i1, i2, i3, ..., ik cu proprietatea că cutia i1 se afl ă în cutia i2; cutia i2 se afl ă în cutia i3 etc. 
Scrieţi un program care determină numărul maxim de cutii kmax ce pot fi  imbricate 
într-un astfel de şir.

Fig. 7.4. Cutii imbricate

Date de intrare. Fişierul text CUTII.IN conţine pe prima linie numărul natural n. 
Fiecare din următoarele n linii conţine cîte trei numere naturale xi, yi, zi  separate prin 
spaţiu.

Date de ieşire. Fişierul text CUTII.OUT va conţine pe o singură linie numărul na-
tural kmax. 

Exemplu:

CUTII.IN CUTII.OUT

5
4 4 4
1 3 5
2 2 3
1 1 1
1 1 2

3

Restricţii: 2  n  500; 1   xi, yi, zi  30 000. Timpul de execuţie nu va depăşi 2 se-
cunde.

